
1 Modeliranje i simulacija mikroelektronskih komponenata i kola 

1. 
UVOD U MODELIRANJE I 
SIMULACIJU KOMPONENATA I 
KOLA 

1.1. Uvod u PSPICE okruženje 

PSPICE (PC Simulation Program for Integrated Circuit Emphasis) je program koji 

omogućava sva tri tipa analize elektronskih kola: odreĎivanje radne tačke, tj. analizu u 

jednosmernom režimu, analizu u naizmeničnom režimu i odreĎivanje vremenskog odziva 

kola. Pored njih raspoložive su i opcije koje daju osetljivosti, izobličenja, funkcije prenosa, 

faktor šuma, analizu najgoreg slučaja i slično. Osnovna verzija PSPICE-a je razvijena na 

Univerzitetu u Berkliju još 1975. godine. Simulacija treba da se nadoveže na teorijska 

razmatranja, a da se uradi pre eksperimenta. Svi primeri u ovom materijalu biće dati u 

OrCAD PSpice AD. Program vrši analizu tako što rešava zadato kolo primenom Kirhofovih 

pravila i principa potencijala čvorova. Svaki element kola SPICE „vidi“ kao strujni ili 

naponski generator (Slika 1.1.). Idealni strujni generator je generator koji daje konstantnu 

struju bez obzira na napon na njegovim krajevima, tj. bez obzira na priključeno opterećenje. 

Idealni naponski generator je generator koji daje konstantni napon na svojim krajevima bez 

obzira na struju koja protiče kroz njega. Izvori mogu biti nezavisni, kao opisani i zavisni. 

Zavisni strujni i naponski izvori su u funkciji bilo kog eksternog napona u kolu (na pr. struja 

drejna MOS tranzistora je funkcija napona izmeĎu gejta i sorsa tog tranzistora). 

 
Slika 1.1. Nezavisni idealni strujni (I1) i idealni naponski generator (V1) 

U SPICE-u, u kolima sa pasivnim elementima (R, C, L) vrednosti tih elemenata su parametri 

strujnih i naponskih izvora, kao što ilustruje Slika 1.2. 

Računanje u SPICE-u zasnovano je na matričnom računu. Sistem jednačina koji opisuje 

ponašanje kola rešava se numerički. Izračunavanje DC vrednosti struja i napona kola svodi 

se na iterativno rešavanje sistema linearnih jednačina. Rešavanje sistema nelineranih 

jednačina svodi se na iterativni postupak u okviru koga se više puta rešava sistem linearnih 

jednačina. Sistem diferencijalnih jednačina rešava se diskretizacijom vremenske ose i 

aproksimacijom izvoda količnikom razlike vrednosti promenljivih i razlike vremena u 

susednim tačkama na vremenskoj osi. Najčešće se koristi Newton-ov (Njutnov) metod. 

Princip nalaženja rešenja Njutonovim metodom biće objašnjen na primeru sa Slike 1.3, za 

funkciju sa slike potrebno je naći x = x0 kada je f(x) = y = 0.  Na početku se zadaje željena 

tačnost ε, a rešavanje počinje od pretpostavljenog početnog rešenja xn. Za početno rešenje xn, 

traži se vrednost funkcije f(xn) i u toj tački se postavlja tangenta. Jednačina tangente je: y = 

f ' (xn)(x-xn)+f(xn). Nakon toga, traži se tačka preseka tangente sa x-osom kada je y=0, tako 

da se dobija sledeće rešenje: xn+1=xn-(f(xn)/f ' (xn)). Ova rekurentna relacija poziva se dok se 

ne postigne željena tačnost, tj. ne ispuni uslov │x0-xn+m│< ε. Prilikom simulacije može se 

javiti problem konvergencije ka rešenju. Konvergencija izostaje kada je početno rešenje 

„daleko“ od tačnog.  

 

Slika 1.2. Pasivni elementi kola i njihovo predstavljanje u SPICE-u 

 

Slika 1.3. Ilustracija Newton-ovog metoda 

SPICE za sve početne vrednosti postavlja nulu, sem kada se posebnom naredom .IC (Initial 

Conditions) zadaju početne vrednosti. Usitnjavanje prilikom približavanja rešenju nije 

beskonačno, već se definiše apsolutna tolerancija struja (ABSTOL) i napona (VNTOL). Ovi 

parametri zadaju se komandom .OPTIONS. Sve naredbe, tj. komande u PSpice-u počinju 

tačkom (.). Podrazumevane vrednosti su: ABSTOL=1E-12 i VNTOL=1E-6. Parametar 

RELTOL predstavlja relativnu toleranciju napona i struja, čija je podrazumevana vrednost 

0.001. Pre svake analize, SPICE sprovodi DC analizu, za koju se definiše maksimalan broj 

iteracija. Ovo se definiše parametrom ITL1, čija je podrazumevana vrednost 40. Ukoliko 

SPICE posle 40 iteracija ne pronaĎe rešenje javiće se greška u simulaciji. Ukoliko izostane 

konvergencija, tj. SPICE „javi“ grešku treba proveriti vezu generatora. Često se dodaju 

„veštački“ elementi, koji ne utiču na elemente u kolu, a doprinose da se desi konvergencija. 

Na primer, SPICE ne dozvoljava rednu vezu dva strujna generatora, pa se kao rešenje dodaje 

veštačka grana sa otpornikom velike vrednosti (1E20) ka masi.  

Ukoliko nema problema sa konvergencijom, ili nakon rešenog problema konvergencije, 

simulacija se izvršava. Ukoliko se vrednost napona ili struje menja iz A u trenutku t1 u B u 

trenutku t2, SPICE će izvršiti interpolaciju izmeĎu tih „tačaka“ A i  B – spojiće ih pravom 

linijom. Brze promene u SPICE-u se teže simuliraju, u odnosu na spore, zbog usitnjavanja 

intervala. 

Tri osnovna tipa analize kola su: DC (analiza jednomsernog režima),  TRAN (analiza u 

vremenskom domenu) i AC (analiza naizmeničnog režima, tj. analiza u frekventnom 

domenu). Ilustracija je data na Slici 1.4. Pored osnovnih analiza u SPICE-u je moguće raditi 

analizu osetljivosti, temperaturnu analizu, itd. 

 

Slika 1.4. Tipovi analiza u SPICE-u 

PSPICE netlistu (lista opisa kola) čini popis komponenata u kolu koje se simulira i veza 

izmeĎu njih, model kartice u kojima su dati parametri komponenti, tip analize i izlazne 

komande za prikaz dobijenih rezultata, kao što je ilustrovano na Slici 1.5.  

Prva linija u ulaznom fajlu je za naziv simulacije i komentare, i ne kompajlira se. Ova linija 

se ne koristi za pisanje naredbi. PSPICE nije „case sensitive“, tj. ne pravi razliku izmeĎu 

velikih i malih slova. Čvor 0 je rezervisan za masu, tj. referentni čvor. Simulacija se neće 

obaviti ukoliko nisu svi čvorovi električno povezani sa referentnim čvorom. Svaki čvor u 

kolu mora biti vezan za bar dva elementa. Linija koja počinje sa * se ne kompajlira, tj. ona 

predstavlja komentar. Komentar se može zadati i korišćenjem ; a tekst koji sledi iza „tačka-

zarez“ se smatra komentarom i PSPICE ga ignoriše. Iza opisa kola, dodaju se model kartice 

sa parametrima komponente, o kojima će više reči biti kasnije. Definiše se tip analize, po 

tačno definisanoj sintaksi, kao i izlazne komande, tj. komande za štampanje rezultata. 

Netlista se mora završiti sa .END. Ulazni fajl može sadržati jednu ili više netlista, tj. više 

kola za simulaciju. Ako je zadato više kola, onda ih PSPICE simulira jedno za drugim. 

Ekstenzija ulaznog fajla je *.cir. 

Slika 1.5. Struktura PSpice netliste  

Napomena: Primeri kola u ovom tekstu dati su radi objašnjenja principa modeliranja i 

simulacije. Cilj zadataka nije numeričko rešavanje i analiza funkcionalnosti kola. Kroz 

rešenja zadataka data su objašnjenja sintakse programa PSpice za pojedine komande i 

komponente. 

ZADATAK 1. Upoznati se sa okruženjem programa PSpice A/D. Opisati kolo sa Slike 1.6. i 

uraditi DC analizu ovog kola, ako se vrednost generatora V1 menja od 0 V do 24 V. 

Prikazati rezultate korišćenjem .PROBE.  

 

Slika 1.6. Kolo za simulaciju uz Zadatak 1. 

Rešenje: 

Najpre se označe čvorovi, kao što je prikazano na Slici 1.7. Masa se uvek označava sa 0. 

Ostale čvorove možemo označiti brojevima ili nerezervisanim slovima i rečima. 

Slika 1.7. Označavanje čvorova u kolu 

Pokrenemo program PSpice A/D. Otvorimo novi ulazni fajl: File/New/Text File i napravimo 

netlistu za kolo (Slika 1.8.). Prva linija je komentar i PSPICE je ignoriše. Da je već u prvoj 

liniji bio opisan generator V1, opis kola bio bi nepotpun i javila bi se greška pri pokretanju 

simulacije. U drugoj liniji je korišćenjem naredbe .OPTIONS definisana vrednost relativne 

tolerancije napona i struja, o čemu je već bilo reči. Nakon toga je opisano kolo sa slike 1.7. 

Prvo su definisani nezavisni naponski generatori za koje je rezervisano slovo V. Iza 

rezervisanog slova, bez razmaka, označava se ime generatora (u ovom primeru V1 i V2). 

Treba napomenuti da imena generatora i ostalih komponenti ne moraju da budu brojevi, već i 

nerezervisana slova i reči (na pr. VIN).  Po konvenciji, kod opisa naponskog generatora prvo 

se navodi čvor generatora koji je na višem potencijalu. U ovom primeru, generator se nalazi 

izmeĎu čvorova 1 i 0, pri čemu je čvor 1 na višem potencijalu. Nezavisni generatori se mogu 

koristiti kao izvori jednosmernog napona ili kao generatori različitih talasnih oblika. Prvo se 

navodi DC vrednost generatora (u ovom primeru 24 V). Po analogiji, isto važi i za generator 

V2. Simulator će javiti grešku ukoliko se nezavisni naponski generator definiše sa U, jer je U 

rezervisana oznaka za komponentu (integrisano kolo). 

U nastavku netliste opisani su otpornici, za koje se koristi rezervisano slovo R. U ovom 

primeru nazivi otpornika su: 1, 2 i 3. Struja kroz otpornik R1 teče od čvora 1 ka čvoru 2 i 

vrednost otpornosti je 10 kΩ. PSPICE dozvoljava korišćenje numeričkih sufiksa. Numerički 

sufiks je jedno ili više slova koja se pišu bez razmaka iza broja i menjaju mu vrednost. U 

ovom primeru, k zamenjuje kilo, tj. ·10
3
. Sufiksi su: f- femto, p- piko, n- nano, u- mikro, m- 

mili, k- kilo, Meg- mega, G- giga i T- tera. Još jednom treba napomenuti, PSPICE nije „case 

sensitive“, pa nema razlike izmeĎu g i G. Iz tog razloga je m – sufiks za mili, a meg – sufiks 

za mega. Tekst iza brojne vrednosti koji nije prepoznat kao sufiks smatra se komentarom. Po 

analogiji, isto važi i za otpornike R2 i R3. 

Primer definisanja komponenti – detaljno u [PSpcRef] 

R1 1 2 3.3k 

*| | |  | vrednost 

*| | | - čvor 

*| | + čvor 

*| ključna reč za komponente i ime 

Slika 1.8. Okruženje programa PSpice A/D i netlista za kolo sa slike 7 

Naredba .DC V1 0 24 1 znači da će biti izvršena DC analiza, tako što će se vrednost 

generatora V1 menjati od 0 V do 24 V, sa korakom 1 V. Bez obzira što se vrednost 

generatora V1 menja tokom simulacije, u opisu kola mora biti zadata bilo koja vrednost, 

inače će se javiti greška. Postoji i drugi oblici zadavanja ove komande, što će biti razmtrano 

u narednim zadacima. 

Primer komande .DC – detaljno u [PSpcRef] 

.DC V1 0 5 0.25 

*      | |  | korak 

*      | | krajnja vrednost 

*      | početna vrednost 

 

Komandom .PROBE simulator formira grafički izlazni fajl sa ekstenzijom *.dat koju kasnije 

obraĎuje Probe, grafički postprocesor simulatora PSpice. Ako se ne navede spisak izlaznih 

promenljivih, prikazuju se kompletni rezultati simulacije, tj. vrednosti svih promenljivih iz 

svih obavljenih analiza. U ovom primeru, biće štampane vrednosti napona izmeĎu tačaka 2 i 

0 i tačaka 2 i 3, kao i vrednost struje kroz generator V1. Treba napomenuti da PSpice može 

da štampa direktno vrednosti struja samo kroz generatore, tj. ne može direktno da prikaže 

vrednost kroz na pr. neki od otpornika. Rešenje ovog problema biće opisano u narednim 

zadacima. 

Napomenimo da prilikom opisa kola, nije dozvoljeno koristiti zareze izmeĎu čvorova, već da 

se razdvajanje isključivo vrši blanko znakom (space). U naredbi .PROBE, prilikom 

definisanja koje će veličine biti štampane, potrebno je koristiti zareze: V(2,0). IzmeĎu 

veličina koje će biti štampane koristiti blanko, a ne zarez. Ukoliko su čvorovi označeni 

nizom karaktera, a ne ciframa tada ih treba navoditi u [] zagradama. Na primer, ako je 

generator izmeĎu čvorova IN i 0, naredba za štampu treba da bude: .PROBE V([IN], 

0). Na kraju, napomenimo da PSpice sve napone daje u odnosu na masu, pa je ispravno 

napisati samo V(2) u naredbi .PROBE, umesto V(2,0). 

Lista opisa kola, netlista, završava se komandom .END. Iza ove komande može da se navede 

opis drugog kola. 

Sačuvati ulazni fajl sa ekstenzijom *.cir (zad1.cir) i pokrenuti simulaciju klikom na ikonicu 

Run, ili Simulation/Run zad1.  

Ukoliko tokom komplajliranja ili simulacije detektuje greška, korisnik će dobiti obaveštenje 

u izlaznom fajlu *.out, kao na Slici 1.9. Ovaj fajl je na prvi pogled isti sa ulaznim, pa je česta 

greška korisnika da korekcije vrše u izlaznom umesto u ulaznom fajlu, tako da će se greška 

javljati prilikom novog pokretanja simulatora bez obzira što je ispravljena. Izlazni fajl nakon 

uočene greške treba zatvoriti i korekcije uneti u ulazni fajl *.cir. 

Otvara se novi prozor programa Probe u kome će rezultati biti prikazani. Izaberemo 

Trace/Add Trace... Izborom iz liste u levom delu prozora (Slika 1.10.) promenljive koje se 

štampaju biće dodate u liniju Trace Expression. U ovom primeru: V(2,0), V(2,3). Ukoliko se 

više puta klikne na odreĎenu promenljivu ona će isto toliko puta biti dodata u listu za 

štampanje. Ova opcija ovako funkcioniše jer je moguće formirati izraze čija će vrednost biti 

štampana, a u kome mogu više puta da figurišu ovi naponi i struje. Iz menija sa desne strane 

prozora moguće je izabrati operacije koje bi se izvršile nad izabranim naponima i strujama. 

Nakon dodavanja željenih napona i struja, dobija se grafički prikaz rezultata simulacije, kao 

što je ilustrovano na Slici 1.11. Klikom na ikonicu Toggle cursor omogućeno je precizno 

očitavanje vrednosti. Ukoliko je na istom grafiku dato više veličina, prelazak kursora za 

očitavanje s jedne na drugu vrednost, vrši se klikom na oznaku veličine u donjem levom uglu 

grafika. Moguće je simulirane veličine prikazati u odvojenim prozorima, ali x-osa ostaje 

zajednička: Plot/Add Plot To Window, pa nakon toga Trace/Add Trace. 

 

Slika 1.9. Izlazni fajl, kada simulator prijavi grešku 

Slika 1.10. Izgled prozora Add Traces 

 

Slika 1.11. Prikaz rezultata simulacije 

1.2. Generatori u SPICE-u 

Generatori u SPICE-u se mogu podeliti na dve grupe: nezavisni i zavisni generatori. 

Nezavisni generatori mogu biti strujni i naponski. Nezavisni strujni generator (ICS – 

Independent Current Source) daje konstantnu struju bez obzira na napon na njegovim 

krajevima, tj. bez obzira na priključeno opterećenje. Nezavisni naponski generator (IVS – 

Independent Voltage Source) daje konstantni napon na svojim krajevima bez obzira na struju  

koja protiče kroz njega. Postoji četiri tipa zavisnih generatora: naponom kontrolisani 

naponski izvor (VCVS – Voltage-Controlled Voltage Source), naponski kontrolisani strujni 

izvor (VCCS – Voltage-controlled Current Source), strujom kontrolisani naponski izvor 

(CCVS – Current-Controlled Voltage Source) i strujom kontrolisani strujni izvor (CCCS – 

Current-Controlled Current Source). Kada se u SPICE-u definiše nezavisni generator navodi 

se rezervisano slovo I ili V, nakon toga naziv izvora. Naziv generatora može da bude broj, 

           V 1

0 V 2 V 4 V 6 V 8 V 1 0 V 1 2 V 1 4 V 1 6 V 1 8 V 2 0 V 2 2 V 2 4 V
V ( 2 , 0 ) V ( 2 , 3 )

- 1 2 V

- 8 V

- 4 V

0 V

4 V

8 V

1 2 V

slovo ili reč. Iza blanko znaka se navodi + čvor, pa – čvor i na kraju inicijalna vrednost. Smer 

struje je od + kraja ka – kraju pa treba voditi računa, prilikom definisanja strujnog 

generatora, o redosledu navoĎenja čvorova. Na primeru sa Slike 1.12, strujni generator I1 

treba da bude definisan kao: I1 0 1 1A, da bi smer struje zaista bio ka bazi tranzistora. 

Ukoliko se navede: I1 1 0 1A, tada bi struja isticala iz baze tranzistora. 

 

Slika 1.12. Ilustracija definisanja strujnog generatora 

Primer definisanja nezavisnog naponskog generatora – detaljno u [PSpcRef] 

V1 0 1 5 

*| | | | inicijalna vrednost 

*| | | - čvor 

*| | + čvor 

*| naziv generatora 

 

Primeri zavisnih generatora su bipolarni i MOS tranzistor. Naime, bipolarni tranzistor je 

strujom kontrolisan strujni izvor, tako da srž njegovog SPICE modela čini CCCS generator, 

kod koga se strujom baze kontroliše vrednost struje kolektora. S druge strane, MOS 

tranzistor je naponom kontrolisani strujni izvor, tako da srž njegovog SPICE modela čini 

VCCS generator, kod koga se naponom izmeĎu gejta i sorsa kontroliše struja drejna. Slično, 

dioda je strujni izvor, čija je vrednost odreĎena naponom na samoj diodi, tako da je u suštini 

dioda naponom kontrolisani strujni izvor.  

Postoji više načina definisanja zavisnih generatora. Pored naziva iza ključne reči navode se 

čvorovi, kontrolišući čvorovi i na neki od načina se definiše kako se vrši kontrola. 

Rezervisano slovo za VCVS je E, a za VCCS je G. Generatori koji se kontrolišu strujom, i 

strujni i naponski imaju istu sintaksu. Rezervisano slovo za CCCS i CCVS je F. Kod 

naponski kontrolisanih generatora, na primer: E1 1 2 10 11 5.0, napon izmeĎu 

čvorova 1 i 2 kontroliše se naponom izmeĎu čvorova 10 i 11 i to sa pojačanjem 5.0, tako da 

je V(1,2)=5·V(10,11). Slično i za VCCS: G1 1 2 10 11 5.0. Drugi, najčešće korišćen 

način definisanja vrednosti zavisnih generatora je zadavanjem izraza. Na primer, E1 1 0 

VALUE={5*SQRT(V(3,2))}, što znači da je napon izmeĎu čvorova 1 i 0 odreĎen 

vrednošću izraza koji se navodi u vitličastim zagradama iza ključne reči VALUE. Napon 

izmeĎu čvorova 1 i 0 biće 5·√V, gde je V napon izmeĎu kontrolišućih čvorova 3 i 2. Ostali 

načini defnisanja zavisnih generatora dati su u [PSpcRef]. Strujom kontrolisani generatori se 

definišu navoĎenjem rezervisanog slova F, naziva generatora, čvorova i naziva kontrolišućeg 

generatora, koji se definiše u kolu, kao i pojačanje – tj. koeficijent srazmere izmeĎu 

kontrolišućeg i napona koji se kontroliše. Na primer, F1 1 0 VSENSE 10.0, napon 

izmeĎu čvorova 1 i 0 odreĎen je kao 10·VSENSE.  

ZADATAK 2. Analizirati kolo sa Slike 1.13. u jednosmernom režimu korišćenjem PSPICE-

a.  

 

Slika 1.13. Razdelnik napona 

Rešenje: 

Električna šema kola za simulaciju sa označenim čvorovima data je na slici 1.14. 

 

Slika 1.14. Uz rešenje zadatka 2 

PSpice netlista za dato kolo je: 

Razdelnik napona 

.options reltol=0.01 stepgmin gmin=1e-12 

V1 1 0 10   

R1 1 2 10k 

R2 2 0 10k 

R3 2 0 10k 

.DC V1 0 10 1  

.PROBE V(2) V(1,2) I(V1) 

.END    

 

Napomena: Biće data objašnjenja samo za delove netliste koji prethodno nisu razmatrani. 

Parametar reltol komande .OPTIONS je relativna tolerancija napona i struja, a gmin 

predstavlja minimalnu dozovljenu provodnost grane. Ključnom reči stepgmin se aktivira 

algoritam za menjanje, tj. „stepovanje“ GMIN-a ukoliko postoji problem konvergencije u 

kolu koje se simulira. Naime, SPICE sve vrednosti napona i struja u kolu postavi na 0, i na 

osnovu takvog početnog rešenja, rešava sisteme jednačina kako bi uspešno numerički rešio 

kolo koje se simulira. MeĎutim, ukoliko se desi da nema konvergencije, ne može da naĎe 

rešenje, postoje specijalni algoritmi kojima se rešava ovaj problem. Kada je ovaj algoritam 

aktiviran, sa stepgmin, SPICE u svaki od čvorova postavlja definisanu odvodnost gmin, i sa 

tim vrednostima nalazi početno rešenje. Na osnovu algoritma, otkače se odvodnosti iz kola, i 

ono se ponovo rešava sa početnim rešenjem koje je dobijeno sa priključenim odovdnostima. 

Ovo početno rešenje je sigurno bliže tačnom rešenju od nule, što rešava problem 

konvergencije u SPICE-u. 

ZADATAK 3. Analizirati kolo Vistonovog mosta sa Slike 1.15. u jednosmernom režimu 

korišćenjem PSPICE-a. R4 je otporni senzor pritiska, čija se otpornost menja u zavisnosti od 

pritiska po zakonu: 10k + 0.01√p. Promenu pritiska p simulirati generatorom čija se vrednost 

menja u opsegu 0-100 kPa. Otpornik R4 definisati kao naponom kontrolisani strujni izvor. 

 

Slika 1.15. Vistonov most 

Rešenje: 

Netlista za kolo sa slike 1.15. je: 

Most - simulacija senzora 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-12 

V1 1 0 5 

V2 10 0 0 

R0 10 0 10K 

R1 1 2 10K 

R2 1 3 10K 

R3 3 0 100K 

G1 2 0 VALUE={V(2,0)/(10K+(0.01*SQRT(V(10,0))))} 

.DC V2 0 100K 1K 

.PROBE V(3,2) 

.END 

 

Promenu pritiska (ili bilo koje druge veličine) simuliraćemo posebnim kolom sastavljenim 

od nezavisnog naponskog generatora V2 i otpornika R0, koji su definisani u SPICE-u na već 

poznati način. Ključna stvar je uloga generatora G1 – zavisnog naponski kontrolisanog 

strujnog izvora. Ovim generatorom simulira se promenljivi otpornik u kolu Vistonovog 

mosta, R4 (Slika 1.15.). Dakle, generator se nalazi izmeĎu čvorova 2 i 0 i način promene 

definisan je izmeĎu vitličastih zagrada. Struja koju generator daje biće odreĎena kao odnos 

napona izmeĎu čvorova generatora V(2,0) i otpornosti R4 koja se menja po zakonu  

10k+0.01√p, gde je p – pritisak. Prilikom unošenja izraza za otpornost preporučuje se 

korišćenje zagrada zbog redosleda izvršavanja operacija. Dozovoljeno je korišćenje 

numeričkih prefiksa (10k). SPICE  ima ugraĎenu biblioteku za veliki broj operacija, meĎu 

kojima je korenovanje (SQRT). Otpornost je funkcija pritiska, koji smo definisali 

korišćenjem nezavisnog naponskog generatora V(10,0). Vrednost pritiska p odreĎena je 

vrednošću generatora V(10,0). Ovaj generator ima ulogu kontrolišućeg naponskog 

generatora, a kako je vrednost VALUE po prirodi struja (napon/otpornost), još jednom 

rezimiramo – G1 je naponom kontrolisani strujni izvor.  

Primer definisanja zavisnog naponski kontrolisanog strujnog generatora – detaljno u 

[PSpcRef] 

G1 2 0 VALUE={V(2,0)/(10K+(0.01*SQRT(V(10,0))))} 

*| | | |      |                       |kontrolišući generator 

*| | | |      |izraz 

*| | | | ključna reč 

*| | | - čvor 

*| | + čvor 

*| naziv generatora 

 

Naredbom .DC vršimo promenu napona V2 od 0 do 100k, sa korakom 1k, čime simuliramo 

promenu pritiska. Komandom .PROBE se štampa vrednost izlaznog napona V(3,2). 

Prikazati vrednost izlaznog napona u funkciji pritiska (V2). Analizom rezultata zaključujemo 

da sa povećanjem pritiska, dolazi do smanjenja izlaznog napona, ali da su promene vrlo 

male, pa je signal potrebno pojačati, što će biti razmatrano u narednim zadacima. 

1.3. Potkola u SPICE-u 

Potkolo ili makro čini više elemenata u kolu koji se više puta ponavljaju. Na primer, kolo 

prstenastog oscilatora (ring oscilator) je složeno i sastoji se iz neparnog broja invertora 

povezanih u strukturu prstena. Kod ovog kola grupa elemenata, tj. dva tranzistora (PMOS i 

NMOS) se više puta ponavljaju. Kako bi kolo jednostavnije bilo opisano, bez nepotrebnog 

definisanja velikog broja tranzistora u kolu, formira se potkolo. Potkolo može sadržati 

osnovne elemente (otpornike, kondenzatore, kalemove, diode, tranzistore, ...) ali i druga 

potkola. Na ovaj način se manji deo kola posmatra u složenom kolu kao „crna kutija“, za 

koju su poznati ulazi i izlazi, dok se sama struktura jednom definiše i kasnije se samo poziva. 

Objasnićemo ukratko funkcionalnost kola prstenastog oscilatora prikazanog na slici 1.16. 

Ako je inicijalno stanje na ulazu A1 logička nula, na izlazu ovog kola posle nekog vremena 

(definisanog kašnjenjem invertora) pojaviće se logička 1. Pošto je izlaz invertora A1 

istovremeno i ulaz invertora A2, na izlazu A2 pojaviće se napon logičke nule. Ova logička 

nula, uslovljava logičku jedinicu na izlazu invertora A3, koja se sad prosleĎuje do ulaza u 

invertor A1, koji zbog toga menja stanje na izlazu u logičku nulu i tako redom... Na ovaj 

način formira se oscilatorno kolo, koje se zbog same strukture naziva prstenasti oscilator. Da 

bi ova vrsta oscilatora ispravno radila neophodan je neparan broj invertora u kolu. Na slici 

1.17. prikazana je unutrašnja struktura (CMOS par tranzistora), koja čini potkolo invertora.  

Slika 1.16. Prstenasti oscilator 

 

Slika 1.17. Jedinični invertor (potkolo invertora) sa unutrašnjom strukturom 

Komandom .SUBCKT definiše se struktura potkola, a jednom definisano potkolo poziva se 

kao element rezervisanim slovom X. Kraj opisa potkola označava se komandom .ENDS. 

Prilikom defnisanja potkola iza komande .SUBCKT navodi se ime potkola, pa nakon toga 

čvorovi potkola (pinovi „crne kutije“ dostupni spolja). Postoji mogućnost definsanja 

opcionih čvorova, zatim vrednosti parametara čije se vrednosti mogu preneti u potkolo, što 

će kasnije biti razmatrano. Na primer, za kolo invertora, prva linija u definisanju potkola je: 

.SUBCKT INVERTOR IN OUT VDD GND. Iza komande za početak opisa potkola mogu 

se naći definicije elemenata i/ili pozivi drugih potkola.  

Oznake čvorova koje se koriste pri definiciji potkola su formalne i pri pozivu potkola 

menjaju se zadatim aktuelnim čvorovima. Zato je primer dobre prakse da se pri definiciji 

potkola daju smislena imena. U našem primeru, jasno je da je IN naziv ulaznog čvora, OUT 

naziv izlaznog čvora, a da su čvorovi VDD i GND rezervisani za napajanje. Pri definisanju 

potkola ne treba voditi računa da li će čvor IN biti vezan za čvor 2, 3 ili 5 u glavnom kolu. 

TakoĎe, potkolo se poziva više puta, pri čemu nikad čvor IN neće biti vezan za isti stvarni 

čvor. U jednom delu kola čvor IN biće zamenjen stvarnim čvorom 2, a u drugom delu kola 

biće zamenjen, na primer, čvorom 5. Ne sme se koristiti čvor sa oznakom 0 u definiciji 

potkola, jer je ova oznaka rezervisana za globalni referentni čvor (masu). Redosled 

navoĎenja čvorova pri definisanju potkola nije bitan, ali pri pozivu potkola je od izuzetne 

važnosti.  

Pri pozivanju potkola navodi se rezervisano ključno slovo X i ime te komponente (ne naziv 

potkola) nakon toga čvorovi u stvarnom kolu i na kraju ime potkola. Na primer, X1 1 2 

10 0 INVERTOR. To znači da se poziva potkolo INVERTOR, pri čemu će čvor IN biti 

zamenjen stvarnim čvorom 1 u glavnom kolu, OUT će biti zamenjen sa 2, VDD sa 10, a 

GND sa 0. Ako se nakon toga navede X2 2 3 10 0 INVERTOR, to znači da će biti 

pozvano potkolo INVERTOR, ali da će IN biti zamenjen čvorom 2, OUT čvorom 3, VDD 

čvorom 10 i GND čvorom 0. 

Ukoliko je neki čvor interni, tj. nalazi se izmeĎu elemenata u strukturi potkola, a nije 

dostupan kao izvod „crne kutije“, onda taj čvor figuriše samo u opisu potkola (izmeĎu 

.SUBCKT i .ENDS), a ne i u okoviru .SUBCKT. Na primer, u kolu sa slike 1.18, čvorovi IN, 

VDD i GND biće definisani u okviru .SUBCKT KOLO IN VDD GND, dok će čvor izmeĎu 

otpornika R1 i drejna tranzistora M1 biti definisan samo interno u opisu strukture potkola, 

izmeĎu .SUBCKT i .ENDS. Definisanje strukture potkola biće objašnjeno u zadatku 4. 

 

Slika 1.18. Ilustracija potkola uz objašnjenje globalnih i internih čvorova 

ZADATAK 4. Simulirati kolo razdelnika napona sa slike 1.19, pri čemu paralelnu vezu 

otpornika definisati kao potkolo, koje se poziva dva puta. Vrednost generatora V1 menja se 

od 0 V do 10 V sa korakom 1 V. Otpornost otpornika u potkolu definisati fiksno i 

parametarski. 

 

Slika 1.19. Razdelnik napona (levo – glavno kolo, sredina – potkolo, desno – kompletno 

kolo) 

Rešenje: 

Netlista za kolo sa slike 1.19. je: 

Potkolo 

.options reltol=0.01 stepgmin gmin=1e-12 

.SUBCKT OTP GORE DOLE 

R1 GORE DOLE 5K 

R2 GORE DOLE 5K 

.ENDS 

V1 1 0 5 

X1 1 2 OTP 

X2 2 0 OTP 

.DC V1 0 10 1 

.PROBE V(1,2) V(2) I(V1) 

.END 

 

Nakon definisanja komande .OPTIONS treba definisati potkolo. Potkolo čine dva otpornika 

R1 i R2 vrednosti 5 kΩ. Definicija potkola sa nazivom OTP, pri čemu će formalni nazivi 

čvorova biti GORE i DOLE biće: .SUBCKT OTP GORE DOLE. Dakle, navodi se komanda 

.SUBCKT, naziv potkola i čvorovi. IzmeĎu .SUBCKT i .ENDS, čime se završava opis 

potkola, navodi se opis strukture potkola. U ovom primeru, strukturu potkola čini otpornik 

R1 vezan izmeĎu formalnih čvorova GORE i DOLE sa vrednošću 5K. Otpornik R2 je 

identičan otporniku R1. Nakon definisanja potkola, treba ih pozvati navoĎenjem 

rezervisanog X pri čemu se umesto formalnih, koriste nazivi čvorova u glavnom kolu. Naziv 

potkola koje se poziva navodi se na kraju: X1 1 2 OTP. Biće pozvano potkolo OTP, pri 

čemu će u prvom slučaju čvor GORE biti zamenjen čvorom 1, a DOLE čvorom 2. Dok će pri 

drugom pozivu potkola čvor GORE biti zamenjen čvorom 2, a DOLE čvorom 0. Ostali 

delovi netliste su ranije razmatrani. 

 Opis potkola može se generalizovati tako što će se u potkolo preneti vrednosti parametara 

nabrojane iza ključne reči PARAMS. Imena parametara iza reči PARAMS mogu se proizvoljno 

koristiti u izrazima unutar definicije potkola. U sledećem primeru definisano je potkolo sa 

imenom OTP sa generalizovanim vrednostima otpornosti R1 i R2: 

.SUBCKT OTP GORE DOLE PARAMS: R1=10K, R2=5K 

R1 GORE DOLE {R1} 

R2 GORE DOLE {R2} 

.ENDS 

 

Iza ključne reči PARAMS navedene su inicijalne vrednosti za R1  i R2, pri čemu ovi 

parametri u opisu strukture potkola navedeni izmeĎu vitličastih zagrada. Ovako definisano 

potkolo može se pozvati sa različitim vrednostima parametara R1 i R2, kao što ilustruje 

primer: 

X1 1 2 OTP PARAMS: R1=100K, R2=50K 

X2 2 0 OTP PARAMS: R1=100 

 

U prvom pozivu X1, vrednosti R1 i R2 biće zamenje novim, dok će u pozivu X2 biti 

zamenjena samo vrednost parametra R1, dok će R2 zadržati inicijalnu vrednost. 

Modifikovati netlistu ovog zadatka i proveriti rezultat. 

Primer definisanja potkola– detaljno u [PSpcRef] 

.SUBCKT INVERTOR IN OUT VDD GND PARAMS: Rul=100K 

*|       |         |               |       |inicijalna vrednost parametara 

*|       |         |               | ključna reč pre navođenja parametara 

*|       |         | formalni nazivi čvorova potkola 

*|       | naziv potkola koje se definiše      

*| ključna reč (komanda) pri definisanju potkola 

 

*** OPIS STRUKTURE POTKOLA  *** 

  

.ENDS 

*| kraj opisa potkola 

 

Primer pozivanja potkola– detaljno u [PSpcRef] 

X1 1 2 10 0 INVERTOR PARAMS: Rul=50K  

*| |        |        |       |stvarna vrednost parametara 

*| |        |        | ključna reč pre navođenja parametara 

*| |        | naziv potkola koje se poziva 

*| | čvorovi u glavnom kolu – voditi računa o redosledu! 

*| poziv potkola: X 

 

1.4. SPICE simulacija u vremenskom domenu 

ZADATAK 5. Analizirati osnovno RC kolo sa slike 1.20. u vremenskom domenu 

korišćenjem PSPICE-a. Ekvivalentni kondenzator čini paralelna veza dva konenzatora, 

kapacitivnosti 47 µF i 22 µF.  

 

Slika 1.20. RC kolo 

Rešenje: 

Električna šema kola za simulaciju sa označenim čvorovima data je na slici 1.21. 

Slika 1.21. Uz rešenje zadatka 5 

PSpice netlista za RC kolo je: 

RC kolo 

.OPTIONS numdgt=4 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

V1 1 0 DC 10   ;opis elemenata kola 

R1 1 2 3.3K 

C1 2 0 47U IC=0 

C2 2 0 22U IC=0 

.TRAN .05 1 UIC  ;tip analize 

.PRINT TRAN V(1,2)  ;izlazne komande 

.PROBE V(1,2) 

.END    ;kraj opisa kola 

 

Parametar numdgt komande .OPTIONS predstavlja broj cifara u rezultatima simulacije 

zapisanim u izlaznom fajlu, reltol je relativna tolerancija napona i struja, itl2 je 

maksimalni broj iteracija pri DC analizi, gmin predstavlja minimalnu dozovljenu 

provodnost grane. Rezervisanom reči stepgmin se aktivira algoritam za menjanje, tj. 

„stepovanje“ GMIN-a ukoliko postoji problem konvergencije u kolu koje se simulira. 

Prilikom definisanja generatora V1, oznaka DC nije neophodna ako se definiše samo 

jednosmerna vrednost napajanja.  

Kondenzatori C1 i C2 vezani su paralelno, tj. izmeĎu čvora 2 i referentnog čvora 0. Nakon 

toga definisane su njihove vrednosti. Sufiks U označava ·10
-6

. IC definiše početnu vrednost 

napona na kondenzatoru. U ovom primeru je IC=0, što znači da su kondenzatori prazni. 

Komandom .TRAN zahteva se analiza u vremenskom domenu i zadaju se njeni parametri. 

Prvi u nizu parametara je vrednost koraka štampanja (u ovom primeru 0.05s) rezultata 

simulacije. Nakon toga se definiše vreme završetka simulacije (u ovom primeru 1 s). 

Simulacija počinje u trenutku 0.0 s, pre početka analize u vremenskom domenu obavezno se, 

interno, obavlja DC analiza kako bi se odredili početni rezultati. Kolo se u vremenskom 

domenu opisuje sistemom diferencijalnih jednačina. Sistem se rešava tako što se željeni 

vremenski interval za koji se traži odziv kola diskretizuje na manje intervale shodno koraku 

numeričke integracije. Zatim se u svakoj tački na vremenskoj osi rešava sistem jednačina. 

Rešavanje se svodi na iterativni postupak u okviru koga se potreban broj puta ponavlja 

rešavanje sistema jednačina. Opciono se kao parametar komande .TRAN može zadati vreme 

početka štampanja, tj. vreme koliko dugo od početka simulacije ne treba štampati rezultate. 

Korisnik može da zada i maksimalnu vrednost koraka. UIC (Use Initial Conditions)  nalaže 

simulatoru da pri DC analizi kojom se odreĎuju granični uslovi za analizu u vremenskom 

domenu koristi zadate vrednosti napona na kondenzatorima i (struja kroz kalemove) u kolu. 

Primer komande .TRAN – detaljno u [PSpcRef] 

.TRAN 50ns 10u UIC 

*      |    | trajanje simulacije  

*      | na koliko se štampa nova vrednost 

Komande .PRINT i .PROBE su izlazne komande. Njihovi pozivanjem se štampaju rezultati 

simulacije. Komanda .PRINT štampa specifirane promenljive u obliku tablele u izlaznom 

*.out fajlu. 

Pokrenuti simulaciju i analizirati dobijene rezultate. Može se uočiti da se kondenzatori tokom 

vremena pune po eksponencijalnom zakonu.  

Generator V1 pored jednosmernih vrednosti, može da se definiše kao signal generator (tj. 

generator talasnih oblika) za analizu u vremenskom domenu. Na raspolaganju je pet različitih 

talasnih oblika: PULSE (impulsni, trapezni talasni oblik), SIN (sinusni talasni oblik), PWL 

(segmentno linearni talasni oblik), EXP (eksponencijalni talasni oblik) i SFFM (frekvencijski 

modulisan talasni oblik). Različiti oblici signala generatora biće razmatrani u narednim 

zadacima. Da bi smo analizirali odziv kondenzatora na impulsu pobudu, razmotrićemo način 

definisanja impulsnog generatora.  

Na primer: V1 1 0 PULSE (0 5 1u 100ns 100ns 1u 2u), znači da se impulsni 

generator V1, vezan izmeĎu čvorova 1 i 0, karakteriše prvim naponskim nivoom 0 V, drugim 

naponskim nivoom 5 V, kašnjenje početka generisanja impulsa je 1us, trajanje rastuće i 

opadajuće ivice je po 100 ns, trajanje pozitivnog impulsa (odgovara trajanju napona logičke 

jedinice) je 1 μs, dok je ukupna perioda signala 2 μs. Parametrima kojima se definiše trajanje 

rastuće, opadajuće ivice, kao i trajanje pozitivnog dela impulsa i ukupno trajanje impulsa – 

perioda ne sme se zadati vrednost 0. Trajanje rastuće ivice predstavlja vreme potrebno da 

signal sa nivoa logičke nule poraste do nivoa napona logičke jedinice. Slično, trajanje 

opadajuće ivice predstavlja vreme potrebno da signal opadne sa napona logičke jedinice na 

nivo napona logičke nule. Na slici 1.22. prikazan je oblik ovog signala i označeni pojedni 

parametri.  

Sa slike 1.22. možemo primetiti da „pozitivni“ i „negativni“ impuls ne traju podjednako jer u 

ukupnu peridu nije uračunato vreme trajanja rastuće i opadajuće ivice. „Pozitivan“ impuls 

traje specificiranu 1 μs, dok „negativni“ impuls traje (2 – 1 – 0.2) μs = 0.8 μs, umesto 1 μs. 

Zato je potrebno definisati ukupnu periodu na 2.2 μs. Implementirati impulsni generator u 

netlistu iz zadatka 5, prikazati i analizirati rezultate. 

Slika 1.22. Signal trapeznog oblika 

Primer definisanja impulsnog generatora – detaljno u [PSpcRef] 

V1 1 0 PULSE (-5 1.5V 1u 100ns 100ns 1u 2u) 

*              |   |  |    |     |    |  |perioda 

*              |   |  |    |     |    | trajanje pozitivnog impulsa 

*              |   |  |    |     | trajanje opadajuće ivice 

*              |   |  |    | trajanje rastuće ivice 

*              |   |  | kašnjenje generisanja signala 

*              |   | 2. vrednost 

*              | 1. vrednost 

 

1.5. SPICE simulacija u frekventnom domenu 

ZADATAK 6. Uraditi AC analizu kola filtra propusnika niskih učestanosti, dato na slici 

1.23.  

 

Slika 1.23. Filtar propusnik niskih učestanosti 

Rešenje: 

Na osnovu označenih čvorova, formira se PSpice netlista za kolo sa slike 1.23: 

Filtar propusnik nisikih ucestanosti 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16  

V1 2 1 AC 24 

V2 1 0 DC 24 

L1 2 3 100M 

L2 3 4 250M 

C1 3 0 100U 

R1 4 0 1K 

.AC LIN 30 500 15K 

.PROBE V(2) V(4) 

.END 

 

Generator V1 je nezavisni naponski generator naizmeničnog signala neophodnog za 

simulaciju u frekventnom domenu. Iza čvorova, gde je prvi + kraj, a drugi – kraj generatora, 

navodi se reyervisana reč AC, a iza nje dva parametra talasnog oblika: amplituda (u ovom 

slučaju izražena u V) i početna faza (zadaje se u stepenima). Ukoliko se ne navede, 

podrazumevana vrednost faze je 0 °. 

Primer definisanja AC signala – detaljno u [PSpcRef] 

V11 2 1 AC 50mV 90  

*            |  | faza 

*            | amplituda signala 

Komandom .AC zahteva se analiza u frekvencijskom domenu i zadaju se potrebni parametri 

analize. Mora se izabrati jedan od načina promene frekvencije: linearan LIN, logaritamski po 

oktavi OCT ili logaritamski po dekadi DEC. Nakon toga navodi se u koliko tačaka treba 

analizirati kolo. Ukoliko se izabere LIN  ova vrednost je broj tačaka na frekvencijskoj osi, a 

ako se izabere OCT ili DEC onda ova vrednost predstavlja broj tačaka po oktavi, odnosno 

dekadi. Na kraju se navode početna i krajnja vrednost frekvencije. U ovom primeru, 

definisana je AC linearna analiza, u 30 tačaka, počevši od 500 Hz do 15 kHz. Napomenimo, 

svakoj AC analizi prethodi DC analiza kako bi se odredile vrednosti dinamičkih parametara 

nelinearnih elemenata u ranoj tački. 

 

Primer komande .AC – detaljno u [PSpcRef] 

.AC DEC 5 100Hz 100Meg 

*   |   |   |     | krajnja frekvencija 

*   |   |   | pocetna frekvencija 

*   |   | broj koraka 

*   | način promene frekvencije 

 

Komandom .PROBE štampaju se vrednosti napona u tačkama V(2) – ulazni napon i V(4) – 

izlazni napon. Pokrenuti simulaciju i analizirati rezultate. Analiziraćemo propusni opseg 

ovog filtra, na osnovu odnosa izlaznog i ulaznog napona. Za prikazivanje vrednosti izraza u 

programu PROBE kliknuti: Trace/Add Trace. U prozoru koji se otvara uneti vrednost izraza, 

korišćenjem ponuĎenih operacija s desne strane i izračunatih promenljivih ponuĎenih s leve 

strane prozra (Slika 1.24). U ovom primeru, prikazivaćemo vrednost izraza 

20*LOG10(V(4)/V(2)). Analizom zaključujemo da ovo kolo radi kao filtar propusnik niskih 

učestanosti. Tipično, ovaj filtar se koristi u ispravljačkim kolima da propusti DC i blokira 

visokofrekventne harmonike (naponske pikove) generisane u procesu AC-DC ispravljanja. 

Propusni opseg kola odreĎen je rezonantnom učestanošću kola, čija je vrednost definisana 

vrednostima L i C u kolu. 

 

Slika 1.24. Rezultat simulacije i izgled prozora za unos izraza koji će biti prikazan 

1.6. Zadaci za vežbu 

ZADATAK 7. Odrediti vrednost struje I u kolu sa slike 1.25, za vrednosti generatora V1: 

1 V, 3 V i 5 V. Vrednost svih otpornosti u kolu je 1 kΩ.  

 
Slika 1.25. Otporna mreža 

 

Uputstvo: Postprocesor PROBE može da prikaže struje samo kroz grane kola sa 

generatorima. Ako se upotrebi nezavisni strujni generator, da bi simulacija bila pokrenuta 

mora se definisati struja koju daje taj generator, a samim tim bi bila definisana struja u toj 

grani kola. Iz tog razloga treba definisati nezavisni naponski generator u grani kola kroz koju 

se odreĎuje struja, pri čemu inicijalna vrednost tog generatora treba da bude 0. Na ovaj način 

se ne narušava regularni rad kola (generator je kratak spoj), a dobijaju se željeni rezultati. U 

.PROBE treba prikazati struju kroz fiktivni generator. U stranoj literaturi ovi generatori se 

nazivaju dummy generatori. 

Rešenje: 
 

Zadatak 7. Otporna mreza 

* odredjivanje struje kroz granu u kojoj nema generatora 

.options reltol=0.01 stepgmin gmin=1e-16  

V1 1 0 3  

R1 1 2 1K 

R2 2 0 1K 

R3 2 3 1K 

R4 1 3 1K 

V0 3 0 0 

.DC V1 1 5 2  

.PROBE I(V0) 

.END 

 

ZADATAK 8. Analizirati odziv kalema na impulsnu pobudu u kolu sa slike 1.26. Generator 

V1 definisati kao impulsni generator sa parametrima: nivo logičke nule 0 V, nivo logičke 

jedinice 1 V, signal se generiše od početka simulacije, vreme porasta prednje i zadnje ivice je 

po 100 ns, duty cycle je 50 %, a period signala je 200 µs. Vrednosti otpornosti i induktivnosti 

date su na slici 1.26. Prikazati ulazni napon, napon i struju kroz kalem za vreme trajanja 5 

impulsa generatora V1.  

 
Slika 1.26. RL kolo 

 

Uputstvo: Prilikom definisanja impulsnog generatora V1 vreme trajanja pozitivnog implusa 

odrediti na osnovu poznatog duty cycle-a i periode. Pri definisanju periode u obzir uzeti i 

vremena porasta prednje i zadnje ivice, kako bi signal generatora V1 bio simetričan, tj. da 

duty cycle zaista bude 50%. Ukupno trajanje simulacije treba da bude 5 izračunata perioda. 

Parametar u okviru .TRAN koji definiše na koliko vremena će se štampati vrednost napona i 

struja odrediti proizvoljno. 

Rešenje: 

Zadatak 8. RL kolo - odziv na impulsnu pobudu 

* ilustracija definisanja impulsnog generatora 

.options reltol=0.01 itl2=120 stepgmin gmin=1e-12  

V1 1 0 PULSE (0 1 0 100N 100N 100U 200.2U) 

R1 1 2 100 

L1 2 0 1M 

.TRAN 100N 1001U UIC 

.PROBE V(1) V(2) I(V1)  

.END 

ZADATAK 9. Analizirati kolo filtra propusnika visokih učestanosti sa slike 1.27. Vrednosti 

elemenata  date su na slici 1.27. Generator V1 definisati kao generator naizmeničnog AC 

signala amplitude 1 V, faze 0 º. Kolo analizirati u 10 tačaka po oktavi u frekvencijskom 

opesegu od 1 Hz do 1 MHz. Prikazati odnos napona V(OUT)/V(IN). Odrediti donju graničnu 

frekvenciju filtra. 

 

Slika 1.27. Filtar propusnik visokih učestanosti 

Uputstvo: Predvideti potrošač u čvoru OUT (RLOAD=1E12). Alternativno, ovo kolo može se 

opisati kao potkolo koga čine kondenzator redno vezan sa paralelnom vezom kalema i 

otpornika. Ovo potkolo pozvati tri puta, pri čemu vrednosti elemenata treba zadati 

parametarski, s obzirom da nisu isti u svakom od segmenata kola. 

Rešenje: 

Zadatak 9. – osnovno resenje 

* ilustracija .AC analize 

.OPTIONS RELTOL=0.01 STEPGMIN 

+GMIN=1E-12 

V1 1 0 AC 1 0 

C1 1 2 0.47U 

L1 2 0 100M 

R1 2 0 6.8K 

C2 2 3 0.1U 

L2 3 0 56M 

R2 3 0 4.7K 

C3 3 4 0.1U 

R3 4 0 6.8K 

L3 4 0 100M 

C4 4 5 0.47U 

RLOAD 5 0 1E12 

 

.AC OCT 10 1 1MEG 

.PROBE V(1,0) V(5,0)  

.END 

 

Zadatak 9. – resenje s potkolom 

.OPTIONS RELTOL=0.01 STEPGMIN 

+GMIN=1E-12 

V1 1 0 AC 1  

X1 1 0 2 RLC PARAMS: C=0.47U 

X2 2 0 3 RLC PARAMS: L=56M,  

+R= 4.7K 

X3 3 0 4 RLC 

C4 4 5 0.47U 

Rout 5 0 1E12 

 

.SUBCKT RLC K1 K2 K3 PARAMS: 

+R=6.8K, +L=100M, C=0.1U 

C1 K1 K3 {C} 

L1 K3 K2 {L} 

R1 K3 K2 {R} 

.ENDS 

 

.AC OCT 10 1 1MEG 

.PROBE V(5) V(1) 

.END 

2. 
MODELIRANJE I SIMULACIJA 
DIODA 

2.1. Dioda kao potkolo 

Model bilo koje komponente treba da opiše njeno realno ponašanje. Modeli komponenata su 

u suštini potkola sastavljena iz različitih elemenata koji povezani na odreĎeni način 

predstavljaju ekvivalentno kolo komponente. Parametri modela odrediće karakteristike 

komponente koja je modelirana, a promenom vrednosti tih parametara postiže se uparivanje 

simuliranih sa realnim (merenim) karakteristikama. Na primer, nagib strujno naponske 

karakteristike diode odreĎen je otpornošću diode, dok je napon voĎenja odreĎen parametrima 

materijala od kog je komponenta napravljena (slika 2.1). Dakle, otpornost diode je jedan od 

parametara modela diode. 

 

Slika 2.1. Ilustracija različitih I-V karakteristika diode odreĎenih različitim parametrima 

modela 

Najprostiji model diode (ekvivalentno kolo) prikazan je na slici 2.2. Dioda D1 se može 

modelirati pomoću naponom kontrolisanog strujnog izvora, otpornika i kondenzatora, 

vezanih izmeĎu anode (A) i katode (K). Suštinu modela čini naponom kontrolisani strujni 

generator G kojim je definisana struja diode, koja zavisi od napona na diodi. Redna otpornost 

diode (otpornost pn spoja, kontakata i izvoda) definisana je otpornikom rd, dok je ukupna 

kapacitivnost pn spoja, koja je nelinarna i zavisna od napona izmeĎu anode i katode diode 

predstavljena kondenzatorom Cd.  

Slika 2.2. Osnovno ekvivalentno kolo diode 

ZADATAK 10. Opisati silicijumsku diodu kao potkolo u PSpice-u. Definisati kolo za 

simulaciju strujno naponske karakteristike diode u kome će ona biti pozvana kao potkolo. 

Poznati su parametri modela: inverzna struja zasićenja (struja curenja diode) IS=1 pA, 

termički potencijal na sobnoj temperaturi VT=26 mV, redna otpornost diode RD=100Ω. U 

prvoj aproksimaciji uzeti konstantnu kapacitivnost pn spoja CD=1 pF. 

Rešenje: 

Definisaćemo potkolo sa nazivom DIODA i dostupnim čvorovima nazvanim ANODA i 

KATODA. U definiciji potkola zadaju se inicijalne vrednosti parametara IS,VT, RD i CD, 

korišćenjem ključne reči PARAMS:. Osnova potkola sa slike 2.2 biće naponom kontrolisani 

strujni generator, čija će vrednost biti definisana korišćenjem ključne reči VALUE=, a 

opisana je jednačinom struje diode: 

     ( 
  
    )  

Jedina promenljiva je napon na diodi, tj. struja VCCS biće odreĎena naponom na samom 

generatoru, dok će inverzna struja zasićenja i termički potencijal biti prosleĎeni kao 

parametri. Generator je vezan izmeĎu internog čvora (na pr. označenog sa 10) i čvora 

KATODA. Kontrolišući napon je napon izmeĎu + čvora (10) i – čvora (KATODA). 

Elementi potkola sa slike 2.2 su i redna otpornost diode koja je vezana izmeĎu čvora 

ANODA i internog čvora (10) i uzima vrednost parametra {RD}; slično kao i kapacitivnost 

diode, predstavljena kondenzatorom vezanim paralelno sa generatorom čija je vrednost 

{CD}. Nazivi parametara čije vrednosti uzimaju elementi kola navode se izmeĎu vitličastih 

zagrada: {RD}, {CD}. Predefinisanje parametara omogućava laku manipulaciju u netlisti u 

smislu promene vrednosti pojedinih parametara. U suprotnom, bilo bi potrebno na više mesta 

promeniti vrednost odreĎenog elementa. SPICE netlista potkola diode je: 

.SUBCKT DIODA ANODA KATODA PARAMS: IS=1P, VT=26M, R0=100, C0=1P 

RD ANODA 10 {R0} 

CD 10 KATODA {C0} 

G1 10 KATODA VALUE={IS*(EXP(V(10,KATODA)/VT)-1)} 

.ENDS 

 

Napomenimo, ukoliko je potrebno razmatrati temperaturnu zavisnost karakteristike diode, 

tada bi termički potencijal bio predstavljen izrazom VT=k·T/q. U tom slučaju bilo bi 

potrebno definisati konstante k (Bolcmanova konstanta) i q (elementarno naelektrisanje), a T 

(temperatura) bila bi još jedna promenljiva u izrazu za struju diode. TakoĎe, kao što je 

napomenuto kapacitivnost diode je zavisna od napona na diodi, pa bi je trebalo predstaviti 

kao strujni izvor čija je vrednost srazmerna sa prvim izvodom napona na diodi po vremenu, a 

koeficijent srazmere bila bi fiksna kapacitivnost... Definisanje izvoda u SPICE-u je 

komplikovano, pa je u prvoj aproksimaciji kapacitivnost predstavljena kondenzatorom fiksne 

kapacitivnosti. 

Osnovno kolo za simulaciju strujno naponske karakteristike diode prikazano je na slici 2.3. 

Dioda će biti pozvana korišćenjem X1 kao potkolo sa nazivom DIODA, pri čemu čvor 1 u 

glavnom kolu odgovara čvoru ANODA u potkolu, dok čvoru 0 odgovara čvor KATODA. Da 

bi simulirali strujno naponsku karakteristiku biće zadata DC analiza, u kojoj će se vrednost 

generatora VD menjati od 0 V do 2 V sa korakom 0.05 V. U .PROBE treba prikazati struju 

kroz generator VD, što je i struja kroz diodu. Kompletna netlista je: 

Dioda kao potkolo 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-16 

VD 1 0 1 

X1 1 0 DIODA 

 

.SUBCKT DIODA ANODA KATODA PARAMS: IS=1P, VT=26M, R0=100, C0=1P 

RD ANODA 10 {R0} 

CD 10 KATODA {C0} 

G1 10 KATODA VALUE={IS*(EXP(V(10,KATODA)/VT)-1)} 

.ENDS 

 

.DC VD 0 2 0.05 

.PROBE I(VD) 

.END 

  

 

Slika 2.3. Kolo za simulaciju I-V karakteristike diode 

Pokrenuti simulaciju i prikazati karakteristiku. U Trace Expression uneti –I(VD). Obratiti 

pažnju na red veličine struje kroz diodu (desetak 10 mA). Promeniti otpornost diode RD=1 i 

ponoviti simulaciju. Sada će struja kroz diodu biti daleko veća u poreĎenju sa prethodnim 

rezultatima. 

2.2. Parametri modela diode 

U jednačini struje diode jedan od parametara je inverzna struja zasićenja (struja curenja) koja 

zavisi od koncentracije manjinskih nosilaca, difuzionih dužina koje su funkcija vremena 

života nosilaca: 

. 

Visoki nivoi injekcije nosilaca uzrokuju pojavu zasićenja struje. Naime, struja ne može 

beskonačno da raste sa povećavanjem napona, već ulazi u zasićenje (ograničena je). Zbog 

velike koncentracije nosilaca blokira se dalji protok struje, tako da ma koliko bila mala redna 

otpornost diode, celokupni napon pašće na ovu otpornost. Pri odreĎenom kritičnom naponu 

disipacija snage postaje velika što dovodi uništenja (pregorevanja) diode. U jednačinu za 

struju diode uvodi se korekcija, kako bi efekti visoke injekcije bili uključeni: 

. 

U ukupnu struju diode, pri definisanju modela ulaze i generaciono-rekombinacone struje, kao 

i struja pri proboju. Postoje dva mehanizma proboja: lavinski i tunelski (Zenerov) proboj. Na 

primer, napon lavinskog proboja zavisi od materijala od koga je dioda napravljena (εs), 

srazmeran je sa kvadratom električnog polja pri proboju (Fbr
2
), a zavisi i od koncentracije 

nosilaca (Nd): 

. 

Kada se uzmu u obzir navedeni uticaji, može se definisati I-V model diode, koji daje 

zavisnost struje kroz diodu u funkciji napona, pri čemu u izrazima figurišu parametri modela 

diode. Dakle, strujno naponska karakteristika diode može se fitovati sa eksperimentalnim 

rezultatima promenom sledećih parametara: IS – inverzna struja zasićenja, IKF – struja u 

kolenu karakteristike pri visokim nivoima injekcije nosilaca, ISR – parametar generaciono-

rekombinacionih (G/R) struja, VJ – ugraĎeni potencijal, NR – emisioni koeficijent za G/R 

struje, M –koeficijent koji definiše strmost pn spoja, IBV – struja u kolenu karakteristike pri 

proboju i BV – probojni napon (u kolenu karakteristike): 

. 

Napomenimo, ove jednačine su „ugraĎene“ u SPICE. Dizajner modela i/ili korisnik zadaju 

parametre na osnovu kojih SPICE izračunava struju kroz diodu. Parametre modela daje 

proizvoĎač tako da karakteristike odgovaraju realnim karakteristikama diode datog 

proizvoĎača. Ove parametre može zadati/promeniti i korisnik, na osnovu eksperimentalnih 

rezultata ili tehničke dokumentacije proizvoĎača. Ovi parametri zadaju se u okviru tzv. 

modelskih kartica o kojima će biti reči u nastavku.   

C-V model diode se definiše na osnovu poznatih kapacitivnosti prelazne (osiromašene) 

oblasti i difuzione kapacitivnosti. Kapacitivnost prelazne oblasti zavisi od njene širine w, od 

površine pn spoja i značajna je pri inverznoj polarizaciji. Ova kapacitivnost obrnuto je 

srazmerna sa korenom iz napona: 

. 

Ova kapacitivnost utiče na frekventni odziv diode, na ovom efektu zasniva se rad varikap 

(varaktor) diode čija se kapacitivnost menja u zavisnosti od inverznog napona na koji su 

priključene. SPICE „ugraĎena“ jednačina za kapacitivnost prelazne oblasti je: 

gde su: CJO – kapacitivnost osiromašene oblasti kada dioda nije polarisana, VJ – kontaktni 

(ugraĎeni) potencijal, FC – koeficijent kapacitivnosti osiromašene oblasti pri direktnoj 

polarizaciji i M –koeficijent koji definiše strmost pn spoja.   

Difuziona kapacitivnost je značajna pri direktnoj polarizaciji diode. Zavisi od brzine prolaska 

nosilaca kroz prelaznu oblast i obrnuto je srazmerna dinamičkoj otpornosti diode. Ova 

kapacitivnost uslovljena je povećanom koncentracijom manjinskih nosilaca uz prelaznu 

oblast, i u SPICE-u je odreĎena kao proizvod parametra TT – Transient Time i provodnosti 

pn spoja: 

 

Parametar TT definiše brzinu rada diode i predstavlja vreme potrebno da nosioci preĎu iz p- 

u n- oblast i obrnuto. Što je ovo vreme veće, biće veća kapacitivnost, koja ne stiže da se 

isprazni brzo što uzrokuje sporiji odziv diode i obrnuto. Difuziona kapacitivnost zajedno sa 

dinamičkom otpornošću odreĎuje admitansu diode, a time i ograničava maksimalnu 

frekvenciju njene primene. Proširen model diode za visoke frekvencije dat je na slici 2.4. Na 

visokim frekvencijama, definiše se tzv. ekvivalentno kolo za male signale u kome pored 

redne otpornosti, VCCS (na slici 2.4 predstavljen kao provodnost Gd) i paralelne 

kapacitivnosti (kapacitivnost osiromašene oblasti i difuziona kapacitivnost) figurišu i 

geometrijska kapacitivnost (Cgeom) koja uzima uticaj geometrije diode i parazitna 

induktivnost (Ls) koja potiče od inuktivnosti kontakata i izvoda.  

 

Slika 2.4. Ekvivalentno kolo diode za male signale 

Napomenimo, temperaturne zavisnosti karakteristike diode se takoĎe mogu simulirati u 

SPICE-u, detaljnije u [PSpcRef]. Temperaturni koeficijent diode (parametar) označen je sa 

XTI i definiše zavisnost inverzne struje zasićenja, a time i struje diode od temperature. 

2.3. Modelska kartica diode 

Vrednosti parametara matematičkog modela elemenata u kolu mogu se definisati na dva 

načina u zavisnosti od složenosti modela. Ukoliko se radi o elementu koji se opisuje samo 

jednim parametrom, tada se navodi samo brojna vrednost tog parametra. U slučaju da model 

elementa kola sadrži veći broj parametara ovaj način nije pogodan, pa se zadavanje vrednosti 

parametara vrši preko modelske kartice. Na primer, za otpornik ne definišemo posebno 

modelsku karticu jer je ova komponenta u potpunosti, u najvećem broju slučajeva, definisana 

ako se zada njena vrednost otpornosti. To ne znači da nije moguće definisati modelsku 

karticu i za pasivne komponente poput otpornika, kondenzatora ili kalemova; i za ove 

komponente postoji skup parametra detaljno objašnjen u [PSpcRef]. S druge strane, na 

primer dioda je odreĎena sa desetak parametara, pa je neophodno opisivanje pomoću 

modelske kartice. 

Napomenimo, složene komponente se mogu opisati kao potkola – kao što je pokazano, ali u 

najvećem broju slučajeva se koriste „ugraĎene“ jednačine kojima se prosleĎuju parametri 

modela. Na ovaj način, netlista je preglednija, rasterećenija, a rezultati su precizniji, jer 

jednačine, kao što je takoĎe pokazano, obuhvataju više raznih uticaja na ponašanje 

komponente. Kada bi se svi uticaji uzeli u obzir, konkretno za diodu, ekvivalentno kolo koje 

bi bilo definisano kao potkolo bilo bi daleko složenije od prikazanog u zadatku 10.  

Za definisanje modelskih kartica koristi se komanda .MODEL. U okviru definicije modelske 

kartice specificira se njeno ime, navodi se tip modelske kartice i vrednosti parametara 

modela. Vrednosti parametara modela navode se izmeĎu malih zagrada. Postoje rezervisane 

reči za tip modelske kartice, tj. vrste komponente. Na primer, D – dioda, RES – otpornik, 

CAP – kondenzator, IND – kalem, NMOS – n- kanalni MOS, PMOS – p- kanalni MOS, 

NPN, PNP – za bipolarne tranzistore [PSpcRef]... Za svaki tip modelske kartice, tj. vrstu 

komponente definisan je skup parametara koji se mogu koristiti. Na primer, ako opisujemo 

diodu, a u okviru modelske kartice definišemo parametar NMOS tranzistora, koji nije 

predviĎen u modelu diode, SPICE će javiti grešku – nepoznati parametar. Ovde su navedeni 

najbitniji parametri modela diode, kompletan spisak parametara diode dat je u [PSpcRef]. U 

toku kursa, biće dat pregled najvažnijih parametara modela za komponente koje budu 

korišćene u kolima. 

Primer modelske kartice diode D1N4148 

.MODEL D1N4148 D (IS=0.1PA RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

* |    |       |  |vrednosti parametara modela 

* |    |       |tip modelske kartice, tj. vrsta komponente 

* |    |naziv modela 

* |naredba  

 

Naziv modela može biti proizvoljno izabran. Parametri modela koji nisu obuhvaćeni se u 

simulaciju (jednačine) uključuju sa podrazumevane (default-nim) vrednostima, koje su date 

tabelarno u [PSpcRef] za sve komponente. Model diode iz primera opisan je sa 6 parametara: 

inverznom strujom zasićenja (IS), rednom otpornošću (RS), kapacitivnosti osiromašene 

oblasti kada nema polarizacije (CJ0), vremenom prelaska nosilaca od anode do katode (TT), 

probojnim naponom (BV) i strujom proboja (IBV). Parametri koji figurišu u „ugraĎenim“ 

jednačinama, a koji nisu navedeni uzimaju podrazumevane vrednosti. Na primer, IKF će 

imati vrednost beskonačno, dok će vrednost korekcionog faktora M biti 0.5 [PSpcRef]. 

Napomenimo, ista modelska kartica može poslužiti za opis više elemenata koji imaju 

ekvivalentni matematički model. 

Primer definisanja (pozivanja) diode – detaljno u [PSpcRef] 

D1 1 0 D1N4148 

*| | | |naziv modela 

*| | |katoda 

*| |anoda 

*|rezervisano slovo D i naziv diode 

 

Vrednosti parametara modela pridružuju se opisu elemenata navoĎenjem imena 

odgovarajuće modelske kartice. Pri definisanju (pozivanju) diode, navodi se ključno 

rezervisano slovo D i naziv diode, zatim čvor anode, čvor katode i na kraju naziv modela. Na 

primer, SPICE će diodu D1, čija je anoda na čvoru 1, a katoda na čvoru 0, posmatrati kao 

diodu čiji su parametri definisani u modelskoj kartici po nazivom D1N4148. Modelska 

kartica treba da bude definisana u istom ulaznom fajlu, najčešće iza opisa kola. Parametri 

definisani u modelskoj kartici se prosleĎuju „ugraĎenim“ jednačinama diode, s obzirom da je 

modelska kartica D1N4148 tipa D- dioda. 

ZADATAK 11. Korišćenjem modelske kartice pod nazivom D1N4148 (primer), opisati kolo 

sa slike 2.3 i  prikazati strujno naponsku karakteristiku diode u direktnom i inverznom 

režimu rada. Promeniti vrednost parametra BV=50 i ponoviti simulaciju. Diskutovati 

rezultat. 

Rešenje: 

Za kolo sa slike 30 definiše se sledeća netlista: 

Modelska kartica diode 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-16 

VD 1 0 1 

D1 1 0 D1N4148 

 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

.DC VD -105 2 0.05 

.PROBE I(VD) 

.END  

 

Da bi bila prikazana karakteristika pri direktnoj i inverznoj polarizaciji, u .DC analizi zadato 

je da se vrednost generatora VD menja od -105 V do 2 V sa korakom 0.05 V. Isti rezultat 



 
32  

dobija se ako se definiše .DC VD 2 -105 -0.05. Rezultat simulacije pokazuje da je 

probojni napon jednak 100 V, kao što je definisano u modelu, kao i da je pri tom naponu 

struja 0.1 pA (slika 2.5). Kada se promeni vrednost probojnog napona na BV=50 V, rezultat 

simulacije pokazuje ovaj napon proboja. 

 

Slika 2.5. Rezultati simulacije za BV=100 V 

ZADATAK 12. Simulirati kolo naponskog regulatora – klipera sa četiri identične diode 

(slika 2.6) u vremenskom domenu u ukupnom trajanju 20 ms. V1 generiše signal sinusnog 

oblika, amplitude 10 V, frekvencije 100 Hz. Vrednost otpornosti R1 je 1 kΩ. Prikazati ulazni 

napon V1 i izlazni napon (Vout):  

a) Ako se koristi modelska kartica 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 

IBV=0.1PA) 

b) Ako se dioda opiše potkolom (osnovno ekvivalentno kolo diode čiji su parametri 

IS=0.1 pA, VT=26 mV, RD= 16 Ω, CD=2 pF). 

Diskutovati rezultate, koje napone „odseca“ ovo kolo i zašto? Koliki su naponi voĎenja 

dioda opisanih pod a) i b). 

 

Slika 2.6. Naponski regulator – klipersko kolo 

 

           V D

- 1 1 0 V - 1 0 0 V - 9 0 V - 8 0 V - 7 0 V - 6 0 V - 5 0 V - 4 0 V - 3 0 V - 2 0 V - 1 0 V 0 V 1 0 V

- I ( V D )

- 3 0 0 m A

- 2 0 0 m A

- 1 0 0 m A

0 m A

1 0 0 m A

Rešenje: 

Generator V1 koji je vezan izmeĎu čvorova 1 i 0 treba da generiše napon sinusnog oblika. 

Sinusni talasni oblik se definiše navoĎenjem ključne reči SIN i parametara signala izmeĎu 

malih zagrada. 

Primer definisanja generatora sinusnog talasnog oblika – detaljno u [PSpcRef] 

V1 1 0 SIN (0 10V 100Hz 0 0 0) 

*           |  |  |     | | |faza 

*           |  |  |     | |faktor slabljenja amplitude 

*           |  |  |     |kašnjenje početka generisanja sinusoide 

*           |  |  |frekvencija 

*           |  |amplituda sinusoide 

*           |napon jednosmernog offset-a 

 

Prvi u nizu parametara predstavlja napon jednosmernog oblika, drugi je amplituda sinusoide, 

zatim se definiše frekvencija signala, dok su preostala tri parametra manje bitna. Naime, iza 

frekvencije se definiše kašnjenje početka generisanja sinusoide, faktor slabljenja amplitude 

sinusnog signala i na kraju faza. Vrednost offseta i amplitude se moraju zadati, dok je 

podrazumevana vrednost frekvencije 1/tstop, gde je tstop trajanje simulacije zadato u 

komandi .TRAN. Poslednja tri parametra imaju podrazumevanu vrednost 0 i navoĎenje 

njihove vrednosti nije neophodno ukoliko zadržavaju podrazumevanu vrednost. 

Vreme štampanja vrednosti definišemo proizvoljno. Ulazni napon je izmeĎu čvorova (1,0), 

dok je izlazni napon izmeĎu čvorova (2,0), što je definisano za štampu u .PROBE. Ostali 

delovi kola su razmatrani, tako da se može napisati konačna netlista: 

a) 

Zadatak 12a. Naponski regulator 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-16 

V1 1 0 SIN(0 10 100) 

R1 1 2 1K 

D1 2 3 D1N4148 

D2 3 4 D1N4148 

D3 4 5 D1N4148 

D4 5 0 D1N4148 

 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

.TRAN 50u 20m UIC 

.PROBE V(1,0) V(2,0) 

.END 

 

b) 
 

Zadatak 12b. Naponski regulator 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-16 

V1 1 0 SIN(0 10 100) 

R1 1 2 1K 

X1 2 3 DIODA 

X2 3 4 DIODA 

X3 4 5 DIODA  

X4 5 0 DIODA 

 

.SUBCKT DIODA ANODA KATODA PARAMS: I0=0.1P, VT=26M, R0=16, C0=2P 

RD ANODA 10 {R0} 

CD 10 KATODA {C0} 

G1 10 KATODA VALUE={I0*(EXP(V(10,KATODA)/VT)-1)} 

.ENDS 

 

.TRAN 50u 20m UIC 

.PROBE V(1,0) V(2,0) 

.END 

 

Rezultati simulacije prikazani su na slici 2.7, dobijaju se isti oblici signala u oba slučaja. Ovo 

kolo „odseca“ sve napone veće od 4·napon voĎenja diode, jer su u pozitivnoj poluperiodi sve 

diode direktno polarisane. U toku negativne poluperiode ne vodi ni jedna dioda, inverzno su 

polarisane, tako da se napon sa ulaza preslikava na izlazu. Pomoću kursora mogu se precizno 

očitati amplitude izlaznog napona. Za diode opisane korišćenjem modelske kartice, dobija se 

amplituda izlaznog napona 3.0233 V, što znači da je napon voĎenja svake diode (identične 

su) 3.0233/4=0.756 V. Za diode opisane potkolom, amplituda izlaznog napona klipera je 

3.0386 V, što znači da je napon voĎenja svake od dioda 0.759 V. Možemo zaključiti da su 

oba modela diode u saglasnosti. 

 

Slika 2.7. Rezultati simulacije naponskog regulatora - klipera 

ZADATAK 13. Simulirati strujno-naponsku karakteristiku diode: 

.MODEL D1N4148 D (IS=0.1PA RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

ako se napon na diodi menja od 0 V do 1.2 V sa korakom 0.01 V. Analizirati uticaj 

parametra RS i M na karakteristike diode. 

Rešenje: 

SPICE netlistom opisaćemo kolo sa slike 2.8. 

           T i m e

0 s 2 m s 4 m s 6 m s 8 m s 1 0 m s 1 2 m s 1 4 m s 1 6 m s 1 8 m s 2 0 m s

V ( 1 , 0 ) V ( 2 , 0 )

- 1 0 V

- 5 V

0 V

5 V

1 0 V

Slika 2.8. Kolo za simulaciju IV karakteristike diode 

IV karakteristika diode 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

V1 1 0 1 

D1 1 0 D1N4148 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.DC V1 0 1.2 0.01 

.PROBE I(V1) 

.END 

 

Parametar RS predstavlja ekvivalentnu rednu otpornost diode, koja uglavnom potiče od 

kontakata i izvoda. U datom modelu diode iznosi 16 Ω. Za ovu vrednost RS se, na primer, 

simulacijom dobija da je pri naponu na diodi 1 V, struja koja protiče kroz diodu 20.4 mA. 

Ukoliko dioda ima loš kontakt, ekvivalentna otpornost diode biće velika. Na primer, 

uzećemo vrednost RS=1600 Ω. Pokretanjem simulacije, očitavamo da je pri naponu na diodi 

1 V, struja kroz diodu 273.7 µA. Zaključujemo da ukoliko dioda ima veliku otpornost, usled 

na primer lošeg kontakta unutar samog kućišta, struja će biti znatno manja.  

Pošto se za analizu rezultata koristi grafičko okruženje (posmatra se karakteristika), ova 

razlika u struji nije očigledna jer karakteristika ima isti oblik za obe vrednosti otpornosti. 

Zato je moguće da svaku od osa definiše korisnik: duplim klikom na brojne vrednosti na x ili 

y osi otvara se prozor kao na slici 2.9. U ovom slučaju podesićemo da se podaci na y osi 

prikazuju u opsegu od 0 do 30 mA, tako da će za različite vrednosti otpornosti promena 

struje biti očigledna. U odeljku Data Range selekujemo User Defined i unesemo opseg 

vrednosti. Ovde je moguće podesiti skalu- razmeru ose (linearna, logaritamska). 

Napomenimo, kolo sa slike 2.8 nije realno, već pomaže da se na najjednostavniji način 

simulira I-V karakteristika. U realnom kolu neophodan je otpornik za ograničenje struje kroz 

diodu, tako da na x-osi ne treba  prikazati napon na generatoru V1, već napon na diodi. 

Duplim klikom na brojne vrednosti na x-osi otvara se prozor sa slike 2.9, sa aktivnom 

karticom za podešavanje x-ose. U donjem levom uglu ovog prozora, kliknuti na dugme Axis 

Variable i definisati veličinu koja će biti prikazana, na primer V(1,0). 

Slika 2.9. Izgled prozora za definisanje osa u grafičkom okruženju 

Idealna dioda ima rednu otpornost RS = 0, što dovodi do ogromnog povećnja struje, s 

obzirom na eksponencijalnu zavisnost struje diode od napona. Default-na vrednost ovog 

parametra diodnog modela je 0. Ukoliko se parametar ne navede, uzima se podrazumevana 

vrednost, što bi u ovom slučaju dovelo do nerealnih karakteristika diode. Slično, 

podrazumevana vrednost dužine kanala MOS tranzistora u SPICE-u je 1m, tako da bi se i u 

slučaju ne navoĎenja parametra L dobile nerealne karakteristike tranzistora, što će biti 

razmatrano u nastavku kursa.  

Na slici 2.10 ilustrovana je zavisnost kapacitivnosti pn spoja u funkciji inverznog napona na 

diodi. Parametar M odreĎuje nagib, tj. funkciju navedene zavisnosti. Podrazmevana vrednost 

je M=0.5 i zadovoljavajuća je za većinu dioda. 

 

Slika 2.10. Ilustracija C-V zavisnosti pn spoja (uz objašnjenje značenja parametra M) 

2.4. Temperaturna zavisnost karakteristika diode  

Najznačajniji parametar u diodnom modelu kojim se definiše temperaturna zavisnost 

karakteristika diode je XTI. XTI parametar (default: 3) figuriše u jednačinama za inverznu 

struju zasićenja diode - definiše zavisnost IS od temperature.  

ZADATAK 14. Simulirati IV karakteristike diode iz zadatka 13 na temperaturama (0, 25, 

75, 100, 125)°C za podrazumevanu vrednost parametra XTI=3. 

Rešenje: 

Zadatak se rešava korišćenjem komande .TEMP, kojom se definiše temperatura radne 

okoline u °C. Ako se zada više od jedne temperature, sve specificirane analize (.DC, .AC, 

.TRAN) obaviće se za svaku od zadatih vrednosti temperature. Ako se ova komanda ne zada, 

analiza se radi za sobnu temperaturu, tj. za 27°C. 

Primer komande .TEMP – detaljno u [PSpcRef] 

.TEMP 0 25 75 100 125 

*     | lista temperatura 

 

Za kolo sa slike 35 može se napisati netlista sa definisanom komandom .TEMP: 

Temperaturna zavisnost IV k-ke 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

V1 1 0 1 

D1 1 0 D1N4148 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.TEMP 0 25 75 100 125 

.DC V1 0 1.2 0.01 

.PROBE I(V1) 

.END 

 

Nakon završene simulacije, a pre prikazivanja rezltata otvara se prozor u kome je moguće 

izabrati koja od karakteristika (na kojoj temperaturi) će biti prikazana. Izabraćemo svih 5 

karakteristika i sve će biti nacrtane na istom grafiku (slika 2.11). Desnim klikom na 

karakteristiku otvara se poruka sa informacijama o selektovanoj krivoj.  

Slika 2.11. I-V karakteristika diode na različitim temperaturama 
           V 1

0 V 0 . 1 V 0 . 2 V 0 . 3 V 0 . 4 V 0 . 5 V 0 . 6 V 0 . 7 V 0 . 8 V 0 . 9 V 1 . 0 V 1 . 1 V 1 . 2 V

- I ( V 1 )

0 A

2 0 m A

4 0 m A

6 0 m A

Analizirajući karakteristike zaključujemo da sa povećanjem temperature dolazi do pomeranja 

karakteristike u levo, tj. da se smanjuje napon voĎenja diode. 

ZADATAK 15. Simulirati diodu iz zadatka 13 kao senzor temperature. Odrediti prenosnu 

karakteristiku senzora (zavisnost napona na diodi u funkciji temperature) za kombinacije 

parametara: XTI=3 (default), XTI=50, EG=1.11 (default) i EG=0.66. Parametar EG definiše 

širinu zabranjene zone i podrazumevana vrednost je 1.11 eV (silicijum). Širina zabranjene 

zone germanijum iznosi 0.66 eV. 

Rešenje: 

Dioda se koristi kao temperaturni senzor kada se napaja konstantnom strujom. Za konstantnu 

struju sa promenom temperature doći će do promene napona na diodi. U kolu sa slike 2.12 

dioda radi kao temperaturni senzor, a SPICE netlista za ovo kolo data je u nastavku. 

 

Slika 2.12. Dioda kao temperaturni senzor 

Temperaturna zavisnost IV k-ke diode 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

I1 0 1 5M 

D1 1 0 D1N4148 

.MODEL D1N4148 D (IS=0.1PA RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA 

XTI=3     + EG=1.11) 

.TEMP 0 25 75 100 125 

.DC I1 5M 5.00001M 0.00001M 

.PROBE V(1) 

.PRINT DC V(1,0) 

.END 

 

Konstantnu struju koja treba da teče kroz diodu generiše nezavisni strujni generator I1 

definisan izmeĎu tačaka 0 i 1, jer je smer struje od + ka – čvoru, kao što je ranije objašnjeno. 

Inicijalna vrednost generatora I1 je 5 mA. Standardno je definisana dioda D1 i uključena 

modelska kartica diode. Komandom .TEMP definisana je lista temperatura na kojima će biti 

simuliran napon na diodi. Na ovaj način obuhvaćen je temperaturni opseg od (0-125)°C u 5 

tačaka. Temperaturna analiza se u SPICE-u ne može izvršavati kao samostalna, već se 

temperatura definisana u .TEMP koristi kao parametar u jednačinama pri .DC, .AC, ili 

.TRAN analizi, tako da se odreĎena analiza ponovi za skup temperaturnih vrednosti. Iz tog 

razloga je pored skupa temperatura potrebno definisati .DC analizu kako bi se simulirao 

napon na diodi za različite temperature ali pri konstantnoj struji. Ako je vrednost struje koja 

se iz generatora I1 usmerava u diodu 5 mA, u .DC analizi je potrebno praktično ne menjati 

ovu vrednost. Iz tog razloga se definiše da se I1 menja od 5 mA do 5.00001 mA, sa korakom 

0.00001 mA. Na ovaj način je omogućena simulacija, a praktično je struja generatora 

konstantna.  

Komandom .PROBE definisano je grafičko štampanje vrednosti napona, a da bi odreĎeni 

rezultati simulacije bili dostupni i u izlaznom fajlu, potrebno je koristiti komandu .PRINT. 

Izlazni fajl ima ekstenziju *.out. Komanda .PRINT štampa vrednosti specificiranih 

promenljivih sa zadatim korakom štampanja, kod analize u vremenskom domenu to je 

vrednost koraka štampanja, kod AC analize štampa se rezultat za svaku frekvenciju, dok pri 

parametarskoj analizi, kakva je temperaturna, štampa se rezultat za svaku novu vrednost 

parametra. U ovom primeru, .PRINT DC V(1,0) štampaju se vrednosti napona izmeĎu 

tačaka 1 i 0 (napon na diodi) dobijeni definisanom DC analizom za svaku vrednost 

temperature navedenu u komandi .TEMP. U ulaznom fajlu može se naći više od jedne 

.PRINT komande, pri čemu će se štampati sve zahtevane promenljive iz svih komandi. Kod 

.PRINT komande obavezno se navodi samo jedan tip analize na koju se komanda odnosi. 

Moguće je štampati veliki broj izlaznih veličina, pored napona i struja čak i pojedine 

parametre kod AC analize gde su rezultat kompleksni brojevi, što je detaljno objašnjeno u 

[PSpcRef]. 

Na slici 2.13 prikazani su obraĎeni rezultati simulacije, zavisnost napona na diodi u funkciji 

temperature za različite vrednosti parametara XTI i EG. Parametar EG definiše materijal, 

tako da je dioda sa EG=1.11 silicijumska, a dioda sa EG=0.66 je germanijumska. Možemo 

zaključiti da je napon na germanijumskoj diodi veći nego na silicijumskoj za iste vrednosti 

parametra XTI, kao i da su osetljivije diode sa većom vrednošću parametra XTI. 

 

Slika 2.13. Rezultati simulacije diode kao senzora temperature (prenosna karakteristika 

senzora) 

0 10 20 30 40 50 60 70 80 90 100 110 120

0,0

0,1

0,2

0,3

0,4

0,5

0,6

0,7

0,8

0,9

V
 [
V

]

T[
o
C]

 XTI=3, EG=1.11

 XTI=50, EG=1.11

 XTI=3, EG=0.66

 XTI=50, EG=0.66



 
40  

2.5. Frekventna karakteristika diode  

Izborom otpornika u kolu ograničava se struja kroz diodu pri direktnoj polarizaciji i postavlja 

se radna tačka diode Q (slika 2.14). Radna tačka odreĎenja je naponom VDQ, pri kome je 

struja IQ. Kada se na jednomserni napon superponira naizmenični mali signal vin, dolazi do 

odreĎene promene struje. Da bi snimili frekventnu karakteristiku diode potrebno je menjati 

frekvenciju ulaznog napona i snimiti struju. Dioda će do odreĎene frekvencije imati struju 

koja prati promene napona, dok od granične frekvencije struja neće pratiti promene napona. 

Granična frekvencija diode je frekvencija na kojoj vrednost struje opadne za -3 dB. 

Napomenimo, pod malim signalima podrazumevaju se naizmenični signali čije su amplitude 

mnogo manje od vrednosti jednosmernih signala u kolu.  

Simulacija frekventnih karateristika u SPICE-u podrazumeva definisanje generatora malih 

signala (posebnog ili u okviru DC generatora)  i zadavanje frekventne analize komandom 

.AC, kao što je ranije ilustrovano. Na ovaj način nije potrebno „ručno“ menjati frekvenciju 

ulaznog signala, već se definišu amplituda i faza ulaznog signala, a kroz .AC komandu 

definiše se u kom opsegu i sa kojim pravilom (po oktavi, dekadi, linearno) će se menjati 

frekvencije malog signala.  

 

Slika 2.14. Mali signali u okolini radne tačke diode 

ZADATAK 16. Simulirati diodu iz zadatka 13 u frekventnom domenu. Radna tačka diode 

postavljena je tako da je napon na diodi 0.6 V. Parametri malog signala za .AC analizu su: 

amplituda 50 mV, faza 0°. Simulaciju uraditi u frekventnom opsegu od 1 kHz do 1 GHz, u 

10 tačaka po dekadi. Prikazati zavisnost struje kroz diodu izraženu u dB u funkciji 

frekvencije. Analizirati uticaj parametara TT, CJ0 i RS na frekventni odziv ove diode. 

Rešenje: 

SPICE netlista za kolo sa slike 2.8 kada se generator V1 definiše kao DC generator sa 

superponiranim AC malim signalom je: 

Frekventna karakteristika diode 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

V1 1 0 DC 0.6 AC 50M 

D1 1 0 D1N4148 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA 

XTI=50 +EG=0.66) 

.AC DEC 10 1K 1G 

.PROBE I(V1) 

.END 

 

V1 generiše jednosmerni napon 0.6V na koji je superponiran naizmenični signal amplitude 

50 mV. Alternativno rešenje je korišćenje jednog DC i drugog AC generatora, ali tada 

postoji čvor više u kolu, a naponi se takoĎe superponiraju. AC analiza vršiće se u 10 tačaka 

po jednoj dekadi u frekventnom opsegu od 1 kHz do 1 GHz. Nakon završene simulacije 

treba prikazati struju kroz diodu izraženu u dB: Trace/Add Traces. Iz odeljka Functions or 

Macros (desna stran prozora) izabrati fukciju DB(). Funkcija se dodaje u Trace expression 

(donji levi ugao). Kursor je aktivan izmeĎu zagrada gde treba ubaciti vrednost struje I(V1). 

Radi prikazivanja realnog smera struje dodati -, tako da konačni izraz koji će biti iscrtan 

bude: -DB(I(V1)). Rezultat simulacije prikazan je na slici 2.15. Korišćenjem kursora 

možemo očitati graničnu frekvenciju ove diode, iznad koje struja neće pratiti promene 

napona. Granična frekvencija ove diode očitava se kad struja, izražena u dB, opadne na 60-

3=57 dB i iznosi 14.9 MHz. 

 

Slika 2.15. Rezultat simulacije – frekventni odziv diode 

Promena, tj. smanjenje prametra modela TT sa 12 ns na 1.2 ns ima veliki uticaj na 

frekventnu karakteristiku. Za graničnu frekvenciju dobija se 137.5 MHz. TT definiše vreme 

potrebno za „prelet“ nosilaca od anode do katode, i što je ovo vreme kraće to će dioda brže 

raditi. Da bi se ovo vreme smanjilo, dimenzije diode (pn spoja) treba smanjiti. Parametri CJ0 

i RS utiču u manjoj meri na frekventni odziv diode: CJ0=0.2 pF, fg=15 MHz; RS=0.16Ω, 

fg=13.1 MHz. 

           F r e q u e n c y

1 . 0 K H z 3 . 0 K H z 1 0 K H z 3 0 K H z 1 0 0 K H z 3 0 0 K H z 1 . 0 M H z 3 . 0 M H z 1 0 M H z 3 0 M H z 1 0 0 M H z 3 0 0 M H z 1 . 0 G H z

- D B ( I ( V 1 ) )

5 0

5 5

6 0

6 5

2.6. Prekidačke karakteristike diode  

ZADATAK 17. Simulirati odziv diode na impulsnu pobudu (kolo sa slike 2.8). Parametri 

impulsnog signala su: napon logičke nule 0 V, napon logičke jedinice 1 V, signal se generiše 

odmah nakon pokretanja simulacije, vreme porasta prednje ivice je 10 ns, vreme opadanja 

zadnje ivice je 10ns, impuls traje 1 μs, dok je ukupna perioda signala 5 μs. Uraditi 

vremensku analizu u trajanju od 15 μs. U posebnim prozorima prikazati napon na diodi i 

struju kroz diodu. Analizirati uticaj parametra modela CJ0 na prekidačke karakteristike 

diode. 

Rešenje: 

SPICE netlista za kolo sa slike 2.8 kada se generator V1 definiše kao impulsni je: 

Odziv diode na impulsnu pobudu 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

V1 1 0 PULSE (0 1 0 10N 10N 1U 5U) 

D1 1 0 D1N4148 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.TRAN 5N 15U  

.PROBE I(V1) V(1) 

.END 

 

Definisanje impulsnog generatora i zadavanje vremenske analize objašnjeno je ranije. Nakon 

završene simulacije, treba prikazati napon i struju u različitim prozorima: Plot/Add Plot to 

Window. Otvaraju se dva prazna prozora, klikom u odreĎeni prozor – on se aktivira i svi 

dalji koraci se odnose na njega. Aktivan prozor označen je sa SEL>> u levom uglu. U gornji 

prozor prikazati napon: Trace/Add Traces, izabrati V(1). Selektovati donji prozor klikom na 

njega i prikazati struju kroz diodu I(V1). Rezultati simulacije prikazani su na slici 2.16, 

uočiti vreme oporavka diode, ili vreme isključenja diode. Koristiti alate za uvećanje (Zoom). 

Pomoću kursora se može odrediti ovo vreme: izabrati kursor, postaviti levim tasterom miša 

prvi kursor na mesto gde se isključuje napon na diodi, a zatim desnim klikom miša drugi 

kursor dovesti do mesta kada struja kroz diodu postane 0. Očitana vrednost vremena 

oporavka diode (razlika izmeĎu kursora) je 19 ns (tipične vrednosti su od nekoliko ns do 

nekoliko desetina ns za Si prekidačke diode). Ovo kašnjenje u odzivu diode javlja se kao 

posledica „punih“ kapacitivnosti diode. Ako je kapacitivnost diode veća, biće potrebno više 

vremena da se isprazni, pa će i vreme oporavka diode biti duže. Ovo se može ilustrovati 

promenom vrednosti parametra CJ0. Napomenimo, definiše se i vreme uključenja diode ali 

je ono daleko manje od vremena isključenja i u praksi se smatra zanemarljivim. 

Slika 2.16. Rezultati simulacije – odziv diode na impulsnu pobudu 

ZADATAK 18. Simulirati u vremenskom domenu ispravljačko kolo sa Grecovim spojem 

prikazano na slici 2.17. Generator V1 predstavlja mrežni napon: sinusoida amplitude 325 V 

(Vrms=230 V), frekvencije 50 Hz. Vrednost induktivnosti primara transformatora je 400 μH, 

dok je induktivnost sekundara 1 μH, a koeficijent sprege induktivnosti iznosi 1 [videti 

PSpcRef]. Grecov spoj opisati kao potkolo, pri čemu se koristi dioda:  

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

U kolu pozvati potkolo Grecovog spoja. Vrednost kapacitivnosti filtarskog kondenzatora je 

1 mF i kolo je priključeno na opterećenje otpornosti 1 kΩ. U jednom prozoru prikazati ulazni 

– mrežni napon, a u drugom prozoru prikazati napon na sekundaru transformatora i 

potrošaču. Kolika je vrednost ispravljenog DC napona? 

 

Slika 2.17. Ispravljačko kolo sa Grecovim spojem 

Rešenje: 

Za kolo sa slike 2.17. može se napisati sledeća netlista: 

Ispravljacko kolo sa Grecovim spojem 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

V1 1 0 SIN(0 325 50) 

R1 1 2 0.1 

L1 2 0 400U 

           T i m e

0 s 1 u s 2 u s 3 u s 4 u s 5 u s 6 u s 7 u s 8 u s 9 u s 1 0 u s 1 1 u s 1 2 u s 1 3 u s 1 4 u s 1 5 u s

-  I ( V 1 )

- 4 0 m A

0 A

4 0 m A

S E L > >

V ( 1 )

0 V

0 . 5 V

1 . 0 V

L2 3 0 1U 

K1 L1 L2 1 

X1 4 5 3 0 GREC 

R2 4 5 1K 

C1 4 5 1M 

 

.SUBCKT GREC PLUS MINUS VDD VSS 

D1 VDD PLUS D1N4148 

D2 MINUS VDD D1N4148 

D3 VSS PLUS D1N4148 

D4 MINUS VSS D1N4148 

.ENDS  

 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

.TRAN 50N 1 UIC 

.PROBE V(4,5) V(3,0) V(1,0) 

.END 

 

Generator V1 definisan je kao mrežni, korišćenjem SIN sa parametrima: jednosmerni offset 

jednak 0, aplitude 325 V i frekvencije 50 Hz. Napomenimo, napon na sekundaru 

transformatora odreĎen je brojem navojaka u primaru i sekundaru, kao i naponom na 

primaru: N1/N2=U1/U2. Induktivnost primara i sekundara srazmerna je sa kvadratom broja 

navojaka: L≡N
2
. Korišćenjem ovih relacija odreĎene su induktivnosti tako da je napon 

sekundara ~12 V: U2=(N2 /N1) ·U1=12 V → N1/N2≈20, pa ako je L2= 1 μH, tada će L1 biti 

400 μH. Koeficijent sprege definiše se korišćenjem: K1 L1 L2 1 što podrazumeva da je 

koeficijent sprege K1 izmeĎu induktivnosti L1 i L2 jednak 1. Vrednost ovog koeficijenta 

kreće se od – 1 do 1 i odreĎuje meĎusobnu induktivnost M12: k=M12/(L1·L2)
1/2

. Za 

transformatore se najčešće uzima vrednost k=1, tada su kalemovi su u idealnoj sprezi jer 

nema gubitaka magnetnog fluksa (fluks jednog kalema je kompletno obuhvaćen zavojcima 

drugog kalema). Kada nema idealne sprege, vrednost k je manja od 1.  

Definisano je potkolo Grecovog spoja, poziv potkola, kao i filtarski kondenzator i potrošač. 

Potkolo se sastoji iz četiri identične diode. Modelska kartica definisana je van potkola. 

Modeli mogu biti definisani u okviru potkola, ali su tada dostupni – vidljivi samo za 

elemente potkola, dok ako se ista modelska kartica koristi za opis elementa u glavnom kolu 

ili drugom potkolu ona neće biti dostupna. Dobra praksa je da se sve modelske kartice 

definišu u glavnom kolu, jer su one tada globalne, tj. dostupne i za sve elemente u kolu i u 

potkolima. 

Vreme simulacije postavljeno je na 1 s. Nakon simulacije, dodat je novi prozor i u njemu 

prikazan napon na primaru, dok su u drugom prozoru prikazani napon na sekundaru i 

potrošaču (slika 2.18). Pomoću kursora očitana je vrednost jednosmernog ispravljenog 

napona, ~10 V. 

Slika 2.18. Rezultati simulacije, gore  - mrežni napon, dole – napon na sekundaru 

transformatora i napon na potrošaču 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

           T i m e

0 s 0 . 1 s 0 . 2 s 0 . 3 s 0 . 4 s 0 . 5 s 0 . 6 s 0 . 7 s 0 . 8 s 0 . 9 s 1 . 0 s

V ( 3 , 0 ) V ( 4 , 5 )

- 2 0 V

0 V

2 0 V
V ( 1 , 0 )

- 4 0 0 V

0 V

4 0 0 V

S E L > >

3. 
MODELIRANJE I SIMULACIJA 
BIPOLARNOG TRANZISTORA 

3.1. Idealni model bipolarnog tranzistora 

Bipolarni tranzistor kao diskretna komponenta ima tri izvoda: emitor, bazu i kolektor (slika 

3.1 - levo). Kada se bipolarni tranzistor proizvodi u okviru integrisanog kola, tada se definiše 

i četvrti izvod – supstrat. U tom slučaju postoji više tranzistora, tako da se kolektor difunduje 

u supstrat (slika 3.1 - desno) i neophodno je definisati i taj izvod zbog postojanja još jednog 

pn spoja (izmeĎu kolektora i supstrata).  

 

Slika 3.1. Ilustracija tehnološke realizacije npn tranzistora kao diskretne komponente i u 

integrisanim kolima [preuzeto iz Z.Prijić, A. Prijić, Uvod u poluprovodničke komponente i 

njihovu primenu] 

Sintaksa za opis bipolarnog tranzistora podrazumeva korišćenje rezervisanog slova Q iza 

koga se navodi naziv tranzistora, zatim redom čvorovi kolektora, baze, emitora. Opciono se 

nakon toga navodi čvor supstrata. Nakon navedenih čvorova unosi se naziv modelske kartice 

i na kraju vrednost parametra area. Parametar area predstavlja relativnu površinu tranzistora 

na čipu i figuriše u jednačinama koje odreĎuju vrednosti elemenata ekvivalentnog kola 

bipolarnog tranzistora. Podrazumevana vrednost ovog parametra je 1. 

Primer definisanja bipolarnog tranzistora – detaljno u [PSpcRef] 

Q1 2 1 0 0 Q2N2222A 1 

*  | | | |     |    |parametar area 

*  | | | |     | naziv modelske kartice 

*  | | | | supstrat 

*  | | | emitor 

*  | | baza 

*  | kolektor 

ZADATAK 19. Na slici 3.2 je dato prosto ekvivalentno kolo bipolarnog tranzistora. 

Simulirati Gamelovu karakteristiku (Gummel Plot), zavisnost pojačanja tranzistora od 

kolektorske struje kao i izlazne karakteristike. 

Rešenje: 

 

Slika 3.2. Uprošćen (idealni) model (ekvivalentno kolo) bipolarnog tranzistora 

Analiziraćemo kolo sa slike 3.2. Otpornost RB potiče od kontakata i izvoda, predstavlja 

baznu, ulaznu otpornost. Bazno – emitorski pn spoj predstavlja se diodom. Da bi tranzistor 

vodio neophodno je da bazno – emitorski spoj bude direktno polarisan, tako da je tipično 

VBE=0.7 V. S druge strane, definiše se izlazna otpornost bipolarnog tranzistora u 

kolektorskom kolu RC. Bipolarni tranzistor se koristi kao prekidač i pojačavač. Kada se 

koristi kao pojačavač, ulazna struja baze biva pojačana β puta, tako da je izlazna kolektorska 

struja IC=βIB. Iz tog razloga, osnovu modela bipolarnog tranzistora čini strujom kontrolisani 

strujni izvor (generator) F. Ovaj generator se definiše navoĎenjem ključnog slova F i naziva 

generatora, nakon čega se navode čvorovi (smer struje je od + ka – čvoru). Iza čvorova 

navodi se naziv kontrolišućeg generatora, tj. strujnog izvora. U ovom primeru struju baze 

daje generator VBE koji se definiše kako bi direkno polarisao bazno – emitorski spoj. 

Poslednji u nizu je parametar pojačanja, ovde ćemo uzeti 100. Na osnovu ovoga može se 

napisati PSPICE netlista za kolo sa slike 3.2, s tim što se za polarizaciju kolektorskog spoja 

tranzistora definiše idealni naponski generator VCC (na primer 3 V): 

Bipolarni tranzistor - ekvivalentno kolo 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

VBE B 0 0.6 

VCC C 0 3 

RB B 1 100 

RC C 2 100 

D1 1 0 D1N4148 

F1 2 0 VBE 100 

.MODEL D1N4148 D (IS=0.1PA RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.DC VBE 0.1 0.8 0.1 

.PROBE I(VBE) I(VCC) 

.END 

 

Za dobijanje Gummel Plot-a potrebno je uraditi DC analizu, pri čemu se menja napon VBE 

od 0.1 V do 0.8 V sa korakom 0.1 V. Gummelova karakteristika je kombinovana 

karakteristika struje kolektora i struje baze bipolarnog tranzistora u funkciji napona izmeĎu 

baze i emitora u polu logaritamskoj razmeri (log(IC)=f(VBE) i log(IB)=f(VBE)). Ova 

karakteristika je vrlo korisna u karakterizaciji komponenti jer se na osnovu nje mogu odrediri 

pojačanja tranzistora β i α, faktori idealnosti za bazu i kolektor, redne otpornosti i struje 

curenja. Na slici 3.3 prikazana je Gummelova karakteristika za simulirani idealni bipolarni 

tranzistor. Na y-osi prikazanse su struje baze (-I(VBE)) i kolektora (I(VCC)), pri čemu je u 

prozoru koji se otvara duplim klikom na brojne vrednosti na y-osi Axis Settings u odeljku 

Scale izabrana opcija Log. Ovaj prozor se otvara i klikom na: Plot/Axis Settings, gde se bira 

osa i podešavaju parametri (opseg, razmera, itd.). Sa karakteristike možemo videti da je 

odnos kolektorske i struje baze jednak 100, što predstavlja definisano pojačanje tranzistora. 

TakoĎe, može se uočiti da struje rastu do napona voĎenja diode koji je 0.6 V, a nakon toga 

ulaze u zasićenje. Zavisnost pojačanja u funkciji struje kolektora prikazaćemo novim 

pokretanjem simulacije. U prozoru Axis Settings za x- osu izabrati opciju Axis Variable i 

promeniti veličinu koja će biti prikazana na x- osi: I(VCC), u LOG razmeri. Korišćenjm Add 

Traces, na y- osi prikazati vrednost: I(VCC)/ (-I(VBE)). Rezultujuća karakteristika prikazana 

je na slici 3.4, gde se vidi da je pojačanje β za sve struje konstantno, što nije slučaj kod 

realnih tranzistora. 

 

Slika 3.3. Gummelova karakteristika za kolo sa slike 3.2 (ekvivalentno kolo bipolarnog 

tranzistora) 

 

Slika 3.4. Zavisnost pojačanja od struje kolektora za idealno kolo bipolarnog tranzistora sa 

slike 3.2 

Prikazanu netlistu je potrebno modifikovati kako bi bile simulirane izlazne strujno – 

naponske karakteristike bipolarnog tranzistora: 

           V B E

1 0 0 m V 1 5 0 m V 2 0 0 m V 2 5 0 m V 3 0 0 m V 3 5 0 m V 4 0 0 m V 4 5 0 m V 5 0 0 m V 5 5 0 m V 6 0 0 m V 6 5 0 m V 7 0 0 m V 7 5 0 m V 8 0 0 m V

- I ( V B E ) I ( V C C )

1 . 0 p A

1 0 n A

1 0 0 u A

1 . 0 A

           V B E

1 0 0 m V 1 5 0 m V 2 0 0 m V 2 5 0 m V 3 0 0 m V 3 5 0 m V 4 0 0 m V 4 5 0 m V 5 0 0 m V 5 5 0 m V 6 0 0 m V 6 5 0 m V 7 0 0 m V 7 5 0 m V 8 0 0 m V

I ( V C C ) / ( - I ( V B E ) )

0

4 0

8 0

1 1 0

Bipolarni tranzistor – izlazne karakteristike 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

VBE B 0 0.6 

VCC C 0 3 

RB B 1 100 

RC C 2 100 

D1 1 0 D1N4148 

F1 2 0 VBE 100 

.MODEL D1N4148 D (IS=0.1PA RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.DC VCC 0 5 0.25 

.STEP VBE 0.1 0.9 0.1 

.PROBE I(VCC) 

.END 

 

Naime, kao što se i prilikom snimanja familije I-V karakteristika ne menjaju istovremeno 

vrednosti generatora za polarizaciju baze i kolektora tako se i u SPICE-u ne zadaju dve DC 

analize. SPICE ne može da reši sistem jednačina koji bi se generisao u tom slučaju. U 

realnim uslovima, eksperimentalno, kada se fiksira polarizacija baze, tako da daje konstantnu 

struju baze pa se izvrši promena napona polarizacije kolektora, tj. napona izmeĎu kolektora i 

emitora dobija se jedna karakteristika. Nakon toga se polarizacija baze fiksira na drugu 

vrednost i ponovo se vrši promena VCE. Na ovaj način se dobija familija izlaznih 

karakteristika.  

Da bi se simulirale izlazne karakteristike prvo se definiše DC analiza pomoću generatora 

VCC, koji se u ovom primeru menja od 0 V do 5 V sa korakom 0.25 V. Komandom .STEP 

se izaziva ponavljanje svih navedenih analiza (komande .DC, .AC, .TRAN) za zadati skup 

vrednosti promenljivog prametra.  

Primer komande .STEP – detaljno u [PSpcRef] 

.STEP VB 0V 1V .25V 

*     |  |  |  | korak stepovanja 

*     |  |  | krajnja vrednost 

*     |  | početna vrednost 

*     | naziv generatora koji se stepuje 

 

U ovom slučaj se stepuje vrednost generatora VBE, od 0.1 V do 0.9 V sa korakom 0.1 V. 

Dakle, prvo se uzima VBE=0.1 V i za tu vrednost se uradi DC analiza pomoću VCC, nakon 

toga se VBE promeni na 0.2 V i ponovi DC analiza... Na ovaj način dobija se izlazna I-V 

karakteristika. Svi razultati biće smešteni u isti izlazni fajli, a grafički postprocesor Probe 

omogućava grafičko prikazivanje famlije krivih (slika 3.5). Radi bolje preglednosti rezultata 

y-osa, tj. vrednost struje IC data je u LOG razmeri. Možemo uočiti da za sve vrednosti 

napona izmeĎu kolektora i emitora je konstanta vrednost struje, što ne odgovara realnim 

karakteristikama tranzistora.  

Slika 3.5. Izlazne I-V karakteristike idealnog ekvivalentnog kola bipolarnog tranzistora 

Ovo kolo opisuje tranzistor kada bi on bio idealan pojačavač, bez prisutne oblasti zasićenja i 

drugih fizičkih efekata koji postoje kod realnog tranzistora. Pored postojanja oblasti 

zasićenja kod realnog tranzistora i u normalnoj aktivnoj oblasti vrednost kolektorske struje 

nije konstantna već blago raste sa povećanjem napona izmeĎu kolektora i emitora. Ovaj 

efekat nastaje usled povećanja širine osiromašene oblasti inverzno polarisanog pn spoja 

baza-kolektor, koja je posledica porasta napona VCE, odnosno smanjenja napona VBC. 

Osiromašena oblast se širi kako na stranu kolektora, tako i na stranu baze. Njeno širenje na 

stranu baze čini da se elektroneutralna oblast baze efektivno skraćuje. Zbog toga veći broj 

elektrona injektovanih iz emitora stiže do kolektora, što se ispoljava kroz blagi rast 

kolektorske struje. Efekat se naziva modulacija širine baze ili Erlijev (Early) efekat. Izlazna 

otpornost je parametar od značaja za primenu tranzistora kao pojačavača, a definiše se za 

male promene napona VCE i promene IC. Idealno, izlazna otpornost tranzistora bi u aktivnoj 

oblasti trebalo da teži beskonačnosti, jer se struja IC ne menja sa promenom VCE. Tada nema 

uticaja priključenog potrošača na pojačanje. MeĎutim, zbog Erlijevog efekta, izlazna 

otpornost ima konačnu vrednost. Erlijev napon se može opisati povlačenjem tangente na 

svaku od karakteristika u aktivnoj oblasti. Idealno, sve tangente bi sa VCE osom trebalo da se 

preseku u tački koja se naziva Erlijev napon (slika 3.6). Što je veći Erlijev napon, to je veća i 

izlazna otpornost tranzistora. Velike vrednosti Erlijevog napona postižu se jakim dopiranjem 

baze i skraćivanjem oblasti baze. 

 

Slika 3.6. Ilustracija definicije Erlijevog napona [preuzeto iz Z.Prijić, A. Prijić, Uvod u 

poluprovodničke komponente i njihovu primenu] 

           V C C

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

I ( V C C )

1 . 0 n A

1 . 0 u A

1 . 0 m A

1 . 0 A

1 0 0 A

3.2. Gummel-Poon-ov model bipolarnog tranzistora 

U PSPICE je ugraĎen modifikovani Gummel-Poon-ov model bipolarnog tranzistora. 

Uprošćena električna šema ovog modela data je na slici 3.7. Detaljna šema sa svim 

parametrima i jednačinama data je u [PSpcRef]. Ovaj model oslikava kroz parametre veliki 

broj fizičkih efekata u realnom tranzistoru. U odnosu na idealni model sa slike 3.2, ovde je 

dodata dioda koja predstavlja pn spoj baza-kolektor. Definišu se kapacitivnosti pn spojeva u 

tranzistoru, kao i dodatne otpornosti. Modelska kartica može se definisati za tri različita tipa 

tranzistora: NPN tranzistor, PNP tranzistor i lateralni PNP tranzistor.  

 

Slika 3.7. Električna šema uprošćenog Gummel-Poon-ovog modela bipolarnog tranzistora 

Ilustracije radi date su jednačine za struju kolektora i baze u Gummel-Poon-ovom modelu, 

kao i odreĎene kapacitivnosti: 

Neki od najbitnijih parametara modela bipolarnog tranzistora su: BF – maksimalna vrednost 

koeficijenta strujnog pojačanja β pri direknoj polarizaciji, čija je podrazumevana vrednost 

100; VAF – direktni Erlijev napon, čija je podrazumevana vrednost beskonačna, IKF – 

prelomna tačka za opadanje direktnog β pri velikim strujama (struja kolektora pri kojoj β 

počinje da opada – videti sliku 3.12). Slično, definišu se ovi parametri pri inverznoj 

polarizaciji: BR – pojačanje pri inverznoj polarizaciji, VAR – inverzni Erlijev napon, IKR – 

prelomna tačka za opadanje inverznog β pri velikim strujama. Pored koeficijenata emisije 

struja curenja za sve spojeve, definišu se i kapacitivnosti i otpornosti pn spojeva. Na primer, 

RE i RC su otpornosti emitora i kolektora, respektivno. S druge strane, definiše se RB – 

maksimalna otpornost baze pri nultoj polarizaciji i RBM – minimalna otpornost baze pri 

velikoj struji. Kao i kod diode, i ovde je definisano vreme preleta nosilaca (TF). Postoji i set 

parametara koji opisuju temperaturnu zavisnost karakteristika bipolarnog tranzistora. Još 

jednom napomenimo, oni parametri koji nisu navedeni u modelskoj kartici uzimaju se sa 

podrazumevanim vrednostima. Primer jedne modelske kartice komercijalnog tranzistora dat 

je u nastavku:  

.MODEL Q2N2222A NPN (IS=14.34F XTI=3  EG=1.11  VAF=74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0 

+IKR=0 RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

 

Napomenimo, znak + znači nastavak reda. U modelskoj kartici je iza ključne reči .MODEL 

definisan naziv modelske kartice i specificiran tip tranzistora. IzmeĎu malih zagrada nalaze 

se vrednosti parametara modela. Ove vrednosti daju proizvoĎači i odreĎene su iz 

eksperimentalnih karaketristika, tako da simulirana karakteristika odgovara realnoj. 

Parametar EG definiše širinu zabranjene zone materijala, iz ovog primer vidimo da je 

EG=1.11 eV što znači da je materijal od koga je izgraĎen tranzistor silicijum. Erlijev napon 

je 74.03 V, a pojačanje pri direktnoj polarizaciji 255.9. Izlazna otpornost RC je 1 Ω, dok je 

ulazna otpornost baze RB=10Ω.  

ZADATAK 20. Simulirati strujno – naponske karakteristike bipolarnog tranzistora sa datom 

modelskom karticom. Analizirati uticaj otpornosti baze RB, kolektora RC i Erlijevog napona 

VAF na karakteristike tranzistora. Vrednost VCC menjati od 0 V do 5 V sa korakom 0.05 V, 

a vrednost generatora VBE stepovati od 0.5 V do 1 V sa korakom  0.1 V. Prikazati 

Gummelovu karakteristiku i zavisnost pojačanja od struje kolektora za isti tranzistor ako se u 

DC analizi menja vrednost generatora VBE od 0.2 do 1.8 V sa korakom 0.025 V. Odrediti za 

koju vrednost napona VBE će pojačanje tranzistora biti maksimalno. 

Rešenje: 

Za kolo sa slike 3.8 može se napisati sledeća netlista: 

Bipolarni tranzistor - modelska kartica 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

VBE B 0 0.6 

VCC C 0 3 

Q1 C B 0 Q2N2222A 

.MODEL Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0 RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.DC VCC 0 5 0.25 

.STEP VBE 0.5 1 0.1 

.PROBE I(VBE) I(VCC) 

.END 

 

 

Slika 3.8. Električna šema za simulaciju karakteristika bipolarnog tranzistora 

Na slici 3.9 prikazani su rezultati simulacije izlazne strujno naponske karakteristike 

bipolarnog tranzistora pri čemu je korišćena nemodifikovana modelska kartica proizvoĎača. 

Promenom parametara RC, RB i VAF uočava se uticaja ovih otpornosti na karakteristike 

tranzistora. Slika 3.10 prikazuje izlged karakteristika kada je vrednost Erlijevog napona 

VAF=7.403. 

Slika 3.9. Izlazna I-V karakteristika bipolarnog tranzistora (nemodifikovana modelska 

kartica) 

 

Slika 3.10. Izlazna I-V karakteristika bipolarnog tranzistora (VAF=7.403) 

Modifikovati PSPICE netlistu, umesto DC analiza pomoću generatora VCC i stepovanja 

VBE generatora, potrebno je uraditi samo DC analizu pomoću generatora VBE:  

.DC VBE 0.2 1.8 0.025 

Na grafiku prikazati struju baze (kroz generator VBE) sa znakom minus, kao i struju 

kolektora (kroz generator VCC) sa znakom minus. Diplim klikom na y- osu podesiti razmeru 

ove ose na LOG. Dobijena Gummelova karakteristika prikazana je na slici 3.11. Deo 

Gummelove karakteristike u kome stuje postaju konstantne (dolazi do zasićenja) oslikavaju 

veliku vrednost impedanse tranzistora, što je neophodno kada tranzistor radi kao pojačavač 

tako da priključeno opterećenje ne utiče na karakteristike tranzistora, na primer pojačanje. 

           V C C

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

-  I ( V C C )

0 A

0 . 5 A

1 . 0 A

           V C C

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

-  I ( V C C )

0 A

0 . 5 A

1 . 0 A

Slika 3.11. Gummelova karakteristika simuliranog bipolarnog tranzistora 

Za dobijanje karakteristike zavisnosti pojačanja β od struje kolektora potrebno je izbrisati 

prikazane struje. Duplim klikom na brojne vrednosti na x- osi (Plot/Axis Settings) izabrati 

opciju Axis Variable. Podesiti da na x-osi bude I(VCC) i izabrati LOG rameru. Korišćenjem 

opcije Add Trace na y-osi prikazati vrednost β=I(VCC)/I(VBE). Dobijena karakteristika 

prikazana je na slici 3.12. Možemo zaključiti da strujno pojačanje nije konstantno za sve 

vrednosti struje kolektora pri zadatom naponu VCE, već se menja kao što prikazuje slika 

3.12. Sa porastom struje kolektora strujno pojačanje raste, zatim je približno konstantno, a 

nakon toga se smanjuje. Ovaj trend je karakterističan za sve bipolarne tranzistore. Smanjenje 

pojačanja usled jakih struja kolektora  posledica je razičitih zavisnosti struja baze i kolektora 

od napona direktne polarizacije baza – emitor. Ova karakteristika sastavni je deo tehničkih 

specifikacija proizvoĎača tranzistora. Promenom x-ose tako da bude prikazana vrednost 

napona VBE u LIN razmeri možemo odrediti vrednost tog napona za koje tranzistor ima 

maksimalno pojačanje. Slika 3.13 prikazuje ovu zavisnost i jasno se vidi da je maksmalno 

pojačanje pri naponu VBE=0.725 V. 

 

Slika 3.12. Zavisnost strjnog pojačanja β od struje kolektora za simulirani tranzistor 

           V B E

0 . 2 V 0 . 3 V 0 . 4 V 0 . 5 V 0 . 6 V 0 . 7 V 0 . 8 V 0 . 9 V 1 . 0 V 1 . 1 V 1 . 2 V 1 . 3 V 1 . 4 V 1 . 5 V 1 . 6 V 1 . 7 V 1 . 8 V

-  I ( V C C ) - I ( V B E )

1 . 0 p A

1 0 n A

1 0 0 u A

1 . 0 A

1 0 K A

           I ( V C C )

- 1 0 p A - 1 0 A

I ( V C C ) /  I ( V B E )

1 . 0

1 0

1 0 0

1 . 0 K

Slika 3.13. Zavisnost strjnog pojačanja β od napon izmeĎu baze i kolektora za simulirani 

tranzistor 

ZADATAK 21.  

a) Analizirati ponašanje kola sa slike 3.14 u vremenskom domenu kada se kolo pobudi 

sinusnim generatorom amplitude 0.1 V i rekvencije 10 kHz. Analizu završiti posle 

0.2 ms, NE koristiti UIC. Na jednom grafiku predstaviti ulazni i izlazni napon. 

Zadati da se izvrši Fourier-ova analiza talasnog oblika napona na potrošaču za 

frekvenciju osnovnog harmonika jednaku frekvenciji pobudnog generatora. Priložiti 

rezultate analize iz izlaznog fajla. Vrednosti elemenata u kolu date su na slici 3.14. 

Koristiti modelsku karticu iz zadatka 20. 

b) Posmatrati kolo pojačavača sa zajedničkim emitorom u jednosmernom režimu 

(šema izmeĎu C1 i C2). Simulirati zavisnost jednosmernog napona na kolektoru 

tranzistora od temperature u opsegu od 0°C do 120°C sa korakom 5°C. Napomena: 

Za temperaturnu analizu koristiti komandu .DC LIN TEMP 0 120 5, jer je 

TEMP validan parametar za DC analizu u PSPICE-u. 

 

Slika 3.14. Pojačavač sa zajedničkim emitorom 

           V B E

0 . 2 V 0 . 3 V 0 . 4 V 0 . 5 V 0 . 6 V 0 . 7 V 0 . 8 V 0 . 9 V 1 . 0 V 1 . 1 V 1 . 2 V 1 . 3 V 1 . 4 V 1 . 5 V 1 . 6 V 1 . 7 V 1 . 8 V

I ( V C C ) /  I ( V B E )

1 . 0

1 0

1 0 0

1 . 0 K

Rešenje: 

a) Za kolo sa slike 3.14 može se napisati sledeća netlista: 

Bipolarni tranzistor - DOMACI ZADATAK 

.OPTIONS RELTOL=0.01 STEPGMIN GMIN=1E-12 

VCC 1 0 20 

V1 2 0 SIN(0 0.1 10K) 

R1 1 3 100K 

R2 3 0 20K 

RC 1 4 3K 

RE1 5 6 500 

RE2 6 0 1K 

RL 7 0 2K 

CE 6 0 10U 

C1 2 3 10U 

C2 4 7 10U 

Q1 4 3 5 Q2N2222A  

.MODEL Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0  RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.TRAN 1P 0.2M  

.FOUR 10K V(7) 

.PROBE V(2,0) V(7,0)  

.END 

 

Bipolarni traznistor polarisan je preko naponskog razdelnika u radnu tačku Q. Uloga 

kondenzatora CE je da prespaja (bypass) naizmenični signal na masu. Reaktanse svih 

kondenzatora u kolu treba da budu takve da oni za opseg učestanosti koji se pojačava 

predstavljaju kratak spoj. Kolom sa slike 3.14 postiže se kompromis izmeĎu temperaturne 

stabilnosti i vrednosti naponskog pojačanja. U ovom kolu je otpornik u emitoru podeljen, pri 

čemu je RE2 za male signale prespojen na masu preko kondenzatora. Oba otpornika utiču na 

položaj radne tačke, ali na vrednost strujnog pojačanja utiče samo RE1. Detaljnije o 

pojačavačima sa bipolarnim tranzistorima u [Z.Prijić, A. Prijić, Uvod u poluprovodničke 

komponente i njihovu primenu].  

Komandom .FOUR zahteva se Fourier-ova analiza talasnog oblika nekih promenljivih i 

izračunavanje koeficijenata harmonijskih izobličenja. Za svaku zahtevanu promenljivu 

simulator će izračunati i odštampati u izlaznom fajlu vrednost jednosmerne komponente, 

amplitudu osnovnog harmonika i aplitude od drugog do devetog harmonika. TakoĎe, 

simulator računa i štampa klir faktore harmonika i ukupni klir fakor. Iza ključne reči .FOUR 

navodi se vrednost frekvencije osnovnog harmonika i niz promeljivih (na primer, napon u 

odreĎenoj tački). Za Fourier-ovu analizu neophodno je da se raspolaže rezltatima analize u 

vremenskom domenu, zato u ulaznom fajlu mora da postoji i .TRAN. Fourier-ova analiza 

obavlja se posle završetka analize u vremenskom domenu koristeći talasne oblike navedenih 

promenljivih. Za Fourier-ovu analizu koristi se samo vremenski interval od kraja analize pa 

unazad za 1/fosnovno i podrazumeva se da se ovaj period beskonačno ponavlja. Da bi Fourier-

ova analiza bila validna potrebno je da ukupno trajanje analize ne bude kraće od ovog 

perioda i da to bude perioda ponavljanja signala u kolu. 

Primer komande .FOUR – detaljno u [PSpcRef] 

.FOUR 10K V(7) 

*          |   |  promenljiva koja se analizira 

*          | vrednost frekvencije osnovnog harmonika 

 

Rezulati simulacije prikazani su na slici 3.15, možemo videti da je izlazni signal suprotan po 

fazi u odnosu na ulazni signal i pojačan 2.3 puta. U nastavku su dati rezultati Fourier-ove 

analize sačuvani u izlaznom fajlu *.out: 

FOURIER COMPONENTS OF TRANSIENT RESPONSE V(7) 

 DC COMPONENT =   5.728092E-04 

 

 HARMONIC   FREQUENCY    FOURIER    NORMALIZED    PHASE        NORMALIZED 

    NO         (HZ)     COMPONENT    COMPONENT    (DEG)       PHASE (DEG) 

 

     1     1.000E+04    2.301E-01    1.000E+00   -1.798E+02    0.000E+00 

     2     2.000E+04    1.985E-04    8.627E-04    8.835E+01    4.480E+02 

     3     3.000E+04    1.488E-05    6.466E-05   -1.754E+02    3.640E+02 

     4     4.000E+04    8.096E-06    3.519E-05   -5.419E+01    6.651E+02 

     5     5.000E+04    7.842E-06    3.409E-05   -3.482E+01    8.643E+02 

     6     6.000E+04    1.508E-05    6.553E-05   -1.586E+02    9.203E+02 

     7     7.000E+04    7.242E-06    3.148E-05   -5.739E+01    1.201E+03 

     8     8.000E+04    4.405E-06    1.915E-05   -3.189E+01    1.407E+03 

     9     9.000E+04    1.069E-05    4.647E-05    2.884E+01    1.647E+03 

 

     TOTAL HARMONIC DISTORTION =   8.710152E-02 PERCENT 

 

 

Slika 3.15. Rezultati simulacije pojačavača sa zajedničkim emitorom 

b) Kao što je ranije rečeno, komanda .DC služi za zadavanje uslova analize rada kola u 

prisustvu sporo promenljivih signala. Njome se bira parametar kola za čiji zadati 

skup vrednosti treba ponoviti jednosmernu analizu. Do sada se u .DC analizi vršila 

promena napona i struja, ali i temperatura radne okoline se može ravnopravno 

koristiti, korišćenjem gobalnog parametra TEMP: 

.DC LIN TEMP 0 120 5 

           T i m e

0 s 2 0 u s 4 0 u s 6 0 u s 8 0 u s 1 0 0 u s 1 2 0 u s 1 4 0 u s 1 6 0 u s 1 8 0 u s 2 0 0 u s

V ( 2 , 0 ) V ( 7 , 0 )

- 4 0 0 m V

- 2 0 0 m V

0 V

2 0 0 m V

4 0 0 m V

U ovom primeru, će se globalni parametar TEMP linearno menjati od 0 do 120 sa 

korakom 5. Dakle, treba definisati ovu analizu i postaviti da nema naizmenične 

polarizacije, tj. da je vrednost generatora V1=0. Nakon pokretanja simulacije, na x-osi 

biće parametar temperature TEMP. Rezultati simulacije prikazani su na slici 3.16, može 

se videti da sa povećanjem temperature dolazi do opadanja vrednosti jednosmernog 

napona na kolektoru tranzistora V(4). 

 

Slika 3.16. Zavisnost jednosmernog napona na kolektoru od temperature 

ZADATAK 22. Simulirati rad bipolarnog tranzistora u AC režimu u opsegu frekvencija od 

1kHz do 50MHz. Jednosmerna polarizacija baze je 0.6V, a amplituda AC superponiranog 

signala 50mV. Kolektor je polarisan jednosmernim naponom od 5V. Prikazati zavisnost 

strujnog pojačanja tranzistora u funkciji frekvencije ulaznog signala. 

 

Slika 3.17. Kolo bipolarnog tranzistora za simulaciju AC režima rada 

Rešenje: 

Za kolo sa slike 3.17 može se napisati sledeća netlista: 

**AC analiza bipolarnog tranzistora 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

Q1 3 2 0 0 Q2N2222A 1 

VAC 2 1 AC 50mV 90 

VBE 1 0 0.6V 

VCE 3 0 5V 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

           T E M P

0 1 0 2 0 3 0 4 0 5 0 6 0 7 0 8 0 9 0 1 0 0 1 1 0 1 2 0

V ( 4 )

1 4 . 4 V

1 4 . 6 V

1 4 . 8 V

1 5 . 0 V

1 5 . 2 V

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0 RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.AC DEC 5 1kHz 50Meg 

.PROBE I(VBE) I(VCE) 

.END 

 

Da bi prikazali zavisnost strujong pojačanje u funkciji frekvencije ulaznog signala u ulaznom 

fajlu je potrebno komandom .PROBE zadati štampanje struje kolektora i struje baze. U 

prozoru za prikaz rezultata simulacije izabere se Trace/Add Trace, i u Trace expression unese 

formula: 20*LOG10(I(VCE)/I(VBE)) koja predstavlja izraz za pojačanje tranzistora. Da bi 

se dobio prikaz kao na slici 3.18, potrebno je desnim klikom na y-osi izabrati opciju User 

Defined i uneti da budu prikazani rezultati u opsegu od 100m do 1k. Grafik sa slike 3.18 

predstavlja Bodeov dijagram sa koga se može očitati granična frekvencija datom bipolarnog 

tranzistora. 

 

 
 

Slika 3.18. Rezultati simulacije rada bipolarnog tranzistora u AC režimu 

ZADATAK 23. Simulirati i analizirati rad bipolarnog tranzistora kao invertora sa slike 3.19. 

Parametri generatora V1 su: napon logičke nule 0V, napon logičke jedinice 0.6V, vreme 

kašnjenja signala 1μs, vreme trajanja rastuće/opadajuće ivice 100ns, vreme trajanje signala 

logičke jedinice 5μs, period signala 10.2μs. Vreme trajanja simulacije je 50μs, pri čemu se 

vrednosti štampaju svakih 50ns. 

 
 

Slika 3.19. Kolo bipolarnog tranzistora kao invertora 

 

           F r e q u e n c y

1 . 0 K H z 3 . 0 K H z 1 0 K H z 3 0 K H z 1 0 0 K H z 3 0 0 K H z 1 . 0 M H z 3 . 0 M H z 1 0 M H z 3 0 M H z 1 0 0 M H z

2 0 * L O G 1 0 ( I ( V C E ) / I ( V B E ) )

- 2 0

0

2 0

4 0

6 0

Rešenje: 

 

Za kolo sa slike 3.19 može se napisati sledeća netlista: 

 
**Bipolarni tranzistor kao invertor 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

Q1 1 2 0 0 Q2N2222A 1 

VBE 2 0 PULSE (0 0.7V 1u 100ns 100ns 0.5u 1.2u) 

VCE 3 0 5V 

R1 3 1 10k 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0  RC=1  CJC=1p  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  MJE=.377   

+VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  RB=10) 

.TRAN 50ns 10u uic 

.PROBE V(1,0) V(2,0) I(VCE) 

.END 

 

Na slici 3.20 prikazani su rezultati simulacije za R1=100kΩ i CJC=1fF. Naime, bipolarni 

tranzistor ima veliku baznu otpornost i kapacitivnost, pa iz tog razloga postoji kašnjenje 

izlaznog signala. Pik na prednjoj ivici izlaznog signala javlja se zbog male vrednosti 

kapacitivnosti CJC, koja se iz tog razloga brzo puni. Da bi se eliminisao pik treba povećati 

ovu kapacitivnost na CJC=1pF. Da bi se kašnjenje još više smanjilo treba smanjiti vrednost 

otpornosti u kolu kolektora R1=10kΩ, čime se povećava struja u kolu. Ako posmatramo 

struju kolektora bipolarnog tranzisotra možemo zaključiti da struja teče kroz tranzistor kada 

je na ulazu napon logičke nule, a ne teče kada je na ulazu napon logičke jedinice. Ovo je 

jedan od razloga zašto se u savremenim kolima koriste CMOS invertori, koji će biti obraĎeni 

u sledećem poglavlju. Na slici 3.21 prikazane su vrednosti napona i struja u kolu bipolarnog 

tranzistora kao invertora sa podešenim tehnološkim parametrom CJC i eksternim otpornikom 

R1. 

 

Slika 3.20. Rezultati simulacije bipolarnog tranzistora kao invertora – inicijalne vrednosti 

parametara 

           T i m e

0 s 1 u s 2 u s 3 u s 4 u s 5 u s 6 u s 7 u s 8 u s 9 u s 1 0 u s

V ( 1 , 0 ) V ( 2 , 0 )

0 V

2 . 5 V

5 . 0 V

S E L > >

- I ( V C E )

0 A

2 5 0 u A

5 0 0 u A



 
62  

 

Slika 3.21. Rezultati simulacije bipolarnog tranzistora kao invertora – podešene vrednosti 

parametara 

ZADATAK 24. Simulirati prenosnu karakteristiku bipolarnog tranzistora kao invertora. 

Šema kola prikazana je na slici 3.19. 

Rešenje: 

Da bi simulirali prenosnu karakteristiku, koja predstavlja zavnisnost izlaznog napna u 

funkciji ulaznog napona, podesićemo trajanje prethodne simulacije na 2μs. Drugi način za 

dobijanje prenosne karakteristike je DC analiza kola. Naime, impulsni generator se menja 

jednosmernim izvorom čija se vrednost menja od 0V do 1V sa korakom 0.01V.  Netlista kola 

prikazana da data je u nastavku, a rezultati simulacije prikazani su na slici 3.22. 

**Bipolarni invertor, PRENOSNA KKA 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

Q1 1 2 0 0 Q2N2222A 1 

VBE 2 0 1V 

VCE 3 0 5V 

R1 3 1 10k 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0 RC=1  CJC=7.306P MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.DC VBE 0 1V 0.01V 

.PROBE V(1,0) V(2,0)  

.END 

 

 

Slika 3.22. Rezultati simulacije prenosne karakteristike bipolarnog tranzistora kao invertora 

           T i m e

0 s 1 u s 2 u s 3 u s 4 u s 5 u s 6 u s 7 u s 8 u s 9 u s 1 0 u s

V ( 1 , 0 ) V ( 2 , 0 )

0 V

2 . 5 V

5 . 0 V

S E L > >

- I ( V C E )

0 A

2 5 0 u A

5 0 0 u A

           V B E

0 V 0 . 1 V 0 . 2 V 0 . 3 V 0 . 4 V 0 . 5 V 0 . 6 V 0 . 7 V 0 . 8 V 0 . 9 V 1 . 0 V

V ( 1 , 0 )

0 V

2 . 0 V

4 . 0 V

6 . 0 V

ZADATAK 25. Simulirati rad kola sa slike 3.19 kada je ono opterećeno sa C1=1pF. 

Rešenje: 

Netlista za kolo sa slike je: 

**Bipolarni invertor sa opterecenjem, vremenska analiza 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

Q1 1 2 0 0 Q2N2222A 1 

VBE 2 0 PULSE (0 0.65V 1u 100ns 100ns 0.5u 1.2u) 

VCE 3 0 5V 

R1 3 1 10k 

C1 1 0 1p 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0 RC=1  CJC=7.306P MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.TRAN 50ns 7u uic 

.PROBE V(1,0) V(2,0)  

.END 

 

Rezultati simulacije prikazani su na slici 3.23.  

Slika 3.23. Rezultati simulacije kola bipolarnog invertora sa opterećenjem 

ZADATAK 26. Simulirati rad kola sa slike 3.24 u vremenskom domenu. Generator V1 je 

jednosmerni izvor vrednosti 5V, dok je V2 impulsni generator (logička nula: 0V, logička 

jedinici: 4.4V, vreme trajanja rasta prednje/zadnje ivice: 100ns, period: 12μs). Iskorišćen je 

NPN bipolarni tranzistor 2N2222A, i PNP bipolarni tranzistor DI_DP350T05. Vrednost 

opterećenja je C1=1pF. 

           T i m e

0 s 0 . 5 u s 1 . 0 u s 1 . 5 u s 2 . 0 u s 2 . 5 u s 3 . 0 u s 3 . 5 u s 4 . 0 u s 4 . 5 u s 5 . 0 u s 5 . 5 u s 6 . 0 u s 6 . 5 u s 7 . 0 u s

V ( 1 , 0 )

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

6 . 0 V

Slika 3.24. Kolo invertora sa bipolarnim tranzistorima 

Rešenje: 

Netlista kola sa slike opisana je u nastavku, dok su rezultati simulacije prikazani na slici 

3.25. 

** 

.OPTIONS numdgt=8 RELTOL=0.01 ITL2=120 stepgmin gmin=1e-16 

Q1 2 1 0 Q2N2222A 

Q2 2 1 3 DI_DP350T05 

VCE 3 0 5V 

VBE 1 0 PULSE (0 4.4V 1u 100n 100n 5u 12u) 

C1 2 0 1pF 

.MODEL DI_DP350T05  PNP (IS=177f NF=1.00 BF=185 VAF=337 

+ IKF=0.182 ISE=68.4p NE=2.00 BR=4.00 NR=1.00 

+ VAR=20.0 IKR=0.450 RE=1.26 RB=5.05 RC=0.505 

+ XTB=1.5 CJE=73.2p VJE=1.10 MJE=0.500 CJC=11.8p VJC=0.300 

+ MJC=0.300 TF=3.04n TR=116n EG=1.12 ) 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

+IKR=0 RC=1  CJC=7.306p  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  

+MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  

+RB=10) 

.TRAN 50n 100u  UIC 

.PROBE V(2,0), V(1,0) I(VCE) 

.END 

 

 

Slika 3.25. Rezultati simulacije kola sa slike 3.24. 

 

           T i m e

0 s 1 0 u s 2 0 u s 3 0 u s 4 0 u s 5 0 u s 6 0 u s 7 0 u s 8 0 u s 9 0 u s 1 0 0 u s

V ( 1 , 0 ) V ( 2 , 0 )

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

6 . 0 V

4.  
MODELIRANJE I SIMULACIJA 
MOS TRANZISTORA 

MOS tranzistor je aktivna elektronska komponenta sa četiri izvoda: gejt, sors, drejn i 

supstrat. Kod diskretnih MOS tranzistora spolja su dostupni izvodi gejta, sorsa i drejna. MOS 

tranzistor je po strukturi sastavljen od metala, oksida i poluprovodnika, odakle potiče 

akronim MOS. Postoje dva tipa MOS tranzistora: NMOS i PMOS. MOS tranzistor je 

naponom kontrolisana komponenta, za razliku od bipolarnog tranzistora koji je strujom 

kontrolisana komponenta. Struja gejta uvek je jednaka nuli. Napon se dovodi na gejt 

tranzistora. Ukoliko je taj napon manji od napona praga tranzisotor ne vodi, nalazi se u 

zakočenju, tako da je struja drejna jednaka nuli. Napon praga je minimalni napon potreban 

da tranzistor provede, tj. minimalni napon koji je potrebno dovesti na gejt tranzistora da bi u 

supstratu, neposredno ispod oksida gejta došlo do inverzije, odnosno da se izgradi provodni 

kanal od sorsa do drejna tranzistora. Naime, kod NMOS oblasti sorsa i drejna prave se od n-

tip poluprovodnika, dok je supstrat p-tipa. Kod PMOS tranzistora je obrnuto: sors i drejn su 

p-tipa, a supstrat je n-tipa. Inverzija znači da se pod dejstvom električnog polja nosioci 

naelektrisanja (šupljine kod NMOS-a) potiskuju ka elektrodi supstrata, tako da nastane kanal 

n-tipa. Jačina struje drejna zavisi i od napona na gejtu i od napona na drejnu. Definiše se 

napon saturacije VDS(sat) kao razlika napona na gejtu (VGS) i napona praga tranzistora (VT). 

Dok je napon na drejnu VDS manji od napona saturacije struja raste sa porastom napona, 

tranzistor radi u triodnoj oblasti. Kada je napon VDS postane veći od napona saturacije, dolazi 

do prekida kanala, odnosno struja kroz tranzistor postaje konstantna, tranzistor radi u 

zasićenju. MOS tranzistor se korisiti kao prekidač (radi u zakočenju/triodnoj oblasti) ili kao 

pojačavač (radi u oblasti zasićenja). Poprečni presek MOS tranzistora prikazan je na slici 4.1. 

 

Slika 4.1. Poprečni presek MOS tranzistora 

Na slici 4.2 prikazana je najprostija ekvivalentna šema MOS tranzistora. Ulazno kolo čine 

kondenzator (sastavljen od metala gejta, oksida gejta i kontakta supstrata) i otpornik 

(otpornost izolacije predstavlja otpornost ovog otpornika). Vrednost ovog otpornika je 

velika, tako da nema proticanja stuje, zapravo postoji samo struja curenja niske vrednosti. 

Osnovu ekvivalentnog kola čini naponom kontrolisani strujni generator. Izlazno kolo čini 

otpornost drejna tranzistora. 

 

Slika 4.2. Najprostije ekvivalentno kolo MOS tranzistora 

Postoji veliki broj složenih modela čiju osnovu čini opisano kolo. Modeli su toliko složeni da 

neki od njih uključuju efekte poput brzine saturacije kod tranzistora sa kratkim kanalom gde 

se vrednost napona praga menja sa naponom na drejnu. MOS tranzistor opisuje se 

modelskom karticom, primer modelske kartice dat je u nastavku: 

Primer modelske kartice MOS tranzistora 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

Modelska kartica počinje naredbom .MODEL, potom se zadaje proizvoljni naziv 

komponente (najčešće komercijalni naziv MOS tranzistora) nakon čega se navodi tip 

komponente (u ovom NMOS tranzistor)  i slede parametri modela. Ovde će biti diskutovani 

najvažniji parametri, dok se detaljni opisi parametara sa podrazumevanim vrednostima i 

jedinicama dostupni u [PSpcRef]. Parametar LEVEL definiše nivo modela, najčešći je BSIM 

(Berkeley SIM model) koji pripada nivou 4, dok su viši nivoi nadogradnja nivoa 4. Do danas 

je razvijeno 72 nivoa modela MOS tranzistora. Model iz primera je nivoa 3. PHI parametar 

odreĎuje površinski potencijal, tj. potencijal pri kome dolazi do inverzije supstrata, u ovom 

modelu ima vrednost 0.6V, što je i podrazumevana vrednost. Parametar TOX predstavlja 

debljinu oksida gejta, i na osnovu ove vrednosti SPICE preračunava kapacitivnost oksida 

gejta. Paramerom XJ zadaje se dubina PN spoja. Na osnovu vrednosti ovog parametra 

izračunava se efektivna vrednost dužine kanala kao Leff=L-2(0.8·XJ). Naime, tokom procesa 

difuzije 0.8·XJ odlazi ispod oksida gejta, što skraćuje kanal i na strani sorsa i na strani 



 
67 Modeliranje i simulacija mikroelektronskih komponenata i kola 

drejna, kao što je ilustrovano na slici 4.3. Inače, vrednost dužine podilaženja difuzije ispod 

oksida gejta može biti iskazana parametrom lateralne difuzije LD.  

 

Slika 4.3. Ilustracija značenja parametara XJ i LD 

TPG parametar definiše tip materijala gejta. Naime, gejt može biti napravljen od metala: 

aluminijuma sa bakrom, ili u savremenim kolima od volframa i titana (TPG=0), materijala 

suprotnog tipa od supstrata (TPG=+1) ili istog tipa kao supstrat (TPG=-1). Podrazumevana 

vrednost, i u ovom primeru je +1, što znači da ako je supstrat p-tipa, onda su sors i drejn n-

tipa, a gejt se pravi od polySi dopiranog primesama n-tipa. Kroz ovaj parametar odreĎen je 

znak razlike izlaznih radova izmeĎu metala i supstrata (ψMS), koja utiče na napon praga 

tranzistora.  

Parametar VTO predstavlja vrednost napona praga MOS tranzistora bez polarizacije, dok je 

KP parametar transkonduktanse izražen u A/V
2
. Kroz ovaj parametar definiše se 

pokretljivost nosilaca u kanalu tranzistora. Sa povećanjem napona polarizacije gejta dolazi 

do smanjenja pokretljivosti nosilaca u kanalu, a samim tim i transkonduktanse gm (opada 

pojačanje- slika 4.4), tako da su izlazne karaktristike zbijenije u familiji karakteristika za 

veće vrednosti napona na gejtu, što će kasnije biti pokazano simulacijom. Parametar 

površinske pokretljivosti je UO. Ukupna pokretljivost zavisna je i od debljine oksida gejta 

TOX i parametra THETA- faktor modulacije pokretljivosti.  

 

Slika 4.4. Zavisnost transkonduktanse od napona polarizacije gejta 

Parametrom RSH definiše se slojna otpornost, a ukupna otpornost oblasti drejna i sorsa 

izračnava se na osnovu dimenzija tranzistora. Dimenzije tranzistora podrazumevaju 

definisanje širine kanala W i dužine kanala L. Ovi parametri mogu se zadati u okviru 



 
68  

modelske kartice ili pri definiciji MOS tranzistora u netlisti. Podrazumevana jedinica za ove 

dimenzije je metar, pa treba koristiti numeričke sufikse (najčešće U što odgovara E-6). 

U okviru modelske kartice definiše se i efektivni broj stanja u oksidu i na meĎupovršini 

Si/SiO2 kroz parametre NSS, NFS, kao i kapacitivnosti u MOS strukturi: CGDO, CGBO, 

CGSO, itd. izražene u F/m. Tako se, na primer, kapacitivnost izmeĎu gejta i supstrata dobija 

množenjem CGBO sa širinom kanala W.  

Na slici 4.5 prikazana je ekvivalentna šema MOS tranzistora u PSpice-u. PSpice ima set 

jednačina u kojima koristi vrednosti parametara zadatih u modelskoj kartici. Na primer,  

definisane su jednačine po kojima se računa struja drejna za napone na gejtu manje od 

napona praga i jednačine za struju drejna kada je napon na gejtu veći od napona praga. Krive 

treba da budu kontinualne, ali se dešava da u tački spajanja nema konvergencije u SPICE-u. 

Ilustracija realnih zavisnosti i zavisnosti dobijenih rešavanjem jednačina data je na slici 4.6.  

 

Slika 4.5. Ekvivalentna šema MOS tranzistora u PSpice-u 

 

Slika 4.6. Ilustracija realnih zavisnosti i zavisnosti dobijenih rešavanjem jednačina 

Primer definisanja (pozivanja) MOS tranzisotra – detaljno u [PSpcRef] 

M1 2 1 0 0 CMOSN L=1E-6 W=10E-6 

*  | | | |     |    |     | širina kanala 

*  | | | |     |    | dužina kanala 

*  | | | |     | naziv modela 

*  | | | | Balk 

*  | | | Sors 

*  | | Gejt 

*  | Drejn 

 

Kao što je ilustrovano MOS tranzistor se poziva navoĎenjem rezervisanog slova M sa 

nazivom komponente u nastavku, nakon čega se redom navode čvorovi u kolu gde su vezani 

drejn, gejt, sors i balk (supstrat). Iza čvorova navodi se naziv modela, koji treba da odgovara 

nazivu navedenom u modelskoj kartici. Pri definisanju MOS tranzistora navodi se dužina i 

širina kanala, korišćenjem L= i W=. Kao što je već napomenuto, podrazumevana jedinica je 

metar, pa treba koristiti numeričke sufikse. 

Treba napomenuti da se pri tehnološkoj realizaciji pravi više tranzistora kratkih kanala, pa se 

vezuju paralelno kako bi na visokim frekvencijama tranzistor radio. Naime, kada se dovede 

napon na gejt tranzistora dolazi do punjenja kapacitivnosti oksida gejta i kada se ona napuni, 

tranzistor se uključuje. Kada su frekvencije visoke, ova kapacitivnost ne bi stigla da se 

napuni, pa tranzistor ne bi radio. Ovo izdvaja MOS od bipolarnog tranzistora koji ne sme da 

se vezuje paralelno, jer jedan od tranzistora može da povuče veću sruju od drugih, pa bi zbog 

velikog pojačanja, došlo do velike disipacije na tom tranzistoru i pregorevanja istog. Kod 

MOS-a pojačanje opada sa povećanjem temperature, pa ovaj problem ne postoji. 

Smanjenjem dužine kanala MOS tranzistora, naponom na gejtu nije moguće vršiti kontrolu 

nad tranzistorom, a da bi se smanjili ovi efekti radi se nisko dopiranje oblasti drejna/sorsa – 

LDD.  

ZADATAK 27. Simulirati izalazne karakteristike NMOS tranzistora (slika 4.7) čija je 

modelska kartica data u primeru. Dužina kanala je 1μm, a širina kanala je 10μm. Napon 

polarizacije gejta menja se od 0V do 5V sa korakom 1V. Napon polarizacije drejna menja sa 

u istom opsegu sa korakom 0.1V.  

 

Slika 4.7. Kolo NMOS tranzistora za snimanje izlazne IV karakteristike 

Rešenje: 

Kolo sa slike 4.7 može se opisati sledećom netlistom: 

**Izlazne karakteristike NMOS tranzistora 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

M1 2 1 0 0 CMOSN L=1E-6 W=10E-6 

VDD 2 0 5V 

VGS 1 0 1V 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.DC VDD 0V 5V 0.1V 

.STEP VGS 0V 5V 1V 

 

.PROBE V(1,0) I(VDD) 

 

.END 

 

Rezultati simulacije prikazani su na slici 4.8. Treba prikazati struju sa znakom minus kako bi 

bili dati realni rezultati. 

 

Slike 4.8. Simulirane izlazne karakteristike NMOS tranzistora 

ZADATAK 28. Korišćenjem kola sa slike 4.7, simulirati prenosnu i Gummelovu 

karakteristiku i prikazati zavisnost transkonduktanse u funkciji napona na gejtu. Simulirati i 

analizirati karakteristike tranzistora sa promenom temperature. 

Rešenje: 

Da bi simulirali tražene karakteristike modifikujemo prehodnu netlistu, radi se DC analiza 

kola, pri čemu se menja vrednost generatora VGS, a vrednost generatora VDD je konstantna: 

 

           V D D

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

-  I ( V D D )

0 A

2 . 0 m A

4 . 0 m A

6 . 0 m A

**Karakteristike NMOS tranzistora 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

M1 2 1 0 0 CMOSN L=1E-6 W=10E-6 

VDD 2 0 5V 

VGS 1 0 1V 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

.DC VGS 0V 5V 0.5V 

.PROBE V(1,0) I(VDD) 

.END 

 

Rezultati simulacije prikazani su na slikama 4.9, 4.10 i 4.11, respektivno. Nakon pokretanja 

simulacije u Trace Expression se za prenosnu karakteristiku unosi se –I(VDD), za 

Gummelovu karakteristiku unosi se LOG10(-I(VDD)), dok se za prikaz zavisnosti 

transkonduktanse štampa vrednost količnika izvoda: D(I(VDD))/D(VGS). 

 

 

 

Slika 4.9. Prenosna karakteristika NMOS tranzistora 

 

Slika 4.10. Gummelova karakteristika NMOS tranzistora 

           V G S

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

-  I ( V D D )

0 A

2 . 0 m A

4 . 0 m A

6 . 0 m A

           V G S

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

L O G 1 0 ( I ( V D D ) )

- 8 . 0

- 6 . 0

- 4 . 0

- 2 . 0

Slika 4.11. Zavisnost transkonduktanse NMOS tranzistora od napona na gejtu 

Da bi simulirali temperaturnu zavisnost karakteristika NMOS tranzistora koristićemo 

naredbu .TEMP -25 25 75 125 150. Rezultati simulacije prikazani su na slici 4.12. 

Može se uočiti da sve sve karakteristike seku u jednoj tački. Pri odreĎenoj polarizaciji, bez 

obzira na temperaturu sve karakterstike seku se u jednoj tački koja se naziva tačka nultog 

temperaturnog koeficijenta – ZTC (Zero Temperature Coeficient). Odavde zaključujemo da 

se MOS tranzistor ne može koristiti kao senzor temperature. MeĎutim, ako se tranzistor 

polariše da radi u potpragovskom režimu rada, tada se može uočiti linearna zavisnost struje 

od temperature. Dakle, u ovom slučaju MOS tranzistor se može koristiti kao senzor 

temperature. Na slici 4.13 prikazani su rezultati simulacije kada je VDD=1V, a VGS se 

menja od 0V do 3V, na više različitih temperatura. Da bi zavisnost bila uočljivija prikazani 

su rezultati u log-lin razmeri. 

 

Slika 4.12. Temperaturna zavisnost karakteristika NMOS tranzistora 

           V G S

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

D ( I ( V D D ) ) /  D ( V G S )

- 1 . 5 m

- 1 . 0 m

- 0 . 5 m

0

0 . 5 m



 
73 Modeliranje i simulacija mikroelektronskih komponenata i kola 

 

Slika 4.13. Temepraturna zavisnost karakteristika NMOS tranzistora u potpragovskom 

režimu rada 

ZADATAK 29. Simulirati prenosnu karakteristiku CMOS invertora prikazanog na slici 4.14. 

Analizirati potrošnju ovog kola.  

 

 

Slika 4.14. Šema CMOS invertora 

Rešenje: 

Kolo sa slike 83 može se opisati sledećom netlistom: 

**CMOS invertor 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

M1 3 1 0 0 CMOSN L=2.5E-6 W=10E-6 

M2 3 1 2 2 CMOSP L=2.5E-6 W=25E-6 

VGS 1 0 1V 

VDD 2 0 5V 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

           V G S

0 V 0 . 2 V 0 . 4 V 0 . 6 V 0 . 8 V 1 . 0 V 1 . 2 V 1 . 4 V 1 . 6 V 1 . 8 V 2 . 0 V 2 . 2 V 2 . 4 V 2 . 6 V 2 . 8 V 3 . 0 V

L O G 1 0 ( - I ( V D D ) )

- 1 2

- 8

- 4

0



 
74  

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.DC VGS 0 5V 0.25V 

.PROBE V(1,0) V(3,0) I(VDD)  

.END 

 

Rezultati simulacije prikazani su na slici 4.15. Strimina karakteristike menjaće se sa 

promenom širine i/ili dužine kanala tranzistora. Ako prikažemo zavisnost struje drejna u 

funkciji napona na gejtu možemo uočiti da oba tranzistora vode samo u prelaznom režimu. 

Korišćenjem PSPICE-a možemo prikazati i zavisnost snage disipacije od napona na gejtu. 

Kao što je ranije objašnjeno, u delu Trace Expression unosimo izraz: I(VDD)*V(3,0) i 

dobijamo željeni grafik. Vidimo da je u piku potrošnja kola 1.2mW, što se može smanjiti. 

Potrošnju možemo smanjiti podešavanjem napona praga na istu vrednost (na primer 0.9V po 

apsolutnoj vrednosti za oba tranzistora) i povećanjem slojne otpornosti (na primer RSH 

parametar modela povećamo 10 puta za oba tipa tranzistora, čime se dobija „šira“ 

karakteristika potrošnje). 

 

Slika 4.15. Rezultati simulacije prenosne karakteristike CMOS invertora 

ZADATAK 30. Simulirati i analizirati rad kola sa slike 4.14 u vremenskom domenu ukoliko 

je na izlaz kola vezano kapacitivno opterećenje kapacitivnosti 1pF. Kolo se pobuĎuje 

impulsnim signalom (logička nula: 0V, logička jedinica: 5V, kašnjenje: 0s, vreme trajanja 

prednje/zadnje ivice: 100ns, period: 1.2μs).  

Rešenje: 

Kolo sa slike 4.14 može se opisati sledećom netlistom: 

**CMOS invertor TRAN analiza 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

           V G S

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

V ( 3 , 0 )

0 V

2 . 5 V

5 . 0 V

S E L > >

-  V ( 3 , 0 ) *  I ( V D D )

0 W

1 . 0 m W

2 . 0 m W
- I ( V D D )

0 A

2 0 0 u A

4 0 0 u A

M1 3 1 0 0 CMOSN L=5E-6 W=2E-6 

M2 3 1 2 2 CMOSP L=5E-6 W=2E-6 

VGS 1 0 PULSE (0 5V 1u 5ns 5ns 0.5u 1.2u) 

VDD 2 0 5V 

C1 3 0 1pF 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.TRAN 50ns 5u uic 

.PROBE V(1,0) V(3,0) I(VDD)  

.END 

 

Rezultati simulcije prikazani su na slici 4.16. Ako posmatramo potrošnju kola, može se 

uočiti da se struja javlja samo u prelaznom režimu, kao i da ima različite vrednosti kada kolo 

prelazi iz stanja „0“ u stanje „1“ i obrnuto. Ako je frekvencija ulaznog signala veća, na 

primer: Trise i Tfall su 5ns, vreme trajanja logičke jedinice 0.1μs, a perioda 0.21μs, kolo neće 

raditi ispravno. Treba smanjiti vrednost opterećenja ukoliko je to moguće (na primer 

0.01pF). Da bi kolo radilo treba promeniti tehnologiju, tj. u simulaciji promeniti vrednost 

parametera transkonduktanse. Povećanjem KP deset puta, kolo će ispravno raditi. Ukoliko 

nije moguća promena tehnologije, funkcionalnost kola postiže se menjanjem dimenzija 

tranzistora. Na primer, ako postavimo W=20μm, kolo će ispravno raditi (slika 4.17). Ako na 

supstratskoj pločici nemamo dovoljno prostora, može se promeniti dužina kanala tranzistora. 

Na primer, dato kolo će ispravno raditi ako L bude jednako 1μm. Što je dužina kanala veća, 

karakteristika će biti „ravnija“.  

 

Slika 4.16. Rezultat simulacije kola CMOS invertora u vremenskom domenu 

           T i m e

0 s 0 . 5 u s 1 . 0 u s 1 . 5 u s 2 . 0 u s 2 . 5 u s 3 . 0 u s 3 . 5 u s 4 . 0 u s 4 . 5 u s 5 . 0 u s

V ( 1 , 0 ) V ( 3 , 0 )

0 V

2 . 5 V

5 . 0 V

-  I ( V D D )

0 A

5 0 u A

1 0 0 u A

S E L > >

Slika 4.17. Rezultat simulacije kola CMOS invertora pri većim frekvencijama, sa 

promenjenim dimenzijama tranzistora 

ZADATAK 31. Kolo CMOS invertora sa slike 4.14 opisati kao potkolo. Korišćenjem tog 

potkola formirati kolo prstenastog oscilatora – ring oscilatora (slika 4.18). Simulirati rad 

ovog kola u vremenskom domenu i analizirati rad kola u zavisnosti od napona napajanja, 

dimenzija tranzistora i temperature. 

 

Slika 4.18. Kolo prstenastog oscilatora 

Rešenje: 

Oscilatori su kola koja ne dostižu ravnotežno stanje. Prstenasti oscilator se sastoji iz 

neparnog broja CMOS invertora kod koga se izlaz prethodnog vezuje na ulaz sledećeg 

invertora u nizu. Izlaz poslednjeg invertora vezuje se na ulaz prvog invertora. Na slici 4.18 

prikazan je prstenasti oscilator sastavljen iz sedam invertora. Kroz svaki stepen, tj. svaki 

invertor postoji kašnjenje signala τ. Kašnjenje svakog identičnog stepena odreĎuje se na 

osnovu frekvencije oscilovanja f=1/2Nτ, gde je N broj invertora u kolu prstenastog 

oscilatora. Ovo kolo koristi se za testiranje tehnologije.  

Kolo invertora opisano je kao potkolo na način kako je ranije opisano. Da bi kolo radilo 

neophodno je napajanje VDD. Kako bi simulacija bila funkcionalna neophodno je zadati 

početne uslove korišćenjem naredbe .IC. Iza naredbe navodi se čvor u kome se zadaju 

početni uslovi i vrednost početnog uslova. U ovom primeru, napon u čvoru 1 je 0V, a u 

čvoru 2, napon je 1V. 

Primer komande .IC – detaljno u [PSpcRef] 

.IC V(1)=0 V(2)=1 

*     | čvor u kome se zadaju početni uslovi 

*     | | vrednost početnog uslova 

           T i m e

0 s 0 . 5 u s 1 . 0 u s 1 . 5 u s 2 . 0 u s 2 . 5 u s 3 . 0 u s 3 . 5 u s 4 . 0 u s 4 . 5 u s 5 . 0 u s

V ( 1 , 0 ) V ( 3 , 0 )

0 V

2 . 5 V

5 . 0 V

-  I ( V D D )
0 A

0 . 5 m A

1 . 0 m A

S E L > >

Kolo prstenastog osilatora može se opisati sledećom netlistom: 

** Ring Oscillator- prstenasti oscilator 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

.SUBCKT INV ulaz izlaz VDD GND 

 

M1 izlaz ulaz GND GND CMOSN L=5E-6 W=2E-6 

M2 izlaz ulaz VDD VDD CMOSP L=5E-6 W=2E-6 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.ENDS 

 

VDD 16 0 5V 

X1 1 2 16 0 INV 

X2 2 3 16 0 INV 

X3 3 4 16 0 INV 

X4 4 5 16 0 INV 

X5 5 6 16 0 INV 

X6 6 7 16 0 INV 

X7 7 1 16 0 INV 

  

.TRAN 50ns 1u  

.IC V(1)=0 V(2)=1 

.PROBE V(1,0) V(4,0) 

.END 

 

Rezultati simulacije prikazani su na slici 4.19. Očitavanjem pomoću kursora dobija se da je 

frekvencija oscilovanja kola f=35.13MHz, pa se dobija da je kašnjenje kroz svaki stepen 

τ=1/2Nf=2.03ns. Sa povećanjem vrednosti generatora napajanja VDD povećava se i 

frekvencija oscilovanja kola. Što je širina kanala tranzistora veća, to će biti i veća frekvencija 

oscilovanja. Ovo se jednostavno može proveriti zamenom vrednosti W i L pri definiciji 

tranzistora. UvoĎenjem naredbe .TEMP 25 125 simulira se ponašanje kola na dve različite 

temperature. Može se zaključiti da se sa povećanjem temperature povećava kašnjenje u kolu 

(slika 4.20). 

Slika 4.19. Rezultati simulacije sedmostepenog prstenastog oscilatora 

 

Slika 4.20. Rezultati simulacije sedmostepenog prstenastog oscilatora na različitim 

temperaturama 

ZADATAK 32. Kolo sa slike 4.21 radi kao senzor temperature. PMOS tranzistori 

predstavljaju ekvivalnente otpornike koji ograničavaju struju kroz NMOS tranzistore, koji 

rade u potpragovskoj oblasti rada tako da kolo radi kao senzor temperature. Promenom 

dužine i širine kanala tranzistora menja se vrednost ekvivalentne otpornosti. 

 

Slika 4.21. MOS tranzistor kao senzor temperature 

 

           T i m e

0 s 0 . 1 u s 0 . 2 u s 0 . 3 u s 0 . 4 u s 0 . 5 u s 0 . 6 u s 0 . 7 u s 0 . 8 u s 0 . 9 u s 1 . 0 u s

V ( 4 , 0 )

0 V

2 . 0 V

4 . 0 V

6 . 0 V

           T i m e

0 s 0 . 1 u s 0 . 2 u s 0 . 3 u s 0 . 4 u s 0 . 5 u s 0 . 6 u s 0 . 7 u s 0 . 8 u s 0 . 9 u s 1 . 0 u s

V ( 4 , 0 )

0 V

2 . 0 V

4 . 0 V

6 . 0 V

Rešenje:  

Dimenzije tranzistora su podešene, a brojne vrednosti date su u netlisti u nastavku. Za 

simulaciju vrednosti izlaznog napona u funkciji temperature korišćena je naredba .DC TEMP 

LIST 0 25 50 75 100, tako da će biti prikazan rezultat izlaznog napona na temperaturama 

zadatim ovom naredbom (slika 4.22). Kolo je slično PTAT kolu koje se obično realizuje 

pomoću bipolarnih tranzistora. U ovom primeru, razlika izlaznog napona je oko 60mV u 

temperaturnom opsegu od 100°C. 

**Senzor temperature sa MOS tranzistorima 

.OPTIONS numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

VDD 1 0 5 

M1 3 2 1 1 CMOSP W=5U L=3U 

M2 3 3 0 0 CMOSN W=5U L=7U 

M3 2 2 1 1 CMOSP W=5U L=3U 

M4 2 3 4 0 CMOSN W=10U L=3U 

R1 4 0 50K 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.DC TEMP LIST 0 25 50 75 100 

.PROBE V(4,0) 

.END 

 

 

Slika 4.22. Rezultat simulacije kola sa MOS tranzitorima kao senzora temperature 

           T E M P

0 1 0 2 0 3 0 4 0 5 0 6 0 7 0 8 0 9 0 1 0 0

V ( 4 , 0 )

1 4 0 m V

1 6 0 m V

1 8 0 m V

2 0 0 m V

ZADATAK 33. Simulirati i analizirati rad Šitovog kola sa slike 4.23. Predstaviti rezultate 

dobijene .DC i .TRAN analizom. 

 

Slika 4.23. Šema Šmitovog kola sa MOS tranzistorima 

Rešenje: 

Šmitovo kolo je kolo sa histerezisnom zavisnošću izlaznog od ulaznog napona. Ako se kolo 

simulira u .DC režimu potrebno je uraditi dve posebne simulacije, a zatim prikazati 

preklopljene rezultate u jednom prozoru. Na primer, prvo će biti uraĎena simulacija u kojoj 

se VGS menja od 0V do 5V sa korakom 0.25V, a nakon toga treba sačuvati dobijene 

rezultate i ponoviti simulaciju u kojoj se VGS menja od 5V do 0V sa korakom -0.25V. Kolo 

sa slike može se opisati sledećom netlistom: 

**Smit trigger DC analiza 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

M1 2 1 0 0 CMOSN L=5E-6 W=2E-6 

M2 2 1 5 0 CMOSN L=5E-6 W=2E-6 

M3 2 5 3 0 CMOSN L=5E-6 W=2E-6 

M4 4 1 5 3 CMOSP L=5E-6 W=2E-6 

M5 4 1 3 3 CMOSP L=5E-6 W=2E-6 

M6 4 5 0 3 CMOSP L=5E-6 W=2E-6 

VGS 1 0 1V 

VDD 3 0 5V 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.DC VGS 5 0 -0.25V 

.PROBE V(5,0) V(1,0) I(VDD)  

.END 

 

Da bi prikazali rezultate obe simulacije kliknemo: File/Append Waveform (.DAT) i 

izaberemo prethnodno sačuvan fajl koji treba preklopiti sa poslednjim rezultatom simulacije. 

Na slici 4.24 prikazani su rezultati simulacije Šmitovog kola. 

 

Slika 4.24. Rezultat simulacije Šmitovog kola 

Da bi se dobili isti rezultati simulacije .TRAN analizom, potrebno je nezavisni generator 

VGS definisati kao PWL (piecewise linear waveform). Kada se generator definiše na ovaj 

način signal se dobija interpolacijom pravih izmeĎu zadatih tačaka. Na primer, VGS 1 0 

PWL (0,0) (0.5u,5V) (1u,0V) znači da će signal koji daje generator VGS izmeĎu čvorova 1 i 

0 imati  vrednost 0V u trenutku 0s, zatim taj signal raste do 5V u tački 0.5μs, i na kraju 

opada do 0V u tački 1μs.  

Primer definisanja generatora tipa PWL – detaljno u [PSpcRef] 

VGS 1 0 PWL(0,0) (0.5u,5V) (1u,0) 

*           | | koliki napon? 

*           | na koliko sekundi? 

 

Naredbu .DC menjamo .TRAN naredbom: .TRAN 50N 1U UIC. Pored promene tipa 

generatora VGS i .TRAN analize ostatak netliste ostaje nepromenjen. Rezultati simulacije 

prikazani su na slici 4.25.  

           V G S

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

V ( 5 , 0 )

0 V

2 . 0 V

4 . 0 V

6 . 0 V

Slika 4.25. Rezultati simulacije Šmitovog kola korišćenjem .TRAN naredbe 

Da bi bila prikazana zavisnost izlaznog napona u funkciji ulaznog napona, menjamo x-osu. 

Duplim klikom na brojne vrednosti na x-osi otvara se dijalog prozor u kome treba kliknuti na 

dugme Axis Variable. Iz liste promenljivih treba izabrati V(1,0) što je ulazni napon. Konačni 

izgled rezultata simulacije Šmitovog kola korišćenjem .TRAN naredbe prikazan je na slici 

4.26. 

 

Slika 4.26. Zavisnost izlaznog napona Šmitovog kola od ulaznog napona 

Promenom napona praga tranzistora dolazi do pomeranja – šiftovanja histerezisne zavisnosti. 

Ako se napon praga PMOS tranzistora promeni na -1.9056V, a NMOS tranzisotra na 

1.786V, histerezisna petlja se pomera u levo. Povećanjem širine kanala kontrolišućih 

tranzistora M3 i M6, dolazi do širenja histerezisne petlje (na primer: W=20E-6). Povećanjem 

dužine kanala ovih tranzistora, histerezisna petlja biće uža. 

ZADATAK 34. Simulirati rad kola naponom kontrolisanog oscilatora – VCO (Voltage 

Controlled Oscillator) korišćenjem Šmitovog kola i strujnih izvora (slika 4.27). Šmitovo kolo 

opisati potkolom, kao i kolo CMOS invertora. Frekvencija VCO oscilatora kotroliše se 

naponom generatora Vcontrol. Početna stanja CMOS invertora podesiti korišćenjem naredbe 

.IC, ulazni napon treba inicijalizovati na 0V, a izlazni na 5V. Vrednost kapacitivnosti 

kondenzatora C1 je 100pF. Vrednost napona napajanja je VDD=5V. 

           T i m e

0 s 0 . 1 u s 0 . 2 u s 0 . 3 u s 0 . 4 u s 0 . 5 u s 0 . 6 u s 0 . 7 u s 0 . 8 u s 0 . 9 u s 1 . 0 u s

V ( 1 , 0 ) V ( 5 , 0 )

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

6 . 0 V

           V ( 1 , 0 )

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

V ( 5 , 0 )

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

6 . 0 V

Slika 4.27. Šema VCO oscilatornog kola 

Rešenje: 

Kolo sa slike 4.27 može se opisati sledećom netlistom: 

**VCO 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

VDD 6 0 5V 

Vcontrol 1 0 2V 

M1 2 1 0 0 CMOSN L=5E-6 W=2E-6 

M2 2 3 4 0 CMOSN L=5E-6 W=2E-6 

M3 5 3 4 6 CMOSP L=5E-6 W=2E-6 

M4 5 7 6 6 CMOSP L=5E-6 W=2E-6 

M5 7 1 0 0 CMOSN L=5E-6 W=2E-6 

M6 7 7 6 6 CMOSP L=5E-6 W=2E-6 

C1 4 0 100pF 

X1 4 8 6 0 SMIT 

X2 8 3 6 0 INV 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.TRAN 50ns 100u 

.IC V(8)=0V V(3)=5V 

.PROBE V(4,0) V(3,0) 

 

.SUBCKT SMIT ulaz izlaz VDD GND 

 

M11 s2 ulaz GND GND CMOSN L=5E-6 W=2E-6 

M12 s2 ulaz izlaz GND CMOSN L=5E-6 W=2E-6 

M13 s2 izlaz VDD GND CMOSN L=10E-6 W=2E-6 

M14 s4 ulaz izlaz VDD CMOSP L=5E-6 W=2E-6 

M15 s4 ulaz VDD VDD CMOSP L=5E-6 W=2E-6 

M16 s4 izlaz GND VDD CMOSP L=10E-6 W=2E-6 

 

.ENDS 

 

.SUBCKT INV INVulaz INVizlaz VDD GND 

 

M21 INVizlaz INVulaz GND GND CMOSN L=5E-6 W=2E-6 

M22 INVizlaz INVulaz VDD VDD CMOSP L=5E-6 W=2E-6 

 

.ENDS 

 

.END 

 

Rezultati simulacije VCO oscilatora dati su na slici 4.28. Napon V(4,0) testerastog oblika je 

na ulazu Šmitovog kola, dok je izlazni napon kola označen sa V(3,0). Naponom izmeĎu 

čvorova 1 i 0 kontroliše se frekvencija izlaznog signala. 

 

Slika 4.28. Rezultati simulacije VCO oscilatornog kola 

 

 

 

 

           T i m e

0 s 1 0 u s 2 0 u s 3 0 u s 4 0 u s 5 0 u s 6 0 u s 7 0 u s 8 0 u s 9 0 u s 1 0 0 u s
V ( 3 , 0 ) V ( 4 , 0 )

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

6 . 0 V

5. 
MODELIRANJE I SIMULACIJA 
KOMPONENATA SNAGE 

Komponente snage su one komponente koje vode struje veće od 1A. Ovde će biti reči o 

IGBT – bipolarnom tranzistoru sa izolovanim gejtom (Insulated Gate Bipolar Transistor). 

Poprečni presek komponente i simbol prikazani su na slici 5.1. Ulazni deo komponente je 

MOS transitor koji uključuje snažni bipolarni tranzistor koji čini izlazni deo IGBT-a. IGBT 

se sastoji iz velikog broja ćelija, koje su meĎusobno povezane metalnim linijama. Bipolarni 

tranzistor je ugraĎen u drejn MOS strukture. Struktura bez p+ oblasti je zapravo VDMOS 

tranzistor. UgraĎena PNP struktura smanjuje unutrašnju otpornost i povećava pojačanje 

komponente. IGBT zauzima veliku površinu, što je nophodno da bi se dobila komponenta sa 

velikim probojnim naponom. Struja kroz strukturu protiče horizonatalno, pa vertikalno, zato 

je fizički model IGBT-a složen pa se ova komponenta najčešće opisuje kao potkolo. Kada bi 

se umesto IGBT-a koristio bipolarni tranzistor snage, bila bi potrebna velika struja baze za 

uključenje tranzistora, što bi dovelo do velike disipacije snage na komponenti. IGBT je 

naponom kontrolisana komponenta, što joj daje prednost u primeni. U strukturi IGBT-a 

postoji i parazitna tiristorska struktura, mada je pokazano da se ona ne uključuje ako je 

otpornost bazno-emitorskog spoja te strukture mala, tako da se ne dostigne 0.6V, koji bi je 

uključili. Prilikom isključenja IGBT-a postoji kašnjenje, sporo se isključuje izbog postojanja 

kondenzatorske strukture koja treba da se isprazni. 

 

Slika 5.1. Poprečni presek i simbol IGBT-a 

ZADATAK 35. Simulirati IGBT kao potkolo sastavljeno iz bipolarnog i MOS tranzistora 

kao što je prikazano na slici 5.2. Generator polarizacije VDD vezan je izmeĎu emitora i 

kolektora, dok se IGBT kontroliše generatorom vezanim izmeĎu gejta i kolektora. Prikazati 

izazne karakteristike i prenosnu karakteristiku IGBT-a. 

 

Slika 5.2. Šema potkola IGBT-a 

Rešenje: 

Kolo sa slike 5.2 može se opisati sledećom netlistom: 

**IGBT: model sa potkolom 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

VDD 3 0 5V 

VGC 1 0 1V 

X1 1 0 3 IGBT 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL DI_DP350T05  PNP (IS=177f NF=1.00 BF=185 VAF=337 

+ IKF=0.182 ISE=68.4p NE=2.00 BR=4.00 NR=1.00 

+ VAR=20.0 IKR=0.450 RE=1.26 RB=5.05 RC=0.505 

+ XTB=1.5 CJE=73.2p VJE=1.10 MJE=0.500 CJC=11.8p VJC=0.300 

+ MJC=0.300 TF=3.04n TR=116n EG=1.12 ) 

 

.DC VDD 0V 5V 0.1V 

.STEP VGC 0V 5V 1V 

.PROBE V(1,0) I(VDD) 

 

.SUBCKT IGBT G S C 

M1 D G S S CMOSN L=5E-6 W=2E-6 

Q1 C D S C DI_DP350T05 

.ENDS 

 

.END 

Netlistu treba modifikovati za simulaciju prenosne karakteristike. Naime, radi se .DC 

analiza, pri čemu se menja vrednost generatora VGC i to od 0V do 10V sa korakom 0.5V, 

dok .STEP naredbu treba komentarisati – da ne bi bila kompajlirana. Simulirane izlazne 

karakteristike prikazane su na slici 5.3, dok je prenosna karakteristika u log-lin razmeri data 

na slici 5.4. 

 

Slika 5.3. Simulirane izlazne karakteristike IGBT-a 

 

Slika 5.4. Simulirana prenosna karakteristika IGBT-a 

Modeli IGBT-a mogu bitni znatno složeniji od opisanog, primer za to je potkolo IGBT-a 

prikazano na slici 5.5. Ovo potkolo sastoji se iz velikog broja zavisnih generatora, dioda i 

tranzistora. Ovde će biti dato potkolo komercijalnog IGBT tranzistora IRGBC40U, s kojim 

će biti rešeni sledeći primeri. 

           V D D

0 V 0 . 5 V 1 . 0 V 1 . 5 V 2 . 0 V 2 . 5 V 3 . 0 V 3 . 5 V 4 . 0 V 4 . 5 V 5 . 0 V

- I ( V D D )

0 A

0 . 4 m A

0 . 8 m A

1 . 2 m A

           V G C

0 V 1 V 2 V 3 V 4 V 5 V 6 V 7 V 8 V 9 V 1 0 V

- I ( V D D )

1 . 0 n A

1 . 0 u A

1 . 0 m A

1 0 0 m A

Slika 5.5. Potkolo komercijalnog IGBT tranzistora IRGBC40U 

ZADATAK 36. Simulirati osnovno kolo IGBT-a korišćenjem komercijalnog modela 

tranzistora, opisanog kao potkolo. Rezultate uporediti sa zavisnostima datim u datasheetu 

IRGBC40U tranzistora. 

Rešenje: 

Netlista kola je data u nastavku: 

**IGBT komercijalni model 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

VDD 2 0 5V 

VGC 1 0 1V 

X1 2 1 0 IRGBC40U 

 

.DC VDD 0 30 1 

.STEP VGC 0 20 5 

.PROBE I(VDD) 

 

.SUBCKT IRGBC40U 71 72 74 

* TERMINALS: C G E 

* 600 Volt 40 Amp 6.04NS N-Channel IGBT 

Q1 83 81 85 QOUT 

M1 81 82 83 83 MFIN L=1U W=1U 

DSD 83 81 DO 

DBE 85 81 DE 

RC 85 71 21.1M 

RE 83 73 2.11M 

RG 72 82 25.6 

CGE 82 83 1.42N 

CGC 82 71 1P 

EGD 91 0 82 81 1 

VFB 93 0 0 

FFB 82 81 VFB 1 

CGD 92 93 1.41N 

R1 92 0 1 

D1 91 92 DLIM 

DHV 94 93 DR 

R2 91 94 1 

D2 94 0 DLIM 

DLV 94 95 DR 13 

RLV 95 0 1 

ESD 96 93 POLY(1) 83 81 19 1 

MLV 95 96 93 93 SW 

LE 73 74 7.5N 

 

.MODEL SW NMOS (LEVEL=3 VTO=0 KP=5) 

.MODEL QOUT PNP (IS=377F NF=1.2 BF=5.1 CJE=3.48N 

+ TF=24.3N XTB=1.3) 

.MODEL MFIN NMOS (LEVEL=3 VMAX=400K THETA=36.1M ETA=2M 

+ VTO=5.2 KP=2.12) 

.MODEL DR D (IS=37.7F CJO=100P VJ=1 M=.82) 

.MODEL DO D (IS=37.7F BV=600 CJO=2.07N VJ=1 M=.7) 

.MODEL DE D (IS=37.7F BV=14.3 N=2) 

.MODEL DLIM D (IS=100N) 

.ENDS 

 

.END 

 

Na slici 5.6 prikazana je simulirana izlazna karakteristika komercijalnog IGBT-a. Za 

poreĎenje grafika iz datasheeta potrebno je modifikovati netlistu. Prvo, VDD treba da bude 

konstantnih 100V, dok u .DC analizi treba menjati napon VGC od 5V do 20V sa korakom 

0.5V. Korišćenjem naredbe .TEMP 25 150 biće simulirane karakteristike na zadatim 

temperaturama. Na slici 5.7 dat je grafik iz datasheeta proizvoĎača, dok slika 5.8 prikazuje 

simulirane karakteristike u log-lin razmeri. Drugo, treba postaviti fiksnu vrednost za 

VGS=15V, dok se VDD menja u .DC analizi od 1V do 10V sa korakom 0.5V. Na slici 5.9 

prikazan je grafik iz datasheeta proizvoĎača, dok slika 5.10 prikazuje simulirane 

karakteristike u log-log razmeri. 

Slika 5.6. Izlazne karakteristike komercijalnog IGBT-a 

 

Slika 5.7. Tipična prenosna karakteristika preuzeta iz datasheeta komponente 

 

Slika 5.8. Simulirana prenosna karakteristika komercijalnog IGBT-a na različitim 

temperaturama 

           V D D

0 V 2 V 4 V 6 V 8 V 1 0 V 1 2 V 1 4 V 1 6 V 1 8 V 2 0 V 2 2 V 2 4 V 2 6 V 2 8 V 3 0 V

- I ( V D D )

0 A

1 0 0 A

2 0 0 A

3 0 0 A

4 0 0 A

           V G C

5 V 6 V 7 V 8 V 9 V 1 0 V 1 1 V 1 2 V 1 3 V 1 4 V 1 5 V 1 6 V 1 7 V 1 8 V 1 9 V 2 0 V
- I ( V D D )

1 0 0 m A

1 . 0 A

1 0 A

1 0 0 A

1 . 0 K A

- I ( V D D )

Slika 5.9. Tipična izlazna karakteristika preuzeta iz datasheeta komponente 

 

Slika 5.10. Simulirana izlazna karakteristika komercijalnog IGBT-a na različitim 

temperaturama 

ZADATAK 37. Simulirati osnovno kolo IGBT-a u vremenskom domenu. Parametri 

impulsnog generatora su: logična nula: 0V, logička jedinica: 12V, kašnjenje: 1ms, vreme 

trajanja prednje/zadnje ivice: 100ns, period: 5ms. Vreme trajanja simulacije je 25ms. 

Analizirati isto kolo u .AC domenu u 5 tačaka po dekadi u frekventnom opsegu od 10Hz do 

100MHz.  

Rešenje: 

Treba modifikovati netlistu iz prethodnog primera. Menja se vrednost VDD=500V, tip 

generatora VGC: VGC 1 0 PULSE (0 12 1M 100N 100N 2.5M 5M), i zadajte .TRAN tip 

analize: .TRAN 50n 25m. Korišćenjem komande .PROBE prikazati izlaznu struju, ali i struju 

ulaznog kola. Rezultati simulacije prikazani su na slici 5.11. Ukoliko zumiramo prelazni 

režim može se uočiti da kroz gejt ipak protiče struja, odnosno dolazi do punjenja/pražnjenja 

kapacitivnosti, što dovodi do kašnjenja prilikom uključenja/isključenja IGBT-a (slika 5.12). 

           V D D

1 . 0 V 3 . 0 V 5 . 0 V 7 . 0 V 9 . 0 V 1 0 V

- I ( V D D )

1 . 0 A

1 0 A

1 0 0 A

1 . 0 K A

Za .AC analizu treba modifikovati netlistu dodavanjem DC/AC generatora DC amplitude 

5V, AC amplitude 1V i zadavanjem komande .AC DEC 5 10Hz 100Meg. Rezultati .AC 

analize prikazani su na slici 5.13, gde se može očitati granična frekvencija kola. 

 

Slika 5.11. Rezultati simulacije kola sa IGBT-om u vremenskom domenu 

 

Slika 5.12. Uvećani deo karakteristike pri isključenju IGBT-a 

 

Slika 5.13. Simulacija kola IGBT-a u AC režimu 

ZADATAK 38. Simulirati kolo konvertor impulsnog u sinusni oblik realizovanog sa 

IGBTom sa slike 5.14. Korišćen je kalem induktivnosti 800nH sa diodom 4148. Kolo se 

napaja iz impulsnog generatora čiji napon logičke jedinice iznosi 10V. Kolo se napaja sa 

250V. 

           T i m e

0 s 2 m s 4 m s 6 m s 8 m s 1 0 m s 1 2 m s 1 4 m s 1 6 m s 1 8 m s 2 0 m s 2 2 m s 2 4 m s 2 6 m s

-  I ( V D D )

- 4 0 0 A

0 A

4 0 0 A

I ( v g c )
- 2 0 0 m A

0 A

2 0 0 m A

S E L > >

           T i m e

8 . 4 9 9 5 0 0 m s 8 . 5 0 0 0 0 0 m s 8 . 5 0 0 5 0 0 m s 8 . 5 0 1 0 0 0 m s 8 . 5 0 1 5 0 0 m s 8 . 5 0 2 0 0 0 m s 8 . 5 0 2 4 4 6 m s
-  I ( V D D )

- 2 0 0 A

0 A

2 0 0 A

I ( v g c )

0 A

5 0 m A

8 9 m A

S E L > >

           F r e q u e n c y

1 0 H z 3 0 H z 1 0 0 H z 3 0 0 H z 1 . 0 K H z 3 . 0 K H z 1 0 K H z 3 0 K H z 1 0 0 K H z 3 0 0 K H z 1 . 0 M H z 3 . 0 M H z 1 0 M H z 3 0 M H z 1 0 0 M H z

I ( V D D )

0 A

4 A

8 A

1 2 A

1 6 A

Slika 5.14. Šema konvertora impulsnog u sinusni oblik signala pomoću IGBT-a 

Rešenje: 

Kolo sa slike 5.14 može se opisati sledećom netlistom: 

**IGBT- pulse -> sine 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

VCC 3 0 250 

VGC 1 0 PULSE (0 10V 1u 10ns 10ns 0.1u 0.2u) 

X1 4 1 0 IRGBC40U 

L1 4 3 800nH 

D1 4 3 D1N4148 

 

.model D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

.TRAN 50ns 10u uic 

.PROBE V(1,0) I(VCC) V(4,0) 

 

.SUBCKT IRGBC40U 71 72 74 

* TERMINALS: C G E 

* 600 Volt 40 Amp 6.04NS N-Channel IGBT  

Q1 83 81 85 QOUT 

M1 81 82 83 83 MFIN L=1U W=1U 

DSD 83 81 DO 

DBE 85 81 DE 

RC 85 71 21.1M 

RE 83 73 2.11M 

RG 72 82 25.6 

CGE 82 83 1.42N 

CGC 82 71 1P 

EGD 91 0 82 81 1 

VFB 93 0 0 

FFB 82 81 VFB 1 

CGD 92 93 1.41N 

R1 92 0 1 

D1 91 92 DLIM 

DHV 94 93 DR 

R2 91 94 1 

D2 94 0 DLIM 

DLV 94 95 DR 13 

RLV 95 0 1 

ESD 96 93 POLY(1) 83 81 19 1 

MLV 95 96 93 93 SW 

LE 73 74 7.5N 

 

.MODEL SW NMOS (LEVEL=3 VTO=0 KP=5) 

.MODEL QOUT PNP (IS=377F NF=1.2 BF=5.1 CJE=3.48N 

+ TF=24.3N XTB=1.3) 

.MODEL MFIN NMOS (LEVEL=3 VMAX=400K THETA=36.1M ETA=2M 

+ VTO=5.2 KP=2.12) 

.MODEL DR D (IS=37.7F CJO=100P VJ=1 M=.82) 

.MODEL DO D (IS=37.7F BV=600 CJO=2.07N VJ=1 M=.7) 

.MODEL DE D (IS=37.7F BV=14.3 N=2) 

.MODEL DLIM D (IS=100N) 

.ENDS 

 

.END 

 

Rezultat simulacije prikazan je na slici 5.15. 

 

Slika 5.15. Rezultat simulacije kola za konverziju impulsnog u sinusni oblik signala sa 

IGBT-om 

 

 

 

 

 

 

 

 

           T i m e

0 s 1 u s 2 u s 3 u s 4 u s 5 u s 6 u s 7 u s 8 u s 9 u s 1 0 u s

V ( 4 , 0 )

0 V

2 0 0 V

4 0 0 V
V ( 1 , 0 )

0 V

5 V

1 0 V

S E L > >

6. 
MODELIRANJE I SIMULACIJA 
OPERACIONIH POJAC AVAC A 

Operacioni počajavač je kolo koje pojačava razliku signala na njegovim ulazima A puta i 

prosleĎuje je na izlaz kola. Ulazi operacionih pojačavača su invertujući (Vin-) i neinvertujući 

(Vin+). Ovo je najprimenjivanije kolo u analognoj mikroelektronici, naziv je dobilo po tome 

što je osmišljeno da izvodi osnovne matematičke operacije nad signalima (sabiranje, 

oduzimanje, množenje, ...). Osnovne karakteristike idealnog operacionog pojčavača su: 

ulazna impedansa je beskonačna (struje kroz ulazne priključke su jednake nuli), izlazna 

impedansa je jednaka nuli (izlaz predstavlja idealni naponski izvor), ne reaguje za signal koji 

je zajednički za oba ulaz, beskonačno pojačanje u otvorenoj petlji i beskonačni propusni 

opseg. Kod realnih operacionih pojačavača sve ovo ne važi, ali su u aproksimaciji mogu 

koristiti osobine idealnih operacionih pojačavača. Komercijalni operacioni pojačavači su 

najčešće opisani kao potkolo za simulatore. 

ZADATAK 39. Formirati potkolo operacionog pojačavača prikazano na slici 6.1 i simulirati 

rad kola u .AC režimu ako se naizmenični signal amplitude 1V superponira na PWL signal 

(0u,0V) (0.01u, 1V). 

 

Slika 6.1. Šema idealnog operacionog pojačavača 

Rešenje: 

Otpornik RIN predstavlja ulaznu otpornost operacionog pojačavača, čija je vrednost idealno 

beskonačna (1E12, 1E15), dok kod realnih pojčavača ima konačnu vrednost tako da kroz 

ulazne terminale protiče vrlo mala struja. Srednji blok sastoji se od naponom kontrolisanog 

naponskog generatora i filtarskog kola. Naime, generator V1 kontroliše se ulaznim naponom 

VIN, sa odreĎenim faktorom pojačanja A. RC filtar odreĎuje graničnu frekvenciju 

operacionog pojačavača, sprečavajući ga da zaosciluje na visokim frekvencijama. Izlazni 

stepen kola čini bafer i izlazna otpornost operacionog pojačavača. Ulogu bafera igra 

generator VOUT koji uzima vrednost sa izlaza srednjeg stepena sa faktorom pojačanja 1 i 

time se sprečava uticaj priključenog opterećenja na rad operacionog pojačavača. Izlazna 

otpornost operacionog pojačavača teži nuli, ali kod realnih pojačavača ima konačnu malu 

vrednost, tako da deo napona ipak osaje unutar operacionog pojačavača. Netlista za kolo sa 

slike 6.1 data je u nastavku: 

**Idealni OPA 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

.SUBCKT OPAMP 1 2 6 

* 1= Vin+, 2=Vin-, 6=Vout 

RIN 1 2 10MEG 

E1 3 0 1 2 100K 

R1 3 4 1K 

C1 4 0 1.5915U 

EOUT 5 0 4 0 1 

ROUT 5 6 10 

.ENDS 

 

VIN 1 0 AC 1 PWL (0u 0V 0.01u 1V) 

XOP 1 0 3 OPAMP 

RL 3 0 1K 

 

.AC DEC 5 1 100MEG 

.PROBE V(3,0) 

.END 

 

Pokretanjem simulacije prikazaćemo u jednom prozoru vrednost 20*LOG10(V(3,0)), tj. 

pojačanje operacionog pojačanje u decibelima, a u drugom vrednost P(V(3,0)), tj. fazu 

izlaznog signala. Rezultati simulacije prikazani su na slici 6.2. Možemo zaključiti da 

pojačanje pada na 0 dB na frekvenciji 10 MHz, ali analizirajući fazni dijagram zaključujemo 

da ovakav operacioni pojačavač okreće fazu već na 5 kHz. 

Slika 6.2. Rezultati simulacije idealnog operacionog pojačavača 

ZADATAK 40. Simulirati rad kola sa realnim operacionim pojačavačem prikaznom na slici 

6.3 u vremenskom domenu. Parametarski zadati vrednosti: .PARAM VDD=2.5 RF=1 

CL=60p RL=10k. Pobudni generator treba da bude imuplsni, parametri signala su: logička 

nula: -2.25V, logička jedinica: 2.25V, kašnjenje: 10μs, vreme trajanja prednje/zadnje ivice: 

1ps, period: 100μs. Kolo simuliarati i u .AC režimu, dodavanjem AC generatora. Analizu 

izvršiti u 5 tačaka po dekadi u frekventnom opsegu od 100mHz do 10MHz.  

 

Slika 6.3. Šema kola sa operacionim pojačavačem 

Rešenje: 

Kolo sa slike 6.3 može se opisati netlistom u nastavku: 

*OPA 

.option numdgt=8 stepgmin gmin=1e-12 reltol=0.01   nomod 

.PARAM VDD=2.5 RF=1 CL=60p RL=10k 

R_RG 0 2 10k 

R_RL 0 OUT {RL} 

C_CL 0 OUT {CL} 

V_VDD 3 0 {VDD} 

V_VSS 0 4 {VDD} 

V_VIN IN 0 PULSE (-2.25 2.25 10u 1p 1p 50u 100u) 

           F r e q u e n c y

1 . 0 H z 1 0 H z 1 0 0 H z 1 . 0 K H z 1 0 K H z 1 0 0 K H z 1 . 0 M H z 1 0 M H z 1 0 0 M H z

2 0 *  L O G 1 0 ( V ( 3 , 0 ) )

- 1 0 0

0

1 0 0

S E L > >

P ( V ( 3 , 0 ) )

- 1 0 0 d

- 5 0 d

0 d

R_RP IN 1 10k 

R_RF 2 5 {RF} 

R_RZ 5 OUT 1m 

XOPA 1 2 3 4 5 MCP6241 

 

 

.TRAN 0 100us  

.PROBE V([in],0), V([out],0) 

 

 

.SUBCKT MCP6241 1 2 3 4 5 

*               | | | | | 

*               | | | | Output 

*               | | | Negative Supply 

*               | | Positive Supply 

*               | Inverting Input 

*               Non-inverting Input 

 

* Input Stage 

V10 3 10 -500M 

R10 10 11 6.90K 

R11 10 12 6.90K 

C11 11 12 7.20P 

C12 1 0 6.00P 

E12 71 14 POLY(4) 20 0 21 0 26 0 27 0 5.00M 34.9 34.9 1 1 

G12 1 0 62 0 1m 

M12 11 14 15 15 NMI 

M14 12 2 15 15 NMI 

G14 2 0 62 0 1m 

C14 2 0 6.00P 

I15 15 4 50.0U 

V16 16 4 -300M 

GD16 16 1 TABLE {V(16,1)} ((-100,-1p)(0,0)(1m,1u)(2m,1m)) 

V13 3 13 -300M 

GD13 2 13 TABLE {V(2,13)} ((-100,-1p)(0,0)(1m,1u)(2m,1m)) 

R71 1 0 20.0E12 

R72 2 0 20.0E12 

R73 1 2 20.0E12 

I80 1 2 500E-15 

* 

* Noise, PSRR, and CMRR 

I20 21 20 423U 

D20 20 0 DN1 

D21 0 21 DN1 

G26 0 26 POLY(2) 3 0 4 0 0.00 -158U -3U 

R26 26 0 1 

G27 0 27 POLY(2) 1 0 2 0 -776U 35.5U 35.5U 

R27 27 0 1 

* 

* Open Loop Gain, Slew Rate 

G30 0 30 12 11 1 

R30 30 0 1.00K 

C30 30 0 10p 

G31 0 31 3 4 2 

I31 0 31 DC 65 

R31 31 0 1 TC=3.67M,5.32U 

GD31 30 0 TABLE {V(30,31)} ((-100,-1u)(0,0)(1m,.1)(2m,2)) 



 
99 Modeliranje i simulacija mikroelektronskih komponenata i kola 

G32 32 0 3 4 -1.9 

I32 32 0 DC 105 

R32 32 0 1 TC=3.43M,4.42U 

GD32 0 30 TABLE {V(30,32)} ((-2m,2)(-1m,.1)(0,0)(100,1u)) 

G33 0 33 30 0 1m 

R33 33 0 1K 

G34 0 34 33 0 316M 

R34 34 0 1K 

C34 34 0 81.8U 

G37 0 37 34 0 1m 

R37 37 0 1K 

C37 37 0 22.7P 

G38 0 38 37 0 1m 

R38 39 0 1K 

L38 38 39 26.5U 

E38 35 0 38 0 1 

G35 33 0 TABLE {V(35,3)} ((-1,-1n)(0,0)(48,1n))(49,1)) 

G36 33 0 TABLE {V(35,4)} ((-49,-1)((-48,-1n)(0,0)(1,1n)) 

* 

* Output Stage 

R80 50 0 100MEG 

G50 0 50 57 96 2 

R58 57 96 0.50 

R57 57 0 1650 

C58 5 0 2.00P 

G57 0 57 POLY(3) 3 0 4 0 35 0 0 0.21M 0.21M 0.6M 

GD55 55 57 TABLE {V(55,57)} ((-2m,-1)(-1m,-1m)(0,0)(10,1n)) 

GD56 57 56 TABLE {V(57,56)} ((-2m,-1)(-1m,-1m)(0,0)(10,1n)) 

E55 55 0 POLY(2) 3 0 51 0 -0.85M 1 -51.0M 

E56 56 0 POLY(2) 4 0 52 0 1.33M 1 -42.0M 

R51 51 0 1k 

R52 52 0 1k 

 

GD51 50 51 TABLE {V(50,51)} ((-10,-1n)(0,0)(1m,1m)(2m,1)) 

GD52 50 52 TABLE {V(50,52)} ((-2m,-1)(-1m,-1m)(0,0)(10,1n)) 

G53 3 0 POLY(1) 51 0 -50.0U 1M 

G54 0 4 POLY(1) 52 0 -50.0U -1M 

* 

* Current Limit 

G99 96 5 99 0 1 

R98 0 98 1 TC=-1.92M,-7.58U 

G97 0 98 TABLE { V(96,5) } ((-11.0,-21.0M)(-1.00M,-

20.7M)(0,0)(1.00M,20.7M)(11.0,21.0M)) 

E97 99 0 VALUE { V(98)*((V(3)-V(4))*166M + 416M)} 

D98 4 5 DESD 

D99 5 3 DESD 

* 

* Temperature / Voltage Sensitive IQuiscent 

R61 0 61 1 TC=3.14M,7.28U 

G61 3 4 61 0 1 

G60 0 61 TABLE {V(3, 4)} 

+ ((0,0)(750M,450N)(800M,1.00U)(900M,4.00U) 

+ (1.2,41.0U)(1.4,45.0U)(5.5,46.0U)) 

* 

* Temperature Sensistive offset voltage 

I73 0 70 DC 1uA 

R74 0 70 1 TC=3.00U 

E75 1 71 70 0 1 

* 

* Temp Sensistive IBias 

I62 0 62 DC 1uA 

R62 0 62 REXP 55.78U 

* 

* Models 

.MODEL NMI NMOS(L=2.00U W=42.0U KP=20.0U LEVEL=1 ) 

.MODEL DESD D N=1 IS=1.00E-15 

.MODEL DN1 D IS=1P KF=146E-18 AF=1 

.MODEL REXP RES TCE=10.14 

.ENDS MCP6241 

.END 

 

Rezultati simulacije u vremenskom domenu prikazani  su na slici 6.4, a u AC domenu na 

slici 6.5. Može se zaključiti da kod realnih pojačavača postoji kašnjenje u odzivu kola, dok 

otpornici odreĎuju odnos izlaznog i ulaznog napona. Na osnovu AC analize možemo 

zaključiti da je granična frekvencija kola 600kHz. 

 

Slika 6.4. Rezultati simulacije kola sa OPA u vremenskom domenu 

 
Slika 6.5. Rezultati simulacije kola sa OPA u AC domenu 

 

 

           T i m e

0 s 1 0 u s 2 0 u s 3 0 u s 4 0 u s 5 0 u s 6 0 u s 7 0 u s 8 0 u s 9 0 u s 1 0 0 u s

V ( o u t , 0 ) V ( i n , 0 )

- 4 . 0 V

- 2 . 0 V

0 V

2 . 0 V

4 . 0 V

           F r e q u e n c y

1 0 0 m H z 1 . 0 H z 1 0 H z 1 0 0 H z 1 . 0 K H z 1 0 K H z 1 0 0 K H z 1 . 0 M H z 1 0 M H z

D B ( V ( o u t , 0 ) / V ( i n , 0 ) )

- 8 0

- 4 0

0

4 0

7. 
RES ENI ZADACI 

ZADATAK 41. Analizirati jednosmerni režim rada (DC) kola razdelnika napona sa Slike 

118. 

a) R2 simulirati kao promenljivi otpornik, parametarskim zadavanjem vrednosti od 

1 kΩ do 5 kΩ sa korakom 1 kΩ. 

b) R2 simulirati kao promenljivi otpornik, definisanjem modela otpornika i menjanjem 

parametra modela od 1 kΩ do 5 kΩ sa korakom 1 kΩ. 

 

Slika 118. Rezdelnik napona 

Rešenje: 

a) Na osnovu označenih čvorova, formira se PSpice netlista za kolo sa slike 118: 

Razdelnik napona – simulacija potenciometra 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16  

.PARAM POT = 1K 

V1 1 0 1 

R1 1 2 10K 

R2 2 0 {POT} 

.DC V1 0 10 1 

.STEP PARAM POT 1K 5K 1K 

.PROBE V(2) 

.END 

 

Jedan od načina za simulaciju promenljivog otpornika, tj. za promenu otpornosti u kolu o 

odreĎenom opsegu sa odreĎenim korakom je korišćenjem globalnog parametra. Komandom 

.PARAM definišu se globalni parametri simulacije. Kao globalni parametar definišu se 

konstante koje se pri opisivanju kola ponavljaju više puta u izrazima. Na primer: .PARAM 

pi=3.14. Druga primena globalnih parametara odnosi se na zadavanje parametarske 

analize komandom .STEP, što se razmatra u ovom primeru. Definisali smo globalni 

parametar POT i inicijalizovali njegovu vrednost na 1 k.  

Vrednost otpornika R2 je zadata parametarski, tako da se umesto konkretne vrednosti, piše 

naziv globalnog parametra u vitličastim zagradama {POT}.  

Vrednost generatora V1 menjaće se od 0 V do 10 V sa korakom 1 V, što je definisano 

komandom .DC.  

Komandom .STEP ponavljaju se sve navedene analize za zadati skup vrednosti promenljivog 

parametra. U ovom primeru, ponavljaće se .DC analiza za parametar POT i to od 1 k do 

5 k sa korakom 1 k. PARAM je ključna reč koja se koristi u okviru .STEP komande kada se 

poziva globalni parametar (u ovom slučaju POT). Zatim se navode, početna vrednost i 

krajnja vrednost promenljivog parametra. Vrednost parametra se u svakom prolazu uvećava 

za korak koji je definisan na kraju.  

Primer komande .STEP promenu globalnih parametara  – detaljno u [PSpcRef] 

.STEP PARAM POT 1K 5K 1K 

*         |     |      |  |  |korak 

*         |     |      |  | krajnja vrednost 

*         |     |      | početna vrednost 

*         |     | naziv globalnog parametra 

*         |ključna reč za stepovanje globalnih parametara 

 

Treba napomenuti, postoje i drugi oblici zadavanja komande .STEP kada se kao menja 

(stepuje) vrednost nezavisnog pobudnog generatora (naponskog V ili strujonog I), parametri 

modela, temperatura, što će biti obraĎeno tokom kursa. Ovde je opisan samo način promene 

(stepovanja) globalnog parametra, korišćenjem komande .PARAM. Postoje i različiti načini 

promene vrednosti parametara: linearno sa korakom, što je opisano, ali i logaritamski, gde se 

analiza obavlja u odreĎenom broju tačaka po oktavi ili dekadi, kao i po listi, kada se analiza 

ponavlja za sve vrednosti parametra nevedene u listi. Sintaksa za ove tipove .STEP analize 

biće obraĎena tokom kursa. 

TakoĎe napomenimo, globalni parametar može figurisati u izrazima, a može se definisati 

korišćenjem izraza. Izraz se upisuje izmeĎu vitličastih zagrada. U izrazima se mogu javiti 

globalni parametri, realni brojevi, operacije i ugraĎene standardne funkcije. Izraz mora biti 

definisan u jednoj liniji, ili je, u suprotnom, potrebno koristiti komandu .FUNC. Pisanje 

izraza i korišćenje komande .FUNC biće objašnjeno tokom kursa. 

Pokrenuti simulaciju i analizirati rezultate. Rezultati svih analiza biće smešteni u isti izlazni 

fajl, jedan za drugim. Program PROBE omogućava prikazi svih rezultata, tj. štampa familije 

krivih. Nakon završene simulacije, pre prikaza rezultata dobija se obaveštenje o svim 

vrednostima stepovanih parametara. Karakteristika za prvu vrednost stepovanog parametra 

označena je prvim u nizu simbola prikazani u donjem desnom uglu prozora za prikaz rezltata 



 
103 Modeliranje i simulacija mikroelektronskih komponenata i kola 

simulacije (najčešće zeleni kvadratić), druga vrednost parametra odgovara drugom simbolu i 

tako redom (Slika 119). 

 

Slika 119. Rezultati simulacije (zelena linija je ta R2=1 kΩ, crvena za R2=2 kΩ, i tako 

redom) 

b) Netlista za kolo sa Slike 16, kada se za otpornik R2 definiše model i vrši stepovanje 

parametra modela, kako bi se simulirao promenljivi otpornik je: 

Razdelnik napona – simulacija potenciometra 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16  

V1 1 0 1 

R1 1 2 10K 

R2 2 0 RMODEL 1k 

.MODEL RMODEL RES (R=1k) 

.DC V1 0 10 1 

.STEP RES RMODEL (R) 1K 5K 1K 

.PROBE V(2) 

.END 

 

Otpornik R2 izmeĎu čvorova 2 i 0 i za njega će biti definisan model pod nazivom RMODEL. 

Vrednost otpornosti ovog opotrnika je 1 kΩ, što je samo inicijalna vrednost, kako simulator 

ne bi javio grešku. Vrednost otpornika biće stepovana tako da se simulira potenciometar. 

Komanda .MODEL koristi se za definisanje kartice modela. Naime, do sada smo radili sa 

elementima koji se opisuju samo jednim parametrom (otpornost, kapacitivnost, 

indukitvnost), tako da nisu definisane kartice modela. Vrednosti parametra matematičkog 

modela elemenata u kolu definisane su navoĎenjem brojne vrednosti pri definisanju 

komponente. Na primer, jedini parametar otpornika R1 ovog kola je njegova vrednost 

otpornosti i iznosi 10 kΩ, kao što je navedeno u netlisti. U slučaju da model elementa kola 

sadrži veći broj parametara vrednosti parametara se zadaju u okviru kartice modela. Nakon 

ključne reči .MODEL navodi se naziv modela, tip komponente i u zagradi parametri i 

njihove vrednosti. U ovom primeru definisan je model otpornika, pod nazivom RMODEL, 

čiji je jedini parametar otpornost R, čija je vrednost R=1 kΩ. Ovakav način definisanja 

modela ima smisla kada bi bili definisani i ostali parametri otpornika kao što je linearni 

temperaturni koeficijent i sl. Ovde je definisan samo bitan parametar, kako bi demonstrirali 

korišćenje stepovanje parametara modela. U nastavku je dat primer kartice modela diode, da 

           V 1

0 V 1 V 2 V 3 V 4 V 5 V 6 V 7 V 8 V 9 V 1 0 V

V ( 2 )

0 V

1 . 0 V

2 . 0 V

3 . 0 V

4 . 0 V



 
104  

bi bila ilustrovana sintaksa .MODEL. Model diode biće detaljno razmatran tokom kursa. 

Ovde se vide mnogobrojni parametri ovog modela diode. Treba napomenuti, otpornici imaju 

kartice modela tipa RES, kondenzatori – CAP, kalemovi – IND, diode – D, bipolarni 

tranzistori – NPN, PNP... Tipovi kartica će, takoĎe, biti razmatrani tokom kursa. 

Primer definisanja kartice modela .MODEL  – detaljno u [PSpcRef] 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 

IBV=0.1PA) 

*           |       |      |  parametri modela 

*           |       | tip komponente 

*           | naziv modela 

Vrednosti parametara modela pridružuju se opisu elemenata navoĎenjem imena 

odgovarajuće kartice modela, kao što je u ovom primeru R2 2 0 RMODEL 1k. Ista kartica 

modela služi za opis više elemenata koji imaju evivalentni matematički model. Dakle, model 

se jednom navodi i sve komponente koje imaju isti model samo pozivaju definisani model. 

Detalji o modelima, pogotovu složenih komponenti kao što su diode i tranzistori biće 

razmatrani tokom kursa.  

Nakon definisanja DC analize, potrebno je definisati stepovanje parametra modela otpornika 

– R . Iza ključne reči .STEP navodi se tip komponente, naziv modela, nakon toga u zagradi 

naziv parametra koji će biti stepovan i na kraju početna, krajnja vrednost i korak sa kojim će 

se odreĎeni parametar modela menjati. U ovom primeru se za otpornik, čiji se model naziva 

RMODEL, stepuje parametar R, od 1 kΩ do 5 kΩ sa korakom 1 kΩ. 

Primer komande .STEP promenu parametara modela  – detaljno u [PSpcRef] 

.STEP RES RMODEL (R) 1K 5K 1K 

*         |   |             |  |  |  |  korak 

*         |   |             |  |  | krajnja vrednost 

*         |   |             |  | početna vrednost 

*         |   |             | naziv parametra koji se stepuje 

*         |   | naziv modela čiji se parametar stepuje 

*         | tip komponente 

Napomenimo, mogu se stepovati svi parametri modela osim L i W za MOSFET i 

temperaturnih koeficijenata TC1 i TC2 za komponente kod kojih ovi koeficijenti mogu da se 

definišu, što će biti diskutovano tokom kursa. 

Pokrenuti simulaciju i analizirati rezultate. Dobija se isti rezultat kao i u pirmeru pod a) ovog 

zadatka. Može se uočiti da sa povećanjem napona V1 raste izlazni napon, kao i da je pri 

istom naponu V1, veći izlazni napon što veća vrednost R2 (Slika 17). 

ZADATAK 42. Izvršiti Furijeovu analizu kola sa slike. 

 
Slika 120. Električno kolo za simulaciju 

Rešenje: 
 

Zadatak 5. LCC oscilatorno kolo 

* ilustracija definisanja generatora sinusnog napona 

* i ilustracija MODEL kartice 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16  

V1 1 0 SIN (0 1 50K) 

L1 1 2 LMODEL 1M 

C1 2 3 10N 

C2 3 0 10N 

R1 3 0 100 

.MODEL LMODEL IND (L=1 DEV 1%) 

.TRAN 100P 50U UIC 

.FOUR 50K V(1) 

.PROBE V(3) I(V1) 

.END 

 

ZADATAK 43. Na slici 121 je prikazano kolo za obradu signala sa senzora pritiska. 

Simulirati rad kola u programu PSpice u vremenskom domenu u trajanju od 10 ms, rezultati 

se štampaju na 0.1 ms. Otpornost senzora pritiska XR2 menja se po zakonu 1K*P. XR2 

predstaviti naponom kontrolisanim strujnim izvorom, dok je promena pritiska P definisana 

pomoćnim generatorom VC1 čija se vrednost menja linearno od 0.9 do 1.1 u vremenu od 

0 ms do 10 ms. Za definisanje signala generatora VC1 koristiti PWL. Prikazati vrednosti 

VC1 i izlaznog napona  na otporniku Riz u funkciji vremena. Operacioni pojačavač opisan je 

datim potkolom: 

* SINGLE-POLE OPERATIONAL AMPLIFIER MACRO-MODEL 

* connections:       non-inverting input 

*                    |   inverting input 

*                    |   |   output 

*                    |   |   | 

.SUBCKT OPAMP1       1   2   6 

* INPUT IMPEDANCE 

RIN 1 2 10MEG 

* DC GAIN (100K) AND POLE 1 (100HZ) 

EP1 3 0 1 2 100000 

RP1 3 4 1K 

CP1 4 0 1.5915UF 

* OUTPUT BUFFER AND RESISTANCE 

EOUT 5 0 4 0 1 

ROUT 5 6 10 

.ENDS 

 

Slika 121. Kolo za obradu podataka sa senzora pritiska 

Rešenje: 

ZADATAK 43 

* 

* SENSOR BRIDGE 

RS1 4 11 1 

R1 11 12 1K 

XR2 12 14 20 0 VCR  

R3 11 13 1K 

R4 13 14 1K 

RS4 14 2 1 

* 

* MEASUREMENT WIRES AND RIN OF AMP 

RS2 12 15 1 

RAMP 15 16 100MEG 

RS3 13 16 1 

* 

* OPAMP CURRENT SERVO 

VREF 3 0 DC 1V 

RSENSE 2 0 100 

XOP1 3 2 4 OPAMP1  

* 

* 

* VOLTAGE CONTROL FOR VCR - INCREASING RESISTANCE CHANGE 

VC1 20  0 PWL(0MS 0.9  10MS 1.1) 

RD1 20  0 1G 

* 

* 

* SUBCIRCUIT FOR VOLTAGE CONTROLLED RESISTOR (VCR) 

R 3

1 K

R 1

1 K

RS2

1

RS3

1

RS1

1

RS4

1

Riz

1 0 0 M

RSENSE

1 0 0

R D 1

1 G

R 4

1 K

V R 2

OPAMP1

V R E F

1 V

V C 1

.SUBCKT VCR   1  2  4  5 

Gres  1 2 VALUE = { v(1,2)/1K/V(4,5) } 

.ENDS 

 

* SINGLE-POLE OPERATIONAL AMPLIFIER MACRO-MODEL 

* connections:       non-inverting input 

*                    |   inverting input 

*                    |   |   output 

*                    |   |   | 

.SUBCKT OPAMP1       1   2   6 

* INPUT IMPEDANCE 

RIN 1 2 10MEG 

* DC GAIN (100K) AND POLE 1 (100HZ) 

EP1 3 0 1 2 100000 

RP1 3 4 1K 

CP1 4 0 1.5915UF 

* OUTPUT BUFFER AND RESISTANCE 

EOUT 5 0 4 0 1 

ROUT 5 6 10 

.ENDS 

*  

* 

* ANALYSIS 

.TRAN  0.1MS  10MS 

* 

.PROBE v(15,16) v(20,0) 

.END 

 

ZADATAK 44. Na slici 122 je prikazano kolo step-up konvertora, osnovna konfiguracija 

prekidačnog izvora napajanja. Simulirati rad kola u programu PSpice u vremenskom domenu 

u trajanju od 5s. Odrediti: 

a) Vrednost napona V2 za koju će kolo  na opterećenju RL dati vrednost izlaznog 

napona  Viz =11.5V   u slučaju  korišćenja  BJT i NMOSFET-a. 

b) Struju kroz potrošač RL. 

 

Slika 122. Step-Up konvertor 

3 V

V 1

1mH

L 1

D 1

3 3 0 u F

C 1

5 K

R L

BJT ili FET

V 2

SINE, 1kHz

B, G

C, D

E, S



 
108  

Dati su modeli: 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0      

+ IKR=0 RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P      

+ MJE=.377 VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3     

+ RB=10) 

 

.MODEL FETN NMOS LEVEL=3 W=3000u L=1u 

+ PHI=0.600000 TOX=2.1500E-08 XJ=0.200000U TPG=1 VTO=0.8063 

+ DELTA=9.4090E-01 LD=1.3540E-07 KP=1.0877E-04 UO=680.4 

+ THETA=8.3620E-02  GAMMA=0.5487 NSUB=2.3180E+16 NFS=1.98E+12 

+ VMAX=1.8700E+05 ETA=5.5740E-02 KAPPA=5.9210E-02                    

+ CGDO=3.2469E-10 CGSO=3.2469E-10 CGBO=3.7124E-10 CJ=3.1786E-04      

+ MJ=1.0148 CJSW=1.3284E-10 MJSW=0.119521 PB=0.800000 

 

Rešenje: 

*** 

.options numdgt=8 reltol=0.01  

V1 1 0 3 

L1 1 2 1M 

D1 2 3 D1N4148 

C1 3 0 330U 

R1 3 5 50k 

*Q1 2 4 0 0 Q2N2222A 1 

M1 2 4 0 0 CMOSN W=3000u L=1u 

*V2 4 0 SIN (0 0.91 1KHz) 

V2 4 0 SIN (0 5 1KHz) 

V3 5 0 0 

.MODEL D1N4148 D (IS=0.1PA, RS=16 CJO=2PF TT=12N BV=100 IBV=0.1PA) 

.model Q2N2222A NPN (IS=14.34F  XTI=3  EG=1.11  VAF= 74.03  BF=255.9   

+NE=1.307  ISE=14.34F  IKF=.2847  XTB=1.5  BR=6.092  NC=2  ISC=0  

IKR=0   

+RC=1  CJC=7.306P  MJC=.3416  VJC=.75  FC=.5  CJE=22.01P  MJE=.377   

+VJE=.75  TR=46.91N  TF=411.1P  ITF=.6  VTF=1.7  XTF=3  RB=10) 

.MODEL CMOSN NMOS LEVEL=3 

+ PHI=0.600000 TOX=2.1500E-08 XJ=0.200000U TPG=1 VTO=0.8063 

+ DELTA=9.4090E-01 LD=1.3540E-07 KP=1.0877E-04 UO=680.4 

+ THETA=8.3620E-02  GAMMA=0.5487 NSUB=2.3180E+16 NFS=1.98E+12 

+ VMAX=1.8700E+05 ETA=5.5740E-02 KAPPA=5.9210E-02 CGDO=3.2469E-10 

+ CGSO=3.2469E-10 CGBO=3.7124E-10 CJ=3.1786E-04 MJ=1.0148 

+ CJSW=1.3284E-10 MJSW=0.119521 PB=0.800000  

.TRAN 50N 5  

.PROBE V(4,0) V(3,0) I(V3) 

.END 

 

ZADATAK 45. Na slici 123 je prikazano NAND logičko kolo u CMOS tehnologiji. 

Simulirati rad kola u programu PSpice u vremenskom domenu u trajanju od 1ms. Dužina 

kanala svih tranzistora je 1μm, dok je širina kanala PMOS tranzistora 10μm, a NMOS 

tranzistora 5μm. Poznato je: VDD=5V, VA=VDD. Generator VB je impulsni sledećih 

karakteristika (logička nula 0V, logička jedinica 5V, vreme porasta/opadanja signala 100ns, 

perioda 0.2ms, duty cycle 50%). Odrediti: 

a) Širinu kanala svih tranzistora tako da napon logičke jedinice na izlazu (napon na 

kondenzatoru C1) dostigne 5V. 

b) Struju drejna tranzistora M4. 

 
Slika 123. NAND logičko kolo u CMOS tehnologiji 

Dati su modeli: 

 
.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

Rešenje: 

*** 

.options numdgt=8 reltol=0.01  

 

M 4

NMOS

M 3

NMOS

M 1

PMOS

M 2

PMOS

V D D

G N D

V A

V B

1 0 0 n F

C 1

G N D

VB 2 0 PULSE(0 5 0 100N 100N 0.1M 0.2M) 

VDD 3 0 5 

M1 4 3 3 3 CMOSP L=1E-6 W=100E-6 

M2 4 2 3 3 CMOSP L=1E-6 W=100E-6 

M3 4 3 5 0 CMOSN L=1E-6 W=50E-6 

M4 6 2 0 0 CMOSN L=1E-6 W=50E-6 

C1 4 0 100n 

Vdummy 5 6 0 

 

.MODEL CMOSN NMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U    

+TPG=1 VTO=0.7860 DELTA=6.9670E-01 LD=1.6470E-07 KP=9.6379E-05 

+UO=591.7 THETA=8.1220E-02 RSH=8.5450E+01 GAMMA=0.5863 

+NSUB=2.7470E+16 NFS=1.98E+12 VMAX=1.7330E+05 ETA=4.3680E-02 

+KAPPA=1.3960E-01 CGDO=4.0241E-10 CGSO=4.0241E-10 

+CGBO=3.6144E-10 CJ=3.8541E-04 MJ=1.1854 CJSW=1.3940E-10 

+MJSW=0.125195 PB=0.800000 

 

.MODEL CMOSP PMOS LEVEL=3 PHI=0.600000 TOX=2.1200E-08 XJ=0.200000U  

+TPG=-1 VTO=-0.9056 DELTA=1.5200E+00 LD=2.2000E-08 KP=2.9352E-05 

+UO=180.2 THETA=1.2480E-01 RSH=1.0470E+02 GAMMA=0.4863 

+NSUB=1.8900E+16 NFS=3.46E+12 VMAX=3.7320E+05 ETA=1.6410E-01 

+KAPPA=9.6940E+00 CGDO=5.3752E-11 CGSO=5.3752E-11 

+CGBO=3.3650E-10 CJ=4.8447E-04 MJ=0.5027 CJSW=1.6457E-10 

+MJSW=0.217168 PB=0.850000 

 

.TRAN 50N 1M uic 

.PROBE V(2,0) V(4,0) I(Vdummy) 

.END 

 

ZADATAK 46. U programu PSpice napraviti netlistu za kolo instrumentacionog pojačavača 

sa slike 124. Operacioni pojačavač opisan je potkolom (u prilogu) i napaja se VDD=+10 V i 

VSS=-10 V.  

a) Uraditi frekventnu analizu instrumentacionog pojačavača u frekventnom opsegu do 

100 MHz. Amplituda naizmeničnog signala je vin1=50 mV, dok je vin2=0V. Prikazati 

amplitudsku karakteristiku, korišćenjem izraza 20·log10(Vout/Vin) i faznu 

karakteristiku, korišćenjem funkcije P(Vout/Vin). 

b) Uraditi vremensku analizu tako da bude prikazano 5 perioda signala. Na ulaz se 

dovodi signal sinusnog oblika, amplitude Vin1=1 V, frekvencije 1 kHz, dok je 

Vin2=0V. Odrediti vrednost pojačanja instrumentacionog pojačavača. 

Izvodi operacionog pojačavača: 
*                   NON-INVERTING INPUT 

*                   |   INVERTING INPUT 

*                   |   |  +VE SUPPLY VOLTAGE 

*                   |   |  |   -VE SUPPLY VOLTAGE 

*                   |   |  |   |   OUTPUT 

*                   |   |  |   |   |  

.SUBCKT LM224    1   2  11  12  24 

Slika 124. Kolo instrumentacionog pojačavača 

 

Rešenje: 

 
*** 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

VDD 10 0 10 

VSS 11 0 -10 

X1 1 2 10 11 5 LM224 

X2 4 3 10 11 6 LM224 

X3 7 8 10 11 9 LM224 

R1 2 3 10K 

R2 5 2 10K 

R3 3 6 10K 

R4 5 8 10K 

R5 6 7 10K 

R6 8 9 20K 

R7 7 0 20K 

VIN1 1 0 SIN(0 1 1K) 

VIN2 4 0 SIN(0 2 1K) 

 

.TRAN 100N 5M UIC 

.PROBE V(1,0) V(4,0) V(9,0)  

 

* Connections: 

*                   Non-Inverting Input 

*                   |   Inverting Input 

*                   |   |  +ve Supply Voltage 

*                   |   |  |   -ve Supply Voltage 

1 0 K

V D D

V D D

V D D

VSS

VSS

VSS

1 0 K

1 0 K

1 0 K

2 0 K

2 0 K

1 0 K

Vout

Vin1

Vin2

+10 V

-10 V

V D D

VSS

*                   |   |  |   |   Output 

*                   |   |  |   |   |  

.SUBCKT LM224       1   2  11  12  24 

 

***** Input Stage ***** 

Q_Q1 4 5 6 QPNP1     

Q_Q2 7 8 9 QPNP2 

I_I1 111 10 DC 1m 

R_RC1 4 12 95.49  

R_RC2 7 12 95.49 

R_RE1 10 6 43.79  

R_RE2 10 9 43.79 

V_Vio 2 8 0Vdc 

E_E4 1 5 16 14 1 

E_EVcc 111 0 11 0 1 

G_Vcc 0 11 poly(1) 11 0 0.278m 2.4u 

 

***** Gain Stage & Frequency Response Stage ***** 

R_R3 13 12 1000   

R_R4 111 13 100k   

E_Eref 14 0 13 0 1 

G_G1 14 15 7 4  0.01047 

R_Rc 14 15 9.549Meg  

C_Cc 14 15 1.667n  

 

***** Output Stage ***** 

E_E1 22 14 15 14 1 

V_F1 23 24 0 

F_F1 11 0 V_F1  1 

R_Rout 22 23 13 

 

***** Common Mode Rejection ***** 

R_R1 3 1 1Meg 

R_R2 2 3 1Meg 

G_G2 14 16 3 14 5.6234n 

R_Rcmr 17 16 10k 

L_Lcmr 14 17 1.59H  

 

***** Output Voltage Limiting ***** 

D_D1 15 18 D10D1 

D_D2 19 15 D10D1  

V_Voh 111 18 2.19 

V_Vol 19 12 0.63 

 

***** Output Current Limiting ***** 

D_D3 15 21 D10D1  

D_D4 20 15 D10D1  

V_V3 21 23 -0.207 

V_V4 23 20 -0.467 

 

***** Models ***** 

.model QPNP1 PNP(Bf=6041.7) 

.model QPNP2 PNP(Bf=5186.8) 

.MODEL D10D1 D IS=1E-15 RS=1.000E-3 VJ=.75 BV=100E6 

.ENDS 

 

.END 

Pojačanje je 6, jer je A=Vout/(Vin2-Vin1)=(20k/10k)*(1+(2*10k/10k)). 

 

ZADATAK 47. U programu PSpice opisati potkolo NAND logičkog gejta u CMOS 

tehnologiji. Dimenzije tranzistora su: L=5 μm, W=3 μm. Vrednost napona generatora VDD je 

5 V. Dati su SPICE modeli tranzistora: 

.MODEL NMOD NMOS(level=2 vto=1 nsub=1e16 tox=8.5e-8 uo=750 

+ cgso=4e-10 cgdo=4e-10 cgbo=2e-10 uexp=0.14 ucrit=5e4 utra=0 

vmax=5e4 rsh=15 

+ cj=4e-4 mj=2 pb=0.7 cjsw=8e-10 mjsw=2 js=1e-6 xj=1u ld=0.7u) 

 

.MODEL PMOD PMOS(level=2 vto=-1 nsub=2e15 tox=8.5e-8 uo=250 

+ cgso=4e-10 cgdo=4e-10 cgbo=2e-10 uexp=0.03 ucrit=1e4 utra=0 

vmax=3e4 rsh=75 

+ cj=1.8e-4 mj=2 pb=0.7 cjsw=6e-10 mjsw=2 js=1e-6 xj=0.9u ld=0.6u) 

 

Korišćenjem formiranog potkola, realizovati logičku mrežu sa slike 125 i izvršiti simulaciju 

u vremenskom domenu. Odrediti kombinaciju ulaznih signala za koju je kašnjenje kola 

maksimalno. Karakteristike promenljivog impulsnog signala su: nizak nivo 0 V, nivo logičke 

jedinice 5 V, vreme porasta i opadanja signala je 100 ns, perioda je 100 ms, a duty cycle 

50%. 

 

   

Slika 125. NAND logičko kolo 

Rešenje: 

*** 

.options numdgt=8 reltol=0.01 itl2=120 stepgmin gmin=1e-16 

 

VDD 6 0 5 

VA 1 0 PULSE (0 5V 20m 100ns 100ns 50m 100m) 

X1 1 1 2 6 0 NAND 

M 1

M 2

M 3 M 4

V D D

A B

A

B

F

x 1

x 2

x 3

x 4

1 0 0 n F

X2 1 1 3 6 0 NAND 

X3 2 3 4 6 0 NAND 

C1 4 0 100n 

 

 

.SUBCKT NAND A B F VDD GND 

 

M1 F A 5 GND NMOD L=5E-6 W=3E-6 

M2 5 B GND GND NMOD L=5E-6 W=3E-6 

M3 F A VDD VDD PMOD L=5E-6 W=3E-6 

M4 F B VDD VDD PMOD L=5E-6 W=3E-6 

 

.MODEL NMOD NMOS( level=2 vto=1 nsub=1e16 tox=8.5e-8 uo=750 

+ cgso=4e-10 cgdo=4e-10 cgbo=2e-10 uexp=0.14 ucrit=5e4 utra=0 

vmax=5e4 rsh=15 

+ cj=4e-4 mj=2 pb=0.7 cjsw=8e-10 mjsw=2 js=1e-6 xj=1u ld=0.7u ) 

 

.MODEL PMOD PMOS( level=2 vto=-1 nsub=2e15 tox=8.5e-8 uo=250 

+ cgso=4e-10 cgdo=4e-10 cgbo=2e-10 uexp=0.03 ucrit=1e4 utra=0 

vmax=3e4 rsh=75 

+ cj=1.8e-4 mj=2 pb=0.7 cjsw=6e-10 mjsw=2 js=1e-6 xj=0.9u ld=0.6u ) 

 

 

.ENDS 

 

.TRAN 1n 150m UIC 

 

.PROBE V(1,0) V(4,0)  

 

.END 

 

 

 


